

Dynamic Epistemic Modelling

Jan van Eijck

CWI and ILLC, Amsterdam, Uil-OTS, Utrecht

Version 1.03, Summer 2005

Contents

1	Introduction	2
2	Definitions	11
2.1	Models and Updates	11
2.2	Operations on Action Models	13
2.3	Automata	14
2.4	Logics for Communication	16
3	Kripke Models	18
3.1	Module Declaration	18
3.2	Agents	18
3.3	Model Datatype	19
4	Display and Visualisation	23
4.1	Module Declaration	23
4.2	Representing Accessibility Relations	23
4.3	Model Display	29
4.4	Model Visualisation	31
5	Model Minimization under Bisimulation	37
5.1	Module Declaration	37
5.2	Partition Refinement	37
5.3	Minimization	39
6	Formulas, Action Models and Epistemic Models	42

6.1	Module Declaration	42
6.2	Formulas	42
6.3	Reducing Formulas to Canonical Form	48
6.4	Action Models and Epistemic Models	53
6.5	Program Transformation	55
6.6	Automata	61
7	Model Minimization under Action Emulation	70
7.1	Module Declaration	70
7.2	Action Emulation	70
7.3	Partition Refinement Again	71
8	Semantics	74
8.1	Module Declaration	74
8.2	Semantics	75
8.3	Tools for Constructing Epistemic Models	78
8.4	From Communicative Actions to Action Models	79
8.5	Operations on Action Models	83
9	Main Module	91
9.1	Module Declaration	91
9.2	Version	92
10	Examples	93
10.1	The Riddle of the Caps	93
10.2	Muddy Children	97
10.3	The Riddle of Sum and Product	102
10.4	Sums and Sums-of-Squares	106
10.5	Card Showing	110
10.6	Elements of Secure Communication	116
10.7	Public/secret key cryptography	118
10.8	The Russian Cards Problem	120

10.9 Update Semantics	128
10.10The Protocol of the Dining Cryptographers	129
10.11A Measure for Ignorance	134
10.12Finding Axiom Schemes for Logics of Communication	135
A Special Treatment for S5, KD45 and K45	137
B The DPLL prover	139
B.1 Module Declaration	139
B.2 Clauses, Clause Sets	139
B.3 Tries	140
B.4 Unit Subsumption and Unit Resolution	142
B.5 Splitting	144
B.6 DPLL	145
Index	148

Abstract

This report introduces and documents *DEMO*, a Dynamic Epistemic Modelling tool. *DEMO* allows modelling epistemic updates, graphical display of update results, graphical display of action models, formula evaluation in epistemic models, translation of dynamic epistemic formulas to PDL formulas, and so on. *DEMO* implements the reduction of dynamic epistemic logic [21, 2, 3, 1] to PDL given in [16]. The reduction of dynamic epistemic logic to automata PDL from [30] is also discussed and implemented. Epistemic models are minimized under bisimulation, and update action models are minimized under action emulation (the appropriate structural notion for having the same update effect, cf. [18]). The report is an exemplar of tool building for epistemic update logic. It contains the full code of an implementation in Haskell [28], in ‘literate programming’ style [29], of *DEMO*.

Keywords: Knowledge representation, epistemic updates, dynamic epistemic modelling, action models. information change, logic of communication.

ACM Classification (1998) E 4, F 4.1, H 1.1.

Chapter 1

Introduction

In this introduction we will demonstrate how *DEMO*, which is short for *Dynamic Epistemic MOdelling*,¹ can be used to check semantic intuitions about what goes on in epistemic update situations.² For didactic purposes, the initial examples have been kept extremely simple. Although the situation of message passing about just two basic propositions with just three epistemic agents already reveals many subtleties, the reader should bear in mind that *DEMO* is capable of modelling much more complex situations.


In a situation where you and I know nothing about a particular aspect of the state of the world (about whether p and q hold, say), our state of knowledge is modelled by a Kripke model where the worlds are the four different possibilities for the truth of p and q (\emptyset , p , q , pq), your epistemic accessibility relation \sim_a is the total relation on these four possibilities, and mine \sim_b is the total relation on these four possibilities as well. There is also c , who like the two of us, is completely ignorant about p and q . This initial model is generated by *DEMO* as follows:

```
DEMO> showM (initE [P 0,Q 0])
==> [0,1,2,3]
[0,1,2,3]
(0,[]) (1,[p]) (2,[q]) (3,[p,q])
(a,[[0,1,2,3]])
(b,[[0,1,2,3]])
(c,[[0,1,2,3]])
```

Here is another representation of this same model. This representation can be generated with *dot* [32] from the file produced by the *DEMO* command `writeP "filename" (initE [P 0,Q 0])`.

¹Or short for *DEMO of Epistemic MOdelling*, for those who prefer co-recursive acronyms.

²The program source code is available from <http://www.cwi.nl/~jve/demo/>.




This is a model where none of the three agents a , b or c can distinguish between the four possibilities about p and q . *DEMO* shows the partitions generated by the accessibility relations \sim_a, \sim_b, \sim_c . Since these three relations are total, the three partitions each consist of a single block. Call this model **e0**.

Now suppose a wants to know whether p is the case. She asks whether p and receives a truthful answer from somebody who is in a position to know. This answer is conveyed to a in a message. b and c have heard a 's question, and so are aware of the fact that an answer may have reached a . b and c have not seen *that* an answer was delivered. This is not a secret communication, for b and c know that a has inquired about p . The situation now changes as follows:

```
DEMO> showM (upd e0 (message a p))
==> [1,4]
[0,1,2,3,4,5]
(0, []) (1, [p]) (2, [p]) (3, [q]) (4, [p,q])
(5, [p,q])
(a, [[0,2,3,5], [1,4]])
(b, [[0,1,2,3,4,5]])
(c, [[0,1,2,3,4,5]])
```

This is again a model where the three accessibility relations are equivalences, but one in which a has restricted her range of possibilities to 1, 4 (these are worlds where p is the case), while for b and c all possibilities are still open.

In graphical display format:



Notice that in this new situation some subtle things have changed for b and c as well. Before the arrival of the message, $\Box_b \neg \Box_a p$ was true, for b knew that a did not know about p . But now b has heard a 's question about p , and is aware of the fact that an answer may have reached a . So in the new situation b is not sure anymore about what a knows about p . In other words, $\Box_b \neg \Box_a p$ has become false. On the other hand it is still the case that b knows that a knows nothing about q : $\Box_b \neg \Box_a q$ is still true in the new situation. The situation for c is similar to that for b . These things can be checked in *DEMO* as follows:

```
DEMO> isTrue (upd e0 (message a p)) (K b (Neg (K a q)))
True
DEMO> isTrue (upd e0 (message a p)) (K b (Neg (K a p)))
False
```

If you receive the same message about p twice, the second time the message gets delivered has no further effect:

```
DEMO> showM (upds e0 [message a p, message a p])
==> [1,4]
```

```

[0,1,2,3,4,5]
(0, []) (1, [p]) (2, [p]) (3, [q]) (4, [p,q])
(5, [p,q])
(a, [[0,2,3,5], [1,4]])
(b, [[0,1,2,3,4,5]])
(c, [[0,1,2,3,4,5]])

```

Now suppose that the second action is a message informing b about p :


```

DEMO> showM (upds e0 [message a p, message b p])
==> [1,6]
[0,1,2,3,4,5,6,7,8,9]
(0, []) (1, [p]) (2, [p]) (3, [p]) (4, [p])
(5, [q]) (6, [p,q]) (7, [p,q]) (8, [p,q]) (9, [p,q])

(a, [[0,3,4,5,8,9], [1,2,6,7]])
(b, [[0,2,4,5,7,9], [1,3,6,8]])
(c, [[0,1,2,3,4,5,6,7,8,9]])

```

The graphical representation of this model is slightly more difficult to fathom at a glance.



In this model a and b both know about p , but they do not know about each other's knowledge about p . c still knows nothing, and both a and b know that c knows nothing. Both $\Box_a\Box_bp$ and $\Box_b\Box_ap$ are false in this model. $\Box_a\neg\Box_bp$ and $\Box_b\neg\Box_ap$ are false as well, but $\Box_a\neg\Box_cp$ and $\Box_b\neg\Box_cp$ are true.

```
DEMO> isTrue (upds e0 [message a p, message b p]) (K a (K b p))
False
DEMO> isTrue (upds e0 [message a p, message b p]) (K b (K a p))
False
DEMO> isTrue (upds e0 [message a p, message b p]) (K b (Neg (K b p)))
False
DEMO> isTrue (upds e0 [message a p, message b p]) (K b (Neg (K c p)))
True
```

The order in which a and b are informed does not matter:

```
DEMO> showM (upds e0 [message b p, message a p])
==> [1,6]
[0,1,2,3,4,5,6,7,8,9]
(0, []) (1, [p]) (2, [p]) (3, [p]) (4, [p])
(5, [q]) (6, [p,q]) (7, [p,q]) (8, [p,q]) (9, [p,q])

(a, [[0,2,4,5,7,9], [1,3,6,8]])
(b, [[0,3,4,5,8,9], [1,2,6,7]])
(c, [[0,1,2,3,4,5,6,7,8,9]])
```

Modulo renaming this is the same as the earlier result. The example shows that the epistemic effects of distributed message passing are quite different from those of a public announcement or a group message.

```
DEMO> showM (upd e0 (public p))
==> [0,1]
[0,1]
(0, [p]) (1, [p,q])
(a, [[0,1]])
(b, [[0,1]])
(c, [[0,1]])
```

The result of the public announcement that p is that a , b and c are informed that p and about each other's knowledge about p .

DEMO allows to compare the action models for public announcement and individual message passing:

```
DEMO> showM (public p)
==> [0]
[0]
```

```

(0,p)
(a,[[0]])
(b,[[0]])
(c,[[0]])

DEMO> showM (cmp [message a p, message b p, message c p])
==> [0]
[0,1,2,3,4,5,6,7]
(0,p)(1,p)(2,p)(3,p)(4,p)
(5,p)(6,p)(7,T)
(a,[[0,1,2,3],[4,5,6,7]])
(b,[[0,1,4,5],[2,3,6,7]])
(c,[[0,2,4,6],[1,3,5,7]])

```

Here `cmp` gives the sequential composition of a list of communicative actions.

More subtly, the situation is also different from a situation where a, b receive the same message that p , with a being aware of the fact that b receives the message and vice versa. Such group messages create common knowledge:

```

DEMO> showM (groupM [a,b] p)
==> [0]
[0,1]
(0,p)(1,T)
(a,[[0],[1]])
(b,[[0],[1]])
(c,[[0,1]])

```

The difference with the case of the two separate messages is that now a and b are aware of each other's knowledge that p :

```

DEMO> isTrue (upd e0 (groupM [a,b] p)) (K a (K b p))
True
DEMO> isTrue (upd e0 (groupM [a,b] p)) (K b (K a p))
True

```

Next, look at the case where two separate messages reach a and b , one informing a that p and the other informing b that $\neg q$:

```

DEMO> showM (upds e0 [message a p, message b (Neg q)])
==> [2]
[0,1,2,3,4,5,6,7,8]
(0,[])(1,[])(2,[p])(3,[p])(4,[p])
(5,[p])(6,[q])(7,[p,q])(8,[p,q])
(a,[[0,1,4,5,6,8],[2,3,7]])
(b,[[0,2,4],[1,3,5,6,7,8]])
(c,[[0,1,2,3,4,5,6,7,8]])

```

Again the order in which these messages are delivered is immaterial for the end result, as you should expect:

```
DEMO> showM (upds e0 [message b (Neg q), message a p])
==> [2]
[0,1,2,3,4,5,6,7,8]
(0, []) (1, []) (2, [p]) (3, [p]) (4, [p])
(5, [p]) (6, [q]) (7, [p,q]) (8, [p,q])
(a, [[0,1,3,5,6,8], [2,4,7]])
(b, [[0,2,3], [1,4,5,6,7,8]])
(c, [[0,1,2,3,4,5,6,7,8]])
```

Modulo a renaming of worlds, this is the same as the previous result.

The logic of public announcements and private messages is related to the so-called logic of knowledge [24]. This logic satisfies the following postulates:

- knowledge distribution $\Box_a(\varphi \Rightarrow \psi) \Rightarrow (\Box_a\varphi \Rightarrow \Box_a\psi)$ (if a knows that φ implies ψ , and she knows φ , then she also knows ψ),
- positive introspection $\Box_a\varphi \Rightarrow \Box_a\Box_a\varphi$ (if a knows φ , then a knows that she knows φ),
- negative introspection $\neg\Box_a\varphi \Rightarrow \Box_a\neg\Box_a\varphi$ (if a does not know φ , then she knows that she does not know),
- truthfulness $\Box_a\varphi \Rightarrow \varphi$ (if a knows φ then φ is true).

As is well known, the first of these is valid on all Kripke frames, the second is valid on precisely the transitive Kripke frames, the third is valid on precisely the euclidean Kripke frames (a relation R is euclidean if it satisfies $\forall x\forall y\forall z((xRy \wedge xRz) \Rightarrow yRz)$), and the fourth is valid on precisely the reflexive Kripke frames. A frame satisfies transitivity, euclideaness and reflexivity iff it is an equivalence relation, hence the logic of knowledge is the logic of the so-called S5 Kripke frames: the Kripke frames with an equivalence \sim_a as epistemic accessibility relation. Multi-agent epistemic logic extends this to multi-S5, with an equivalence \sim_b for every $b \in B$, where B is the set of epistemic agents.

Now suppose that instead of open messages, we use *secret* messages. If a secret message is passed to a , b and c are not even aware that any communication is going on. This is the result when a receives a secret message that p in the initial situation:

```
DEMO> showM (upd e0 (secret [a] p))
==> [1,4]
[0,1,2,3,4,5]
(0, []) (1, [p]) (2, [p]) (3, [q]) (4, [p,q])
(5, [p,q])
(a, [[[]], [0,2,3,5]], ([], [1,4]))
(b, [[1,4], [0,2,3,5]])
(c, [[1,4], [0,2,3,5]])
```

This is not an S5 model anymore. The accessibility for a is still an equivalence, but the accessibility for b is lacking the property of reflexivity. The worlds 1,4 that make up a 's conceptual space (for these are the worlds accessible for a from the actual world 1) are precisely the worlds where the b and c arrows are not reflexive. b enters his conceptual space from the vantage point 1, but b does not see the actual world itself. Similarly for c . In the *DEMO* representation, a list $(\mathbf{xs}, \mathbf{ys})$ gives the entry points \mathbf{xs} into conceptual space \mathbf{ys} .

The secret message has no effect on what b and c believe about the facts of the world, but it has effected b 's and c 's beliefs about the beliefs of a in a disastrous way. These beliefs have become inaccurate. For instance, b now believes that a does *not* know that p , but he is mistaken! The formula $\Box_b \neg \Box_a p$ is true in the actual world, but $\neg \Box_a p$ is false in the actual world, for a *does* know that p , because of the secret message. Here is what *DEMO* says about the situation:

```
DEMO> isTrue (upd e0 (secret [a] p)) (K b (Neg (K a p)))
True
DEMO> isTrue (upd e0 (secret [a] p)) (Neg (K a p))
False
```

This illustrates a regress from the world of knowledge to the world of consistent belief: the result of the update with a secret propositional message does not satisfy the postulate of truthfulness anymore.

The logic of consistent belief satisfies the following postulates:

- knowledge distribution $\Box_a(\varphi \Rightarrow \psi) \Rightarrow (\Box_a\varphi \Rightarrow \Box_a\psi)$,
- positive introspection $\Box_a\varphi \Rightarrow \Box_a\Box_a\varphi$,
- negative introspection $\neg\Box_a\varphi \Rightarrow \Box_a\neg\Box_a\varphi$,
- consistency $\Box_a\varphi \Rightarrow \Diamond_a\varphi$ (if a believes that φ then there is a world where φ is true, i.e., φ is consistent).

Consistent belief is like knowledge, except for the fact that it replaces the postulate of truthfulness $\Box_a\varphi \Rightarrow \varphi$ by the weaker postulate of consistency.

Since the postulate of consistency determines the serial Kripke frames (a relation R is serial if $\forall x \exists y xRy$), the principles of consistent belief determine the Kripke frames that are transitive, euclidean and serial, the so-called KD45 frames.

In the conceptual world of secrecy, inconsistent beliefs are not far away. Suppose that a , after having received a secret message informing her about p , sends a message to b to the effect that $\Box_a p$. The trouble is that this is *inconsistent* with what b believes.

```
DEMO> showM (upds e0 [secret [a] p, message b (K a p)])
==> [1,5]
[0,1,2,3,4,5,6,7]
(0, []) (1, [p]) (2, [p]) (3, [p]) (4, [q])
(5, [p,q]) (6, [p,q]) (7, [p,q])
```

$(a, ([], [([[], [0, 3, 4, 7]), ([[], [1, 2, 5, 6])]))$
 $(b, ([1, 5], [([2, 6], [0, 3, 4, 7])]))$
 $(c, ([], [([1, 2, 5, 6], [0, 3, 4, 7])]))$

This is not a KD45 model anymore, for it lacks the property of seriality for b 's belief relation. b 's belief contains two isolated worlds 1, 5. Since 1 is the actual world, this means that b 's belief state has become inconsistent: from now on, b will believe *anything*.

So we have arrived at a still weaker logic. The logic of possibly inconsistent belief satisfies the following postulates:

- knowledge distribution $\Box_a(\varphi \Rightarrow \psi) \Rightarrow (\Box_a\varphi \Rightarrow \Box_a\psi)$,
- positive introspection $\Box_a\varphi \Rightarrow \Box_a\Box_a\varphi$,
- negative introspection $\neg\Box_a\varphi \Rightarrow \Box_a\neg\Box_a\varphi$.

This is the logic of K45 frames: frames that are transitive and euclidean.

In [15] some results and a list of questions are given about the possible deterioration of knowledge and belief caused by different kind of message passing. E.g., the result of updating an S5 model with a public announcement or a non-secret message, if defined, is again S5. The result of updating an S5 model with a secret message to some of the agents, if defined, need not even be KD45. One can prove that the result is KD45 iff the model we start out with satisfies certain epistemic conditions. The update result always is K45. Such observations illustrate why S5, KD45 and K45 are ubiquitous in epistemic modelling. See [6, 22] for general background on modal logic, and [8, 19] for specific background on these systems.

If this introduction has convinced the reader that the logic of public announcements, private messages and secret communications is rich and subtle enough to justify the building of the conceptual modelling tools to be presented in the rest of the report, then it has served its purpose.

In the rest of the report, we first fix a formal version of epistemic update logic as an implementation goal. After that, we are ready for the implementation.

Some facts about S5, KD45 and K45 relations that are used in the implementation for presenting S5, KD45 and K45 models in a perspicuous way are given in Appendix A.

Further information on various aspects of dynamic epistemic logic is provided in [1, 2, 4, 5, 11, 19, 20, 31].

Chapter 2

Definitions

2.1 Models and Updates

In this section we formalize the version of dynamic epistemic logic that we are going to implement.

Let p range over a set of basic propositions P and let a range over a set of agents Ag . Then the language of PDL over P, Ag is given by:

$$\begin{aligned}\varphi &::= \top \mid p \mid \neg\varphi \mid \varphi_1 \wedge \varphi_2 \mid [\pi]\varphi \\ \pi &::= a \mid ?\varphi \mid \pi_1; \pi_2 \mid \pi_1 \cup \pi_2 \mid \pi^*\end{aligned}$$

Employ the usual abbreviations: \perp is shorthand for $\neg\top$, $\varphi_1 \vee \varphi_2$ is shorthand for $\neg(\neg\varphi_1 \wedge \neg\varphi_2)$, $\varphi_1 \rightarrow \varphi_2$ is shorthand for $\neg(\varphi_1 \wedge \neg\varphi_2)$, $\varphi_1 \leftrightarrow \varphi_2$ is shorthand for $(\varphi_1 \rightarrow \varphi_2) \wedge (\varphi_2 \rightarrow \varphi_1)$, and $\langle\pi\rangle\varphi$ is shorthand for $\neg[\pi]\neg\varphi$. Also, if $B \subseteq Ag$ and B is finite, use B as shorthand for $b_1 \cup b_2 \cup \dots$. Under this convention, the general knowledge operator $E_B\varphi$ takes the shape $[B]\varphi$, while the common knowledge operator $C_B\varphi$ appears as $[B^*]\varphi$, i.e., $[B]\varphi$ expresses that it is general knowledge among agents B that φ , and $[B^*]\varphi$ expresses that it is common knowledge among agents B that φ . In the special case where $B = \emptyset$, B turns out equivalent to $?\perp$, the program that always fails.

The semantics of PDL over P, Ag is given relative to labelled transition systems $\mathbf{M} = (W, V, R)$, where W is a set of worlds (or states), $V : W \rightarrow \mathcal{P}(P)$ is a valuation function, and $R = \{\xrightarrow{a} \subseteq W \times W \mid a \in Ag\}$ is a set of labelled transitions, i.e., binary relations on W , one for each label a . In what follows, we will take the labeled transitions for a to represent the epistemic alternatives of an agent a .

The formulae of PDL are interpreted as subsets of $W_{\mathbf{M}}$ (the state set of \mathbf{M}), the actions of PDL

as binary relations on $W_{\mathbf{M}}$, as follows:

$$\begin{aligned}
\llbracket \top \rrbracket^{\mathbf{M}} &= W_{\mathbf{M}} \\
\llbracket p \rrbracket^{\mathbf{M}} &= \{w \in W_{\mathbf{M}} \mid p \in V_{\mathbf{M}}(w)\} \\
\llbracket \neg \varphi \rrbracket^{\mathbf{M}} &= W_{\mathbf{M}} - \llbracket \varphi \rrbracket^{\mathbf{M}} \\
\llbracket \varphi_1 \wedge \varphi_2 \rrbracket^{\mathbf{M}} &= \llbracket \varphi_1 \rrbracket^{\mathbf{M}} \cap \llbracket \varphi_2 \rrbracket^{\mathbf{M}} \\
\llbracket [\pi] \varphi \rrbracket^{\mathbf{M}} &= \{w \in W_{\mathbf{M}} \mid \forall v (\text{if } (w, v) \in \llbracket \pi \rrbracket^{\mathbf{M}} \text{ then } v \in \llbracket \varphi \rrbracket^{\mathbf{M}}) \} \\
\llbracket a \rrbracket^{\mathbf{M}} &= \xrightarrow{a}_{\mathbf{M}} \\
\llbracket ?\varphi \rrbracket^{\mathbf{M}} &= \{(w, w) \in W_{\mathbf{M}} \times W_{\mathbf{M}} \mid w \in \llbracket \varphi \rrbracket^{\mathbf{M}}\} \\
\llbracket \pi_1; \pi_2 \rrbracket^{\mathbf{M}} &= \llbracket \pi_1 \rrbracket^{\mathbf{M}} \circ \llbracket \pi_2 \rrbracket^{\mathbf{M}} \\
\llbracket \pi_1 \cup \pi_2 \rrbracket^{\mathbf{M}} &= \llbracket \pi_1 \rrbracket^{\mathbf{M}} \cup \llbracket \pi_2 \rrbracket^{\mathbf{M}} \\
\llbracket \pi^* \rrbracket^{\mathbf{M}} &= (\llbracket \pi \rrbracket^{\mathbf{M}})^*
\end{aligned}$$

If $w \in W_{\mathbf{M}}$ then we use $\mathbf{M} \models_w \varphi$ for $w \in \llbracket \varphi \rrbracket^{\mathbf{M}}$.

[3] proposes to model epistemic actions as epistemic models, with valuations replaced by preconditions. See also: [4, 5, 11, 16, 19, 20, 31, 38].

Action models for a given language \mathcal{L} Let a set of agents Ag and an epistemic language \mathcal{L} be given. An action model for \mathcal{L} is a triple $A = ([s_0, \dots, s_{n-1}], \text{pre}, T)$ where $[s_0, \dots, s_{n-1}]$ is a finite list of action states, $\text{pre} : \{s_0, \dots, s_{n-1}\} \rightarrow \mathcal{L}$ assigns a precondition to each action state, and $T : Ag \rightarrow \mathcal{P}(\{s_0, \dots, s_{n-1}\}^2)$ assigns an accessibility relation \xrightarrow{a} to each agent $a \in Ag$.

A pair $\mathbf{A} = (A, s)$ with $s \in \{s_0, \dots, s_{n-1}\}$ is a pointed action model, where s is the action that actually takes place.

The list ordering of the action states in an action model will play an important role in the definition of the program transformations associated with the action models.

In the definition of action models, \mathcal{L} can be any language that can be interpreted in PDL models. Actions can be executed in PDL models by means of the following product construction:

Action Update Let a PDL model $\mathbf{M} = (W, V, R)$, a world $w \in W$, and a pointed action model (A, s) , with $A = ([s_0, \dots, s_{n-1}], \text{pre}, T)$, be given. Then the result of executing (A, s) in (\mathbf{M}, w) is the model $(\mathbf{M} \otimes A, (w, s))$, with $\mathbf{M} \otimes A = (W', V', R')$, where

$$\begin{aligned}
W' &= \{(w, s) \mid s \in \{s_0, \dots, s_{n-1}\}, w \in \llbracket \text{pre}(s) \rrbracket^{\mathbf{M}}\} \\
V'(w, s) &= V(w) \\
R'(a) &= \{((w, s), (w', s')) \mid (w, w') \in R(a), (s, s') \in T(a)\}.
\end{aligned}$$

The language of PDL^{DEL} (update PDL) is given by extending the PDL language with update constructions $[A, s]\varphi$, where (A, s) is a pointed action model. The interpretation of $[A, s]\varphi$ in \mathbf{M} is given by:

$$\llbracket [A, s]\varphi \rrbracket^{\mathbf{M}} = \{w \in W_{\mathbf{M}} \mid \text{if } \mathbf{M} \models_w \text{pre}(s) \text{ then } (w, s) \in \llbracket \varphi \rrbracket^{\mathbf{M} \otimes A}\}.$$

Using $\langle A, s \rangle \varphi$ as shorthand for $\neg[A, s]\neg\varphi$, we see that the interpretation for $\langle A, s \rangle \varphi$ turns out as:

$$\llbracket \langle A, s \rangle \varphi \rrbracket^{\mathbf{M}} = \{w \in W_{\mathbf{M}} \mid \mathbf{M} \models_w \text{pre}(s) \text{ and } (w, s) \in \llbracket \varphi \rrbracket^{\mathbf{M} \otimes A}\}.$$

Updating with multiple pointed update actions is also possible. A multiple pointed action is a pair (A, S) , with A an action model, and S a subset of the state set of A . Extend the language with updates $[A, S]\varphi$, and interpret this as follows:

$$\llbracket [A, S]\varphi \rrbracket^{\mathbf{M}} = \{w \in W_{\mathbf{M}} \mid \forall s \in S \text{ (if } \mathbf{M} \models_w \text{pre}(s) \text{ then } \mathbf{M} \otimes A \models_{(w,s)} \varphi)\}.$$

In [16] it is shown how dynamic epistemic logic can be reduced to PDL by program transformation. Each action model \mathbf{A} has associated program transformers $T_{ij}^{\mathbf{A}}$ for all states s_i, s_j in the action model, such that the following hold:

Lemma 1 (Program Transformation, Van Eijck [16]) *Assume A has n states s_0, \dots, s_{n-1} . Then:*

$$\mathbf{M} \models_w [A, s_i][\pi]\varphi \text{ iff } \mathbf{M} \models_w \bigwedge_{j=0}^{n-1} [T_{ij}^A(\pi)][A, s_j]\varphi.$$

This lemma allows a reduction of dynamic epistemic logic to PDL, a reduction that we will implement in the code below.

2.2 Operations on Action Models

Sequential Composition If (\mathbf{A}, S) and (\mathbf{B}, T) are multiple pointed action models, their sequential composition $(\mathbf{A}, S) \odot (\mathbf{B}, T)$ is given by:

$$(\mathbf{A}, S) \odot (\mathbf{B}, T) := ((W, \text{pre}, R), S \times T),$$

where

- $W = W_{\mathbf{A}} \times W_{\mathbf{B}}$,
- $\text{pre}(s, t) = \text{pre}(s) \wedge \langle \mathbf{A}, S \rangle \text{pre}(t)$,
- R is given by: $(s, t) \xrightarrow{a} (s', t') \in R$ iff $s \xrightarrow{a} s' \in R_{\mathbf{A}}$ and $t \xrightarrow{a} t' \in R_{\mathbf{B}}$.

The unit element for this operation is the action model

$$\mathbf{1} = ((\{0\}, 0 \mapsto \top, \{0 \xrightarrow{a} 0 \mid a \in Ag\}), \{0\}).$$

Updating an arbitrary epistemic model \mathbf{M} with $\mathbf{1}$ changes nothing.

Non-deterministic Sum The non-deterministic sum \oplus of multiple-pointed action models (\mathbf{A}, S) and (\mathbf{B}, T) is the action model $(\mathbf{A}, S) \oplus (\mathbf{B}, T)$ is given by:

$$(\mathbf{A}, S) \oplus (\mathbf{B}, T) := ((W, \text{pre}, R), S \uplus T),$$

where \uplus denotes disjoint union, and where

- $W = W_{\mathbf{A}} \uplus W_{\mathbf{B}}$,
- $\text{pre} = \text{pre}_{\mathbf{A}} \uplus \text{pre}_{\mathbf{B}}$,
- $R = R_{\mathbf{A}} \uplus R_{\mathbf{B}}$.

The unit element for this operation is called $\mathbf{0}$: the multiple pointed action model given by $((\emptyset, \emptyset, \emptyset), \emptyset)$.

2.3 Automata

The reduction of dynamic epistemic logic to PDL from [16] was inspired by a more involved reduction to Automata PDL (PDL, with the atomic programs replaced by nondeterministic finite automata, cf. [23, Chapter 10.3]. That reduction is also implemented below. In this section we fix some terminology about automata.

The general knowledge and common knowledge operators can be encoded as automata. Define a nondeterministic finite automaton or NFA over alphabet Σ as a quadruple consisting of a set of states S , a start state $s \in S$, a set δ of transitions (u, σ, v) with $u \in S, v \in S$, and $\sigma \in \Sigma$, and a final state $f \in S$.

The language accepted by an automaton over Σ is the set of strings from Σ^* that the automaton recognizes (or: accepts), where (S, s, δ, f) recognizes the empty string ϵ iff $s = f$, and (S, s, δ, f) recognizes the string $(\sigma; \vec{\sigma})$ iff there is a $u \in S$ such that $(s, \sigma, u) \in \delta$ and (S, u, δ, f) recognizes $\vec{\sigma}$. If $\vec{\sigma}$ is accepted by Aut, we say that $\vec{\sigma} \in \text{Aut}$.

A NFA N with set of accept states F can be modelled as the set of NFAs $\{N_f \mid f \in F\}$, where N_f is like N except for the fact that it has a single final state f .

Some example automata that are relevant in the present context:

- The automaton for general knowledge among agents B is the automaton with start state 0, final state 1, and transitions $\{(0, b, 1) \mid b \in B\}$.
- The automaton for common knowledge among agents B is the automaton with start state 0, final state 0 and transitions $\{(0, b, 0) \mid b \in B\}$.
- The automaton for common knowledge relativised to φ among agents B (see [30]) is the automaton with start state 0, final state 1, and transitions $\{(0, ?\varphi, 1)\} \cup \{(1, b, 0) \mid b \in B\}$.

Following [30], we extend the language \mathcal{LANG} with formulas $[\text{Aut}]\varphi$, where Aut is an automaton. Let $\Sigma = A \cup \{?\varphi \mid \varphi \in \mathcal{LANG}\}$. Define the interpretation of a string $\vec{\sigma}$ from Σ^* in a model M as follows:

$$\begin{aligned} \llbracket \epsilon \rrbracket^M &= \{(w, w) \mid w \in W_M\} \\ \llbracket a; \vec{\sigma} \rrbracket^M &= \xrightarrow{a}_M \circ \llbracket \vec{\sigma} \rrbracket^M \\ \llbracket ?\varphi; \vec{\sigma} \rrbracket^M &= \{(w, w) \mid M \models_w \varphi\} \circ \llbracket \vec{\sigma} \rrbracket^M \end{aligned}$$

The truth definition for $[\text{Aut}]\varphi$ is now given by:

$$\begin{aligned} M \models_w [\text{Aut}]\varphi &:\equiv \text{ for all } v \in W_M \text{ and all } \vec{\sigma} \in \text{Aut} : \\ &\text{ if } (w, v) \in \llbracket \vec{\sigma} \rrbracket^M \text{ then } M \models_v \varphi. \end{aligned}$$

As it stands, it is not immediately clear how to implement this as a finite check, as both the list of strings accepted by an automaton and the list of paths between pairs of points in a model are generally infinite. In Section 8.2 we propose a modification of a graph reachability algorithm that computes the sets of worlds reachable from a given world in a model through a path accepted by a given automaton.

In the spirit of [30] (but with a slight modification), define a function AUT from quadruples consisting of an action model \mathbf{A} , a first state in \mathbf{A} , a second state in \mathbf{A} , and an automaton $\text{Aut} = (S, s, \delta, f)$, to automata, as follows:

$$\text{AUT}(\mathbf{A}, \mathbf{s}, \mathbf{t}, (S, s, \delta, f)) = (S', s', \delta', f')$$

where

$$\begin{aligned} S' &= S_{\mathbf{A}} \times \{0, 1\} \times S, \\ s' &= (\mathbf{s}, 0, s), \\ f' &= (\mathbf{t}, 1, f), \\ \delta' &= \{((\mathbf{u}, 1, t), a, (\mathbf{u}', 0, t')) \mid \mathbf{u} \xrightarrow{a} \mathbf{u}', (t, a, t') \in \delta\} \\ &\cup \{((\mathbf{u}, 0, t), ?\varphi, (\mathbf{u}, 1, t)) \mid \mathbf{u} \in S_{\mathbf{A}}, p_{\mathbf{u}} = \varphi, t \in S\} \\ &\cup \{((\mathbf{u}, 1, t), ?\langle \mathbf{A}, u \rangle \varphi, (\mathbf{u}, 1, t')) \mid \mathbf{u} \in S_{\mathbf{A}}, (t, ?\varphi, t') \in \delta\}. \end{aligned}$$

Then one can prove the following Lemma:

Lemma 2 (Reduction; Kooi and Van Benthem)

$$[\mathbf{A}, \mathbf{s}][[S, s, \delta, f]]\varphi \leftrightarrow \bigwedge_{\mathbf{t} \in S_{\mathbf{A}}} [\text{AUT}(\mathbf{A}, \mathbf{s}, \mathbf{t}, (S, s, \delta, f))][\mathbf{A}, \mathbf{t}]\varphi.$$

This lemma allows one to reduce epistemic action logic to APDL, the logic of PDL over automata [23, Chapter 10.3].

2.4 Logics for Communication

Here are some specific action models that can be used to define various languages of communication.

Public announcement of φ : action model $(\mathbf{S}, \{0\})$, with

$$S_{\mathbf{S}} = \{0\}, p_{\mathbf{S}} = 0 \mapsto \varphi, R_{\mathbf{S}} = \{0 \xrightarrow{a} 0 \mid a \in A\}.$$

Individual message to b that φ : action model $(\mathbf{S}, \{0\})$, with

$$S_{\mathbf{S}} = \{0, 1\}, p_{\mathbf{S}} = 0 \mapsto \varphi, 1 \mapsto \top, R_{\mathbf{S}} = \{0 \sim_a 1 \mid a \in A - \{b\}\}.$$

Group message to B that φ : action model $(\mathbf{S}, \{0\})$, with

$$S_{\mathbf{S}} = \{0, 1\}, p_{\mathbf{S}} = 0 \mapsto \varphi, 1 \mapsto \top, R_{\mathbf{S}} = \{0 \sim_a 1 \mid a \in A - B\}.$$

Secret individual communication to b that φ : action model $(\mathbf{S}, \{0\})$, with

$$\begin{aligned} S_{\mathbf{S}} &= \{0, 1\}, \\ p_{\mathbf{S}} &= 0 \mapsto \varphi, 1 \mapsto \top, \\ R_{\mathbf{S}} &= \{0 \xrightarrow{b} 0\} \cup \{0 \xrightarrow{a} 1 \mid a \in A - \{b\}\} \cup \{1 \xrightarrow{a} 1 \mid a \in A\}. \end{aligned}$$

Secret group communication to B that φ : action model $(\mathbf{S}, \{0\})$, with

$$\begin{aligned} S_{\mathbf{S}} &= \{0, 1\}, \\ p_{\mathbf{S}} &= 0 \mapsto \varphi, 1 \mapsto \top, \\ R_{\mathbf{S}} &= \{0 \xrightarrow{b} 0 \mid b \in B\} \cup \{0 \xrightarrow{a} 1 \mid a \in A - B\} \cup \{1 \xrightarrow{a} 1 \mid a \in A\}. \end{aligned}$$

Test of φ : action model $(\mathbf{S}, \{0\})$, with

$$S_{\mathbf{S}} = \{0, 1\}, p_{\mathbf{S}} = 0 \mapsto \varphi, 1 \mapsto \top, R_{\mathbf{S}} = \{0 \xrightarrow{a} 1 \mid a \in A\} \cup \{1 \xrightarrow{a} 1 \mid a \in A\}.$$

Individual revelation to b of a choice from $\{\varphi_1, \dots, \varphi_n\}$: action model $(\mathbf{S}, \{1, \dots, n\})$, with

$$\begin{aligned} S_{\mathbf{S}} &= \{1, \dots, n\}, \\ p_{\mathbf{S}} &= 1 \mapsto \varphi_1, \dots, n \mapsto \varphi_n, \\ R_{\mathbf{S}} &= \{s \xrightarrow{b} s \mid s \in S_{\mathbf{S}}\} \cup \{s \xrightarrow{a} s' \mid s, s' \in S_{\mathbf{S}}, a \in A - \{b\}\}. \end{aligned}$$

Group revelation to B of a choice from $\{\varphi_1, \dots, \varphi_n\}$: action model $(\mathbf{S}, \{1, \dots, n\})$, with

$$\begin{aligned} S_{\mathbf{S}} &= \{1, \dots, n\}, \\ p_{\mathbf{S}} &= 1 \mapsto \varphi_1, \dots, n \mapsto \varphi_n, \\ R_{\mathbf{S}} &= \{s \xrightarrow{b} s \mid s \in S_{\mathbf{S}}, b \in B\} \cup \{s \xrightarrow{a} s' \mid s, s' \in S_{\mathbf{S}}, a \in A - B\}. \end{aligned}$$

Transparent informedness of B about φ : action model $(\mathbf{S}, \{0, 1\})$, with

$$\begin{aligned} S_{\mathbf{S}} &= \{0, 1\}, \\ p_{\mathbf{S}} &= 0 \mapsto \varphi, 1 \mapsto \neg\varphi, \\ R_{\mathbf{S}} &= \{0 \xrightarrow{a} 0 \mid a \in A\} \cup \{0 \xrightarrow{a} 1 \mid a \in A - B\} \cup \{1 \xrightarrow{a} 0 \mid a \in A - B\} \cup \{1 \xrightarrow{a} 1 \mid a \in A\}. \end{aligned}$$

Transparent informedness of B about φ is the special case of a group revelation of B of a choice from $\{\varphi, \neg\varphi\}$. Note that all but the revelation action models and the transparent informedness action models are single pointed (their sets of actual states are singletons).

The language for the logic of public announcements:

$$\begin{aligned} \varphi &::= \top \mid p \mid \neg\varphi \mid \bigwedge[\varphi_1, \dots, \varphi_n] \mid \bigvee[\varphi_1, \dots, \varphi_n] \mid \Box_a\varphi \mid E_B\varphi \mid C_B\varphi \mid [\pi]\varphi \\ \pi &::= \mathbf{1} \mid \mathbf{0} \mid \text{public } B \varphi \mid \odot[\pi_1, \dots, \pi_n] \mid \oplus[\pi_1, \dots, \pi_n] \end{aligned}$$

Semantics for this: use the semantics of $\mathbf{1}$, $\mathbf{0}$, **public** $B \varphi$, and the operations on multiple pointed action models from Section 2.2.

The logic of tests and public announcements: as above, but now also allowing tests $?\varphi$ as basic programs. Semantics: add the semantics of $?\varphi$ to the above repertoire.

The logic of individual messages: as above, but now the basic actions are messages to individual agents. Semantics: start out from the semantics of **message** $a \varphi$.

The logic of tests, public announcements, and group revelations as above, but now also allowing revelations from alternatives. Semantics: use the semantics of **reveal** $B \{\varphi_1, \dots, \varphi_n\}$.

Chapter 3

Kripke Models

3.1 Module Declaration

```
module Models where

import List
```

3.2 Agents

```
data Agent = A | B | C | D | E deriving (Eq,Ord,Enum)
```

Give the agents appropriate names:

```
a, alice, b, bob, c, carol, d, dave, e, ernie  :: Agent
a = A; alice = A
b = B; bob   = B
c = C; carol = C
d = D; dave  = D
e = E; ernie = E
```

Make agents showable in an appropriate way:

```
instance Show Agent where
  show A = "a"; show B = "b"; show C = "c"; show D = "d" ; show E = "e"
```

A function for listing all agents (uncomment the appropriate definition of `last_agent`, depending on the number of agents you need):

```
all_agents :: [Agent]
all_agents = [a .. last_agent]

last_agent :: Agent
--last_agent = a
last_agent = b
--last_agent = c
--last_agent = d
--last_agent = e
```

3.3 Model Datatype

It will prove useful to generalize over states. We first define general models, and then specialize to action models and static models. In the following definition, `state` and `formula` are variables over types.

```
data Model state formula = Mo
    [state]
    [(state,formula)]
    [(Agent,state,state)]
    deriving (Eq,Ord,Show)
```

Model with a number of singled-out points:


```

data Pmod state formula = Pmod
    [state]
    [(state,formula)]
    [(Agent,state,state)]
    [state]
    deriving (Eq,Ord,Show)

```

Creating a pointed model from a model and a list of points:

```

mod2pmod :: Model state formula -> [state] -> Pmod state formula
mod2pmod (Mo states prec accs) points = Pmod states prec accs points

```

Separating a pointed model into model and points:

```

pmod2mp :: Pmod state formula -> (Model state formula, [state])
pmod2mp (Pmod states prec accs points) = (Mo states prec accs, points)

```

Decomposing a pointed model into a list of pairs (m,w):

```

decompose :: Pmod state formula -> [(Model state formula, state)]
decompose (Pmod states prec accs points) =
    [(Mo states prec accs, point) | point <- points ]

```

It is useful to be able to map the precondition table to a function. Here is general tool for that. Note that the resulting function is partial; if the function argument does not occur in the table, the value is undefined.

```

table2fct :: Eq a => [(a,b)] -> a -> b
table2fct t = \ x -> maybe undefined id (lookup x t)

```

Another useful utility is a function that creates a partition out of an equivalence relation:

```

rel2part :: (Eq a) => [a] -> (a -> a -> Bool) -> [[a]]
rel2part [] r = []
rel2part (x:xs) r = xblock : rel2part rest r
  where
    (xblock,rest) = partition (\ y -> r x y) (x:xs)

```

The *domain* of a model is its list of states:

```

domain :: Model state formula -> [state]
domain (Mo states _ _) = states

```

The *eval* of a model is its list of state/formula pairs:

```

eval :: Model state formula -> [(state,formula)]
eval (Mo _ pre _) = pre

```

The *access* of a model is its labelled transition component:

```

access :: Model state formula -> [(Agent,state,state)]
access (Mo _ _ rel) = rel

```

The points of a Pmod:

```

points :: Pmod state formula -> [state]
points (Pmod _ _ _ pnts) = pnts

```

When we are looking at pointed models, we are only interested in generated submodels, with as their domain the designated state(s) plus everything that is reachable by an accessibility path.

```

gsm :: Ord state => Pmod state formula -> Pmod state formula
gsm (Pmod states pre rel points) = (Pmod states' pre' rel' points)
  where
    states' = closure rel all_agents points
    pre'    = [(s,f)      | (s,f)      <- pre,
                        elem s states'
                        ]
    rel'    = [(ag,s,s') | (ag,s,s') <- rel,
                        elem s states',
                        elem s' states'
                        ]

```

The closure of a state list, given a relation and a list of agents:

```

closure :: Ord state =>
  [(Agent,state,state)] -> [Agent] -> [state] -> [state]
closure rel agents xs
  | xs' == xs = xs
  | otherwise = closure rel agents xs'
  where
    xs' = (nub . sort) (xs ++ (expand rel agents xs))

```

The expansion of a relation R given a state set S and a set of agents B is given by $\{t \mid s \xrightarrow{b} t \in R, s \in S, b \in B\}$. Implementation:

```

expand :: Ord state =>
  [(Agent,state,state)] -> [Agent] -> [state] -> [state]
expand rel agnts ys =
  (nub . sort . concat)
  [ alternatives rel ag state | ag      <- agnts,
                              state <- ys
                              ]

```

The epistemic alternatives for agent a in state s are the states in sR_a (the states reachable through R_a from s):

```

alternatives :: Eq state =>
  [(Agent,state,state)] -> Agent -> state -> [state]
alternatives rel ag current =
  [ s' | (a,s,s') <- rel, a == ag, s == current ]

```

Chapter 4

Display and Visualisation

4.1 Module Declaration

```
module Display
where

import Char
import List
import Models
```

4.2 Representing Accessibility Relations

Formal background for this section is in Appendix A.

Filter out the accessibility relation for a particular agent label: `idxaccFor`

```
accFor :: Eq a => a -> [(a,b,b)] -> [(b,b)]
accFor label triples = [ (x,y) | (label',x,y) <- triples, label == label' ]
```

An implementation of \subseteq for lists:

```

containedIn :: Eq a => [a] -> [a] -> Bool
containedIn [] ys      = True
containedIn (x:xs) ys = elem x ys && containedIn xs ys

```

The smallest reflexive relation on a list:

```

idR :: Eq a => [a] -> [(a,a)]
idR = map (\x -> (x,x))

```

Test for reflexivity of a relation.

```

reflR :: Eq a => [a] -> [(a,a)] -> Bool
reflR xs r = containedIn (idR xs) r

```

Test for symmetry of a relation:

```

symR :: Eq a => [(a,a)] -> Bool
symR [] = True
symR ((x,y):pairs) | x == y    = symR (pairs)
                    | otherwise = elem (y,x) pairs
                    && symR (pairs \\ [(y,x)])

```

A check for transitivity of R tests for each couple of pairs $(x,y) \in R, (u,v) \in R$ whether $(x,v) \in R$ if $y = u$:

```

transR :: Eq a => [(a,a)] -> Bool
transR [] = True
transR s = and [ trans pair s | pair <- s ]
  where
    trans (x,y) r = and [ elem (x,v) r | (u,v) <- r, u == y ]

```

Put these together in a test for equivalence:

```

equivalenceR :: Eq a => [a] -> [(a,a)] -> Bool
equivalenceR xs r = reflR xs r && symR r && transR r

```

Checking whether a model is S5:

```

isS5 :: (Eq a) => [a] -> [(Agent,a,a)] -> Bool
isS5 xs triples =
  all (equivalenceR xs) rels
  where rels = [ accFor i triples | i <- all_agents ]

```

From a relation as a list of pairs to a characteristic function:

```

pairs2rel :: (Eq a, Eq b) => [(a,b)] -> a -> b -> Bool
pairs2rel pairs = \ x y -> elem (x,y) pairs

```

From an equivalence relation (represented as a list of pairs) to the corresponding partition:

```

equiv2part :: Eq a => [a] -> [(a,a)] -> [[a]]
equiv2part xs r = rel2part xs (pairs2rel r)

```

One way to test whether a model is KD45 is by means of tests for euclideaness, transitivity and seriality. The test for euclideaness is a variation on the test for transitivity.

```

euclideanR :: Eq a => [(a,a)] -> Bool
euclideanR s = and [ eucl pair s | pair <- s ]
  where
    eucl (x,y) r = and [ elem (y,v) r | (u,v) <- r, u == x ]

```

The test for seriality:

```

serialR :: Eq a => [a] -> [(a,a)] -> Bool
serialR [] s = True
serialR (x:xs) s = any (\ p -> (fst p) == x) s && serialR xs s

```

The test for KD45:

```

kd45R :: Eq a => [a] -> [(a,a)] -> Bool
kd45R xs r = transR r && serialR xs r && euclideanR r

```

A test for K45:

```

k45R :: Eq a => [(a,a)] -> Bool
k45R r = transR r && euclideanR r

```

Test for being an isolated point:

```

isolated :: Eq a => [(a,a)] -> a -> Bool
isolated r x = notElem x (map fst r ++ map snd r)

```

Checking for K45; if successful, return a list of isolated points and a list of balloons (see below).

```

k45PointsBalloons :: Eq a => [a] -> [(a,a)] -> Maybe ([a],[[a],[a]])
k45PointsBalloons xs r =
  let
    orphans = filter (isolated r) xs
    ys = xs \\ orphans
  in
    case kd45Balloons ys r of
      Just balloons -> Just (orphans,balloons)
      Nothing        -> Nothing

```

The entry pairs of a relation:

```

entryPair :: Eq a => [(a,a)] -> (a,a) -> Bool
entryPair r = \ (x,y) -> notElem (y,x) r

```

Checking for KD45 by testing the non-entry pairs for equivalence. If successful, the function returns just a list of balloons, where a balloon is a list pair, with the first element the entry points into the states in the second element.

```

kd45Balloons :: Eq a => [a] -> [(a,a)] -> Maybe [[a],[a]]
kd45Balloons xs r =
  let
    (s,t)          = partition (entryPair r) r
    entryPoints     = map fst s
    nonentryPoints = xs \\ entryPoints
    s5part xs r = if equivalenceR xs r
                  then Just (equiv2part xs t)
                  else Nothing
  in
    case s5part nonentryPoints t of
      Just part ->
        Just [ (nub (map fst (filter (\ (x,y) -> elem y block) s)),
              block) |      block <- part
        ]
      Nothing ->
        Nothing

```

This gives, e.g.:

```

DEMO> kd45Balloons [0,1,2] [(1,1),(1,2),(2,1),(2,2),(0,1)]
Just [[0],[1,2]]
DEMO> kd45Balloons [0,1,2,3] [(1,1),(1,2),(2,1),(2,2),(0,1),(3,1)]
Just [[0,3],[1,2]]
DEMO> kd45Balloons [0,1,2,3] [(1,1),(1,2),(2,1),(2,2),(0,1),(3,1),(1,3)]
Nothing

```

If the accessibility relations in a list of triples are all K45, just return the corresponding isolated points plus balloons for each agent. Otherwise return nothing.


```

kd45 :: (Eq a, Ord a) => [a] ->
  [(Agent,a,a)] -> Maybe [(Agent,([a],[[a],[a]]))]
kd45 xs triples =
  if and [ maybe False (\ x -> True) b | (a,b) <- results ]
  then Just [ (a, maybe undefined id b) | (a,b) <- results ]
  else Nothing
    where rels      = [ (a, accFor a triples) | a      <- all_agents ]
          results   = [ (a, kd45PointsBalloons xs r) | (a,r) <- rels   ]

```

If the accessibility relations in a list of triples are all KD45, just return the corresponding balloons for each agent. Otherwise return nothing.

```

kd45 :: (Eq a, Ord a) => [a] -> [(Agent,a,a)] -> Maybe [(Agent,([a],[a]))]
kd45 xs triples =
  if and [ maybe False (\ x -> True) b | (a,b) <- balloons ]
  then Just [ (a, maybe undefined id b) | (a,b) <- balloons ]
  else Nothing
    where rels      = [ (a, accFor a triples) | a      <- all_agents ]
          balloons  = [ (a, kd45Balloons xs r) | (a,r) <- rels      ]

```

Given a list of pairs consisting of isolated points and balloons, here is a cheap check to see whether it corresponds to a balloon list:

```

kd45psbs2balloons :: (Eq a, Ord a) =>
  [(Agent,([a],[[a],[a]]))] -> Maybe [(Agent,([a],[a]))]
kd45psbs2balloons psbs =
  if all (\ x -> x == []) entryList
  then Just balloons
  else Nothing
  where
    entryList = [ fst bs      | (a,bs) <- psbs ]
    balloons  = [ (a, snd bs) | (a,bs) <- psbs ]

```

Given a list of balloons, here is a cheap check to see whether it corresponds to a partition:

```

s5ball2part :: (Eq a, Ord a) =>
  [(Agent,[[[a],[a]]])] -> Maybe [(Agent,[[a]])]
s5ball2part balloons =
  if all (\ x -> x == []) entryList
    then Just partitions
    else Nothing
where
  entryList  = [ concat (map fst bs) | (a,bs) <- balloons ]
  partitions = [ (a, map snd bs)      | (a,bs) <- balloons ]

```

4.3 Model Display

Since models can become quite large, we need a way of displaying them in a convenient fashion. The tool for this is the following display function, useful for formatting large lists of showable things:

```

display :: Show a => Int -> [a] -> IO()
display n = if n < 1 then error "parameter not positive"
            else display' n n

where
  display' :: Show a => Int -> Int -> [a] -> IO()
  display' n m [] = putChar '\n'
  display' n 1 (x:xs) = do (putStr . show) x
                           putChar '\n'
                           display' n n xs
  display' n m (x:xs) = do (putStr . show) x
                           display' n (m-1) xs

```

Use this for displaying models, where the accessibility relations of S5 models are displayed as partitions and the accessibility relations of KD45 models as balloons (see Section A).

```

showMo :: (Eq state, Show state, Ord state, Show formula) =>
  Model state formula -> IO()
showMo = displayM 10

```

Showing Pmods:

```

showM :: (Eq state, Show state, Ord state, Show formula) =>
        Pmod state formula -> IO()
showM (Pmod sts pre acc pnts) = do putStr "==> "
        print pnts
        showMo (Mo sts pre acc)

```

Showing lists of Pmods:

```

showMs :: (Eq state, Show state, Ord state, Show formula) =>
        [Pmod state formula] -> IO()
showMs ms = sequence_ (map showM ms)

```

Displaying models:

```

displayM :: (Eq state, Show state, Ord state, Show formula) =>
        Int -> Model state formula -> IO()
displayM n (Mo states pre rel) =
    do print states
    display (div n 2) pre
    case (k45 states rel) of
        Nothing      -> display n rel
        Just psbs    -> case kd45psbs2balloons psbs of
            Nothing   -> displayPB (div n 2) psbs
            Just balloons -> case s5ball2part balloons of
                Nothing      -> displayB (div n 2) balloons
                Just parts   -> displayP (2*n) parts

```

Displaying lists of partitions:

```

displayP :: Show a => Int -> [(Agent,[[a]])] -> IO()
displayP n parts = sequence_ (map (display n) (map (\x -> [x]) parts))

```

Displaying lists of KD45 balloons:

```
displayB :: Show a => Int -> [(Agent,[[a],[a]])] -> IO()
displayB n balloons = sequence_ (map (display n) (map (\x -> [x]) balloons))
```

Displaying a list of K45 isolated points plus balloons:

```
displayPB :: Show a => Int -> [(Agent,([a],[[a],[a]]))] -> IO()
displayPB n psbs = sequence_ (map (display n) (map (\x -> [x]) psbs))
```

4.4 Model Visualisation

For graphical visualisation of static models and action models, we use the graphic visualisation tool *dot* [32]. Define a class **GraphViz**, as a subclass of **Show**.

```
class Show a => GraphViz a where
  graphviz :: a -> String
```

Every instance of the **GraphViz** class should have the function **graphviz** defined on it.

The following functions uses a string as glue between a list of strings, to produce a single long string.

```
glueWith :: String -> [String] -> String
glueWith _ []      = []
glueWith _ [y]     = y
glueWith s (y:ys) = y ++ s ++ glueWith s ys
```

Listing states:

```

listState :: (Show a, Show b, Eq a, Eq b) => a -> [(a,b)] -> String
listState w val =
  let
    props = head (maybe [] (\ x -> [x]) (lookup w val))
    label = filter (isAlphaNum) (show props)
  in
    if null label
    then show w
    else show w
      ++ "[label =\"" ++ (show w) ++ ":" ++ (show props) ++ "\"]"

```

Get the links from a labelled relation:

```

links :: (Eq a, Eq b) => [(a,b,b)] -> [(a,b,b)]
links [] = []
links ((x,y,z):xyzs) | y == z    = links xyzs
                      | otherwise =
                        (x,y,z): links (filter (/= (x,z,y)) xyzs)

```

Compress the links from a labelled relation:

```

cmpl :: Eq b => [(Agent,b,b)] -> [(Agent,b,b)]
cmpl [] = []
cmpl ((x,y,z):xyzs) = (xs,y,z):(cmpl xyzs')
  where xs = x: [ a | a <- all_agents, elem a (map f xyzs1) ]
        xyzs1 = filter (\ (u,v,w) ->
                          (v == y && w == z)
                          ||
                          (v == z && w == y)) xyzs
        f (x,_,_) = x
        xyzs' = xyzs \\ xyzs1

```

Put models in the class `GraphViz`, and define the `graphviz` function for them:

```

instance (Show a, Show b, Eq a, Eq b) => GraphViz (Model a b) where
  graphviz (Mo states val rel) = if isS5 states rel
  then
    "digraph G { "
    ++
    glueWith " ; " [ listState s val | s <- states ]
    ++ " ; " ++
    glueWith " ; " [ (show s) ++ " -> " ++ (show s')
                      ++ " [label="
                      ++ (filter isAlpha (show ags))
                      ++ ",dir=none ]" |
                      s <- states, s' <- states,
                      (ags,t,t') <- (cml . links) rel,
                      s == t, s' == t' ]
    ++ " }"
  else
    "digraph G { "
    ++
    glueWith " ; " [ listState s val | s <- states ]
    ++ " ; " ++
    glueWith " ; " [ (show s) ++ " -> " ++ (show s')
                      ++ " [label=" ++ (show ag) ++ "]" |
                      s <- states, s' <- states,
                      (ag,t,t') <- rel,
                      s == t, s' == t' ]
    ++ " }"

```

Listing pointed states:

```

listPState  :: (Show a, Show b, Eq a, Eq b) =>
              a -> [(a,b)] -> Bool -> String
listPState w val pointed =
  let
    props = head (maybe [] (\ x -> [x]) (lookup w val))
    label = filter (isAlphaNum) (show props)
  in
    if null label
    then if pointed then show w ++ "[peripheries = 2]"
         else          show w
    else if pointed then
      show w
      ++ "[label =\"" ++ (show w) ++ ":" ++ (show props) ++
         "\",peripheries = 2]"
    else show w
      ++ "[label =\"" ++ (show w) ++ ":" ++ (show props) ++ "\"]"

```

Listing pointed models:

```

instance (Show a, Show b, Eq a, Eq b) => GraphViz (Pmod a b) where
  graphviz (Pmod states val rel points) = if isS5 states rel
  then
    "digraph G { "
    ++
    glueWith " ; " [ listPState s val (elem s points) | s <- states ]
    ++ " ; " ++
    glueWith " ; " [ (show s) ++ " -> " ++ (show s')
                      ++ " [label="
                      ++ (filter isAlpha (show ags))
                      ++ ",dir=none ]" |
                      s <- states, s' <- states,
                      (ags,t,t') <- (cml . links) rel,
                      s == t, s' == t'
    ++ " }"
  else
    "digraph G { "
    ++
    glueWith " ; " [ listPState s val (elem s points) | s <- states ]
    ++ " ; " ++
    glueWith " ; " [ (show s) ++ " -> " ++ (show s')
                      ++ " [label=" ++ (show ag) ++ "]" |
                      s <- states, s' <- states,
                      (ag,t,t') <- rel,
                      s == t, s' == t'
    ++ " }"

```

Write graph to file:

```

writeGraph :: String -> IO()
writeGraph cts = writeFile "graph.dot" cts

```

```

writeGr :: String -> String -> IO()
writeGr name cts = writeFile name cts

```

Write model to file:


```
writeModel :: (Show a, Show b, Eq a, Eq b) => Model a b -> IO()
writeModel m = writeGraph (graphviz m)
```

Write Pmod to file:

```
writePmod :: (Show a, Show b, Eq a, Eq b) => (Pmod a b) -> IO()
writePmod m = writeGraph (graphviz m)
```

```
writeP :: (Show a, Show b, Eq a, Eq b) => String -> (Pmod a b) -> IO()
writeP name m = writeGr (name ++ ".dot") (graphviz m)
```

To Do 1 *Improve the display of S5, KD45 and K45 models along the lines of the terminal display code for models from the previous section.*

To Do 2 *Add visualisations of automata.*

Chapter 5

Model Minimization under Bisimulation

5.1 Module Declaration

```
module MinBis where

import List
import Models
```

5.2 Partition Refinement

Any Kripke model can be simplified by replacing each state s by its bisimulation class $[s]$. The problem of finding the smallest Kripke model modulo bisimulation is similar to the problem of minimizing the number of states in a finite automaton [27]. We will use partition refinement, in the spirit of [34]. Here is the algorithm:

- Start out with a partition of the state set where all states with the same precondition function are in the same class. The equality relation to be used to evaluate the precondition function is given as a parameter to the algorithm.
- Given a partition Π , for each block b in Π , partition b into sub-blocks such that two states s, t of b are in the same sub-block iff for all agents a it holds that s and t have \xrightarrow{a} transitions to states in the same block of Π . Update Π to Π' by replacing each b in Π by the newly found set of sub-blocks for b .
- Halt as soon as $\Pi = \Pi'$.

Looking up and checking of two formulas against a given equivalence relation:

```
lookupFs :: (Eq a, Eq b) => a -> a -> [(a,b)] -> (b -> b -> Bool) -> Bool
lookupFs i j table r = case lookup i table of
  Nothing -> lookup j table == Nothing
  Just f1 -> case lookup j table of
    Nothing -> False
    Just f2 -> r f1 f2
```

Computing the initial partition, using a particular relation for equivalence of formulas:

```
initPartition :: (Eq a, Eq b) => Model a b -> (b -> b -> Bool) -> [[a]]
initPartition (Mo states pre rel) r =
  rel2part states (\ x y -> lookupFs x y pre r)
```

Refining a partition:

```
refinePartition :: (Eq a, Eq b) => Model a b -> [[a]] -> [[a]]
refinePartition m p = refineP m p p
  where
    refineP :: (Eq a, Eq b) => Model a b -> [[a]] -> [[a]] -> [[a]]
    refineP m part [] = []
    refineP m@(Mo states pre rel) part (block:blocks) =
      newblocks ++ (refineP m part blocks)
      where
        newblocks =
          rel2part block (\ x y -> sameAccBlocks m part x y)
```

Function that checks whether two states have the same accessible blocks under a partition:

```
sameAccBlocks :: (Eq a, Eq b) =>
  Model a b -> [[a]] -> a -> a -> Bool
sameAccBlocks m@(Mo states pre rel) part s t =
  and [ accBlocks m part s ag == accBlocks m part t ag |
        ag <- all_agents ]
```

The accessible blocks for an agent from a given state, given a model and a partition:

```
accBlocks :: (Eq a, Eq b) => Model a b -> [[a]] -> a -> Agent -> [[a]]
accBlocks m@(Mo states pre rel) part s ag =
  nub [ bl part y | (ag',x,y) <- rel, ag' == ag, x == s ]
```

The block of an object in a partition:

```
bl :: (Eq a) => [[a]] -> a -> [a]
bl part x = head (filter (\ b -> elem x b) part)
```

Initializing and refining a partition:

```
initRefine :: (Eq a, Eq b) => Model a b -> (b -> b -> Bool) -> [[a]]
initRefine m r = refine m (initPartition m r)
```

The refining process:

```
refine :: (Eq a, Eq b) => Model a b -> [[a]] -> [[a]]
refine m part = if rpart == part
  then part
  else refine m rpart
  where rpart = refinePartition m part
```

5.3 Minimization

Use this to construct the minimal model. Notice the dependence on relational parameter \mathbf{r} .

```

minimalModel :: (Eq a, Ord a, Eq b, Ord b) =>
    (b -> b -> Bool) -> Model a b -> Model [a] b
minimalModel r m@(Mo states pre rel) =
    (Mo states' pre' rel')
    where
        partition = initRefine m r
        states'   = partition
        f         = bl partition
        rel'      = (nub.sort) (map (\ (x,y,z) -> (x, f y, f z)) rel)
        pre'      = (nub.sort) (map (\ (x,y)   -> (f x, y))      pre)

```

Bisimulation-minimal Pmods:

```

minimalPmod :: (Eq a, Ord a, Eq b, Ord b) =>
    (b -> b -> Bool) -> Pmod a b -> Pmod [a] b
minimalPmod r (Pmod sts pre rel pts) = (Pmod sts' pre' rel' pts')
    where (Mo sts' pre' rel') = minimalModel r (Mo sts pre rel)
          pts' = map (bl sts') pts

```

Converting a's into integers, using their position in a given list of a's.

```

convert :: (Eq a, Show a) => [a] -> a -> Integer
convert = convrt 0
    where
        convrt :: (Eq a, Show a) => Integer -> [a] -> a -> Integer
        convrt n []      x = error (show x ++ " not in list")
        convrt n (y:ys) x | x == y      = n
                          | otherwise = convrt (n+1) ys x

```

Converting an object of type Model a b into an object of type Model Integer b:

```

conv :: (Eq a, Show a) => Model a b -> Model Integer b
conv (Mo worlds val acc) =
    (Mo (map f worlds)
        (map (\ (x,y)   -> (f x, y)) val)
        (map (\ (x,y,z) -> (x, f y, f z)) acc))
    where f = convert worlds

```

Conversion by renaming of Pmods:

```
convPmod :: (Eq a, Show a) => Pmod a b -> Pmod Integer b
convPmod (Pmod sts pre rel pts) = (Pmod sts' pre' rel' pts')
  where (Mo sts' pre' rel') = conv (Mo sts pre rel)
        pts' = nub (map (convert sts) pts)
```

Use this to rename the blocks into integers:

```
bisim :: (Eq a, Ord a, Show a, Eq b, Ord b) =>
        (b -> b -> Bool) -> Model a b -> Model Integer b
bisim r = conv . (minimalModel r)
```

Reducing Pmods under bisimulation:

```
bisimPmod :: (Eq a, Ord a, Show a, Eq b, Ord b) =>
            (b -> b -> Bool) -> Pmod a b -> Pmod Integer b
bisimPmod r = convPmod . (minimalPmod r)
```

Chapter 6

Formulas, Action Models and Epistemic Models

6.1 Module Declaration

```
module ActEpist
where

import List
import Models
import MinBis
import DPLL
```

Module `List` is a standard Haskell module. Module `Models` is described in Chapter 3, Module `MinBis` in Chapter 5, and module `DPLL` in Appendix B.

6.2 Formulas

Basic propositions:

```
data Prop = P Int | Q Int | R Int deriving (Eq,Ord)
```

Show these in the standard way, in lower case, with index 0 omitted.

```

instance Show Prop where
  show (P 0) = "p"; show (P i) = "p" ++ show i
  show (Q 0) = "q"; show (Q i) = "q" ++ show i
  show (R 0) = "r"; show (R i) = "r" ++ show i

```

Formulas, according to the definition:

$$\begin{aligned}
\varphi &::= \top \mid p \mid \neg\varphi \mid \bigwedge[\varphi_1, \dots, \varphi_n] \mid \bigvee[\varphi_1, \dots, \varphi_n] \mid [\pi]\varphi \mid [\mathbf{A}]\varphi \mid [\text{Aut}]\varphi \\
\pi &::= a \mid B \mid ?\varphi \mid \bigcirc[\pi_1, \dots, \pi_n] \mid \bigcup[\pi_1, \dots, \pi_n] \mid \pi^*
\end{aligned}$$

Here, p ranges over basic propositions, a ranges over agents, B ranges over non-empty sets of agents, \mathbf{A} is a multiple pointed action model (see below), and Aut is an automaton. \bigcirc denotes sequential composition of a list of programs. We will often write $\bigcirc[\pi_1, \pi_2]$ as $\pi_1; \pi_2$, and $\bigcup[\pi_1, \pi_2]$ as $\pi_1 \cup \pi_2$.

Note that general knowledge among agents B that φ is expressed in this language as $[B]\varphi$, and common knowledge among agents B that φ as $[B^*]\varphi$. Thus, $[B]\varphi$ can be viewed as shorthand for $[\bigcup_{b \in B} b]\varphi$. In case $B = \emptyset$, $[B]\varphi$ turns out to be equivalent to $[?\perp]\varphi$.

For convenience, we have also left in the more traditional way of expressing individual knowledge $\Box_a\varphi$, general knowledge $E_B\varphi$ and common knowledge $C_B\varphi$.

```

data Form = Top
  | Prop Prop
  | Neg Form
  | Conj [Form]
  | Disj [Form]
  | Pr Program Form
  | K Agent Form
  | EK [Agent] Form
  | CK [Agent] Form
  | Up PoAM Form
  | Aut (NFA State) Form
  deriving (Eq,Ord)

```



```

data Program = Ag Agent
              | Ags [Agent]
              | Test Form
              | Conc [Program]
              | Sum [Program]
              | Star Program
              deriving (Eq,Ord)

```

Some useful abbreviations:

```

impl :: Form -> Form -> Form
impl form1 form2 = Disj [Neg form1, form2]

equiv :: Form -> Form -> Form
equiv form1 form2 = Conj [form1 'impl' form2, form2 'impl' form1]

```

The negation of a formula:

```

negation :: Form -> Form
negation (Neg form) = form
negation form      = Neg form

```

Show formulas in the standard way:

```

instance Show Form where
  show Top = "T" ; show (Prop p) = show p; show (Neg f) = '-':(show f);
  show (Conj fs)      = '&': show fs
  show (Disj fs)      = 'v': show fs
  show (Pr p f)       = '[': show p ++ "]" ++ show f
  show (K agent f)    = '[': show agent ++ "]" ++ show f
  show (EK agents f)  = 'E': show agents ++ show f
  show (CK agents f)  = 'C': show agents ++ show f
  show (Up pam f)     = 'A': show (points pam) ++ show f
  show (Aut aut f)    = '[': show aut ++ "]" ++ show f

```

Show programs in a standard way:

```

instance Show Program where
  show (Ag a)      = show a
  show (Ags as)    = show as
  show (Test f)    = '?': show f
  show (Conc ps)   = 'C': show ps
  show (Sum ps)    = 'U': show ps
  show (Star p)    = '(' : show p ++ ")*"

```

Programs can get very unwieldy very quickly. As is well known, there is no normalisation procedure for regular expressions. Still, here are some rewriting steps for simplification of programs:

$$\begin{aligned}
& \emptyset \rightarrow ?\perp \\
& ?\varphi_1 \cup ?\varphi_2 \rightarrow ?(\varphi_1 \vee \varphi_2) \\
& ?\perp \cup \pi \rightarrow \pi \\
& \pi \cup ?\perp \rightarrow \pi \\
& \bigcup[\pi_1, \dots, \pi_k, \bigcup[\pi_{k+1}, \dots, \pi_{k+m}], \pi_{k+m+1}, \dots, \pi_{k+m+n}] \rightarrow \bigcup[\pi_1, \dots, \pi_{k+m+n}] \\
& \bigcup[] \rightarrow ?\perp \\
& \bigcup[\pi] \rightarrow \pi \\
& ?\varphi_1; ?\varphi_2 \rightarrow ?(\varphi_1 \wedge \varphi_2) \\
& ?\top; \pi \rightarrow \pi \\
& \pi; ?\top \rightarrow \pi \\
& ?\perp; \pi \rightarrow ?\perp \\
& \pi; ?\perp \rightarrow ?\perp \\
& \bigcirc[\pi_1, \dots, \pi_k, \bigcirc[\pi_{k+1}, \dots, \pi_{k+m}], \pi_{k+m+1}, \dots, \pi_{k+m+n}] \rightarrow \bigcirc[\pi_1, \dots, \pi_{k+m+n}] \\
& \bigcirc[] \rightarrow ?\top \\
& \bigcirc[\pi] \rightarrow \pi \\
& (? \varphi)^* \rightarrow ?\top \\
& (? \varphi \cup \pi)^* \rightarrow \pi^* \\
& (\pi \cup ? \varphi)^* \rightarrow \pi^* \\
& \pi^{**} \rightarrow \pi^*
\end{aligned}$$

Simplifying unions by splitting up in test part, accessibility part and rest:

```

splitU :: [Program] -> ([Form],[Agent],[Program])
splitU [] = ([],[],[ ])
splitU (Test f: ps) = (f:fs,ags,prs)
    where (fs,ags,prs) = splitU ps
splitU (Ag x: ps) = (fs,union [x] ags,prs)
    where (fs,ags,prs) = splitU ps
splitU (Ags xs: ps) = (fs,union xs ags,prs)
    where (fs,ags,prs) = splitU ps
splitU (Sum ps: ps') = splitU (union ps ps')
splitU (p:ps)         = (fs,ags,p:prs)
    where (fs,ags,prs) = splitU ps

```

Simplifying compositions:

```

comprC :: [Program] -> [Program]
comprC [] = []
comprC (Test Top: ps)      = comprC ps
comprC (Test (Neg Top): ps) = [Test (Neg Top)]
comprC (Test f: Test f': rest) = comprC (Test (canonF (Conj [f,f'])): rest)
comprC (Conc ps : ps')     = comprC (ps ++ ps')
comprC (p:ps)              = let ps' = comprC ps
                             in
                             if ps' == [Test (Neg Top)]
                             then [Test (Neg Top)]
                             else p: ps'

```

Use this in the code for program simplification:

```

simpl :: Program -> Program
simpl (Ag x)          = Ag x
simpl (Ags [])        = Test (Neg Top)
simpl (Ags [x])       = Ag x
simpl (Ags xs)        = Ags xs
simpl (Test f)        = Test (canonF f)

```

Simplifying unions:

```

simpl (Sum prs) =
  let (fs,xs,rest) = splitU (map simpl prs)
      f             = canonF (Disj fs)
  in
    if xs == [] && rest == []
    then Test f
    else if xs == [] && f == Neg Top && length rest == 1
    then (head rest)
    else if xs == [] && f == Neg Top
    then Sum rest
    else if xs == []
    then Sum (Test f: rest)
    else if length xs == 1 && f == Neg Top
    then Sum (Ag (head xs): rest)
    else if length xs == 1
    then Sum (Test f: Ag (head xs): rest)
    else if f == Neg Top
    then Sum (Ags xs: rest)
    else Sum (Test f: Ags xs: rest)

```

Simplifying sequential compositions:

```

simpl (Conc prs) =
  let prs' = comprC (map simpl prs)
  in
    if prs' == [] then Test Top
    else if length prs' == 1 then head prs'
    else if head prs' == Test Top then Conc (tail prs')
    else Conc prs'

```

Simplifying stars:

```

simpl (Star pr) = case simpl pr of
  Test f           -> Test Top
  Sum [Test f, pr'] -> Star pr'
  Sum (Test f: prs') -> Star (Sum prs')
  Star pr'         -> Star pr'
  pr'              -> Star pr'

```

Property of being a purely propositional formula:

```
pureProp :: Form -> Bool
pureProp Top      = True
pureProp (Prop _) = True
pureProp (Neg f)   = pureProp f
pureProp (Conj fs) = and (map pureProp fs)
pureProp (Disj fs) = and (map pureProp fs)
pureProp _        = False
```

Some example formulas and formula-forming operators:

```
bot, p0, p, p1, p2, p3, p4, p5, p6 :: Form
bot = Neg Top
p0 = Prop (P 0); p = p0; p1 = Prop (P 1); p2 = Prop (P 2)
p3 = Prop (P 3); p4 = Prop (P 4); p5 = Prop (P 5); p6 = Prop (P 6)

q0, q, q1, q2, q3, q4, q5, q6 :: Form
q0 = Prop (Q 0); q = q0; q1 = Prop (Q 1); q2 = Prop (Q 2);
q3 = Prop (Q 3); q4 = Prop (Q 4); q5 = Prop (Q 5); q6 = Prop (Q 6)

r0, r, r1, r2, r3, r4, r5, r6 :: Form
r0 = Prop (R 0); r = r0; r1 = Prop (R 1); r2 = Prop (R 2)
r3 = Prop (R 3); r4 = Prop (R 4); r5 = Prop (R 5); r6 = Prop (R 6)

u = Up :: PoAM -> Form -> Form

nkap = Neg (K a p)
nkanp = Neg (K a (Neg p))
nka_p = Conj [nkap, nkanp]
```

6.3 Reducing Formulas to Canonical Form

For computing bisimulations, it is useful to have some notion of equivalence (however crude) for the logical language. For this, we reduce formulas to a canonical form. We will derive canonical forms that are unique up to propositional equivalence, employing a propositional reasoning engine. This is still rather crude, for any modal formula will be treated as a propositional literal.

The DPLL (Davis, Putnam, Logemann, Loveland) engine in Appendix B expects clauses represented as lists of integers, so we first have to translate to this format. This translation should start with computing a mapping from positive literals to integers.

For the non-propositional operators we use a little bootstrapping, by putting the formula inside the operator in canonical form, using the function `canonF` to be defined below. Also, since the non-propositional operators all behave as Box modalities, we can reduce $\Box \top$ to \top .

```
mapping :: Form -> [(Form,Integer)]
mapping f = zip lits [1..k]
  where
    lits = (sort . nub . collect) f
    k     = toInteger (length lits)
    collect :: Form -> [Form]
    collect Top          = []
    collect (Prop p)     = [Prop p]
    collect (Neg f)      = collect f
    collect (Conj fs)    = concat (map collect fs)
    collect (Disj fs)    = concat (map collect fs)
    collect (Pr pr f)    = if canonF f == Top then [] else [Pr pr (canonF f)]
    collect (K ag f)     = if canonF f == Top then [] else [K ag (canonF f)]
    collect (EK ags f)   = if canonF f == Top then [] else [EK ags (canonF f)]
    collect (CK ags f)   = if canonF f == Top then [] else [CK ags (canonF f)]
    collect (Up pam f)   = if canonF f == Top then [] else [Up pam (canonF f)]
    collect (Aut nfa f)  = if nfa == nullAut || canonF f == Top
                          then [] else [Aut nfa (canonF f)]
```

Putting in clausal form, given a mapping for the literals, and using bootstrapping for formulas in the scope of a non-propositional operator. Note that $\Box \top$ is reduced to \top , $\neg \Box \top$ to \perp , and $[\text{Aut}] \varphi$ to \top (and $\neg [\text{Aut}] \varphi$ to \perp) if `Aut` is the automaton for the empty language.

```

cf :: (Form -> Integer) -> Form -> [[Integer]]
cf g (Top) = []
cf g (Prop p) = [[g (Prop p)]]
cf g (Pr pr f) = if canonF f == Top then []
                  else [[g (Pr pr (canonF f))]]
cf g (K ag f) = if canonF f == Top then []
                  else [[g (K ag (canonF f))]]
cf g (EK ags f) = if canonF f == Top then []
                   else [[g (EK ags (canonF f))]]
cf g (CK ags f) = if canonF f == Top then []
                   else [[g (CK ags (canonF f))]]
cf g (Up am f) = if canonF f == Top then []
                  else [[g (Up am (canonF f))]]
cf g (Aut nfa f) = if nfa == nullAut || canonF f == Top then []
                   else [[g (Aut nfa (canonF f))]]
cf g (Conj fs) = concat (map (cf g) fs)
cf g (Disj fs) = deMorgan (map (cf g) fs)

```

Negated formulas:

```

cf g (Neg Top) = [[]]
cf g (Neg (Prop p)) = [[- g (Prop p)]]
cf g (Neg (Pr pr f)) = if canonF f == Top then [[]]
                      else [[- g (Pr pr (canonF f))]]
cf g (Neg (K ag f)) = if canonF f == Top then [[]]
                      else [[- g (K ag (canonF f))]]
cf g (Neg (EK ags f)) = if canonF f == Top then [[]]
                       else [[- g (EK ags (canonF f))]]
cf g (Neg (CK ags f)) = if canonF f == Top then [[]]
                       else [[- g (CK ags (canonF f))]]
cf g (Neg (Up am f)) = if canonF f == Top then [[]]
                      else [[- g (Up am (canonF f))]]
cf g (Neg (Aut nfa f)) = if nfa == nullAut || canonF f == Top then [[]]
                       else [[- g (Aut nfa (canonF f))]]
cf g (Neg (Conj fs)) = deMorgan (map (\ f -> cf g (Neg f)) fs)
cf g (Neg (Disj fs)) = concat (map (\ f -> cf g (Neg f)) fs)
cf g (Neg (Neg f)) = cf g f

```

De Morgan's disjunction distribution:

$$\varphi \vee (\psi_1 \wedge \dots \wedge \psi_n) \leftrightarrow (\varphi \vee \psi_1) \wedge \dots \wedge (\varphi \vee \psi_n).$$

De Morgan's disjunction distribution, for the case of a disjunction of a list of clause sets.

```
deMorgan :: [[[Integer]]] -> [[Integer]]
deMorgan [] = []
deMorgan [cls] = cls
deMorgan (cls:clss) = deMorg cls (deMorgan clss)
  where
    deMorg :: [[Integer]] -> [[Integer]] -> [[Integer]]
    deMorg cls1 cls2 = (nub . concat) [ deM cl cls2 | cl <- cls1 ]
    deM :: [Integer] -> [[Integer]] -> [[Integer]]
    deM cl cls = map (fuseLists cl) cls
```

Function `fuseLists` keeps the literals in the clauses ordered.

```
fuseLists :: [Integer] -> [Integer] -> [Integer]
fuseLists [] ys = ys
fuseLists xs [] = xs
fuseLists (x:xs) (y:ys) | abs x < abs y = x:(fuseLists xs (y:ys))
                        | abs x == abs y = if x == y
                                           then x:(fuseLists xs ys)
                                           else if x > y
                                           then x:y:(fuseLists xs ys)
                                           else y:x:(fuseLists xs ys)
                        | abs x > abs y = y:(fuseLists (x:xs) ys)
```

Given a mapping for the positive literals, the satisfying valuations of a formula can be collected from the output of the DPLL process. Here `dp` is the function imported from the module `DPLL`.

```
satVals :: (Form,Integer) -> Form -> [[Integer]]
satVals t f = (map fst . dp) (cf (table2fct t) f)
```

Two formulas are propositionally equivalent if they have the same sets of satisfying valuations, computed on the basis of a literal mapping for their conjunction:

```
propEquiv :: Form -> Form -> Bool
propEquiv f1 f2 = satVals g f1 == satVals g f2
  where g = mapping (Conj [f1,f2])
```


A formula is a (propositional) contradiction if it is propositionally equivalent to `Neg Top`, or equivalently, to `Disj []`:

```
contrad :: Form -> Bool
contrad f = propEquiv f (Disj [])
```

A formula is (propositionally) consistent if it is not a propositional contradiction:

```
consistent :: Form -> Bool
consistent = not . contrad
```

Use the set of satisfying valuations to derive a canonical form:

```
canonF :: Form -> Form
canonF f = if (contrad (Neg f))
  then Top
  else if fs == []
  then Neg Top
  else if length fs == 1
  then head fs
  else Disj fs
  where g = mapping f
        nss = satVals g f
        g' = \ i -> head [ form | (form,j) <- g, i == j ]
        h = \ i -> if i < 0 then Neg (g' (abs i)) else g' i
        h' = \ xs -> map h xs
        k = \ xs -> if xs == []
          then Top
          else if length xs == 1
            then head xs
            else Conj xs
        fs = map k (map h' nss)
```

This gives:

```
ActEpist> canonF p
p
ActEpist> canonF (Conj [p,Top])
```

```

P
ActEpist> canonF (Conj [p,q,Neg r])
&[p,q,-r]
ActEpist> canonF (Neg (Disj [p,(Neg p)]))
-T
ActEpist> canonF (Disj [p,q,Neg r])
v[p,&[-p,q],&[-p,-q,-r]]
ActEpist> canonF (K a (Disj [p,q,Neg r]))
[a]v[p,&[-p,q],&[-p,-q,-r]]
ActEpist> canonF (Conj [p, Conj [q,Neg r]])
&[p,q,-r]
ActEpist> canonF (Conj [p, Disj [q,Neg (K a (Disj []))]])
v[&[p,q],&[p,-q,-[a]-T]]
ActEpist> canonF (Conj [p, Disj [q,Neg (K a (Conj []))]])
&[p,q]

```

To Do 3 *Extend this with a further treatment of:*

- \Box_a modalities,
- E_B modalities,
- C_B modalities,
- $[M]$ update modalities.

Propositional decomposition in the scope of these operators is already being performed. Further decomposition is a first step. At a later stage we may consider hooking up to a proof engine for modal logic, or extending the present propositional prover with rules for the modalities.

6.4 Action Models and Epistemic Models

Action models and epistemic models are built from states. We assume states are represented by integers:

```
type State = Integer
```

Static models are models where the states are of type **State**, and the precondition function assigns lists of basic propositions (this specializes the precondition function to a valuation).

```
type SM = Model State [Prop]
```

Epistemic models are static models with a set of distinguished points:

```
type EpistM = Pmod State [Prop]
```

Find the valuation of an epistemic model:

```
valuation  :: EpistM -> [(State,[Prop])]
valuation pmod = eval $ fst (pmod2mp pmod)
```

Action models are models where the states are of type `State`, and the precondition function assigns objects of type `Form`. The only difference between an action model and a static model is in the fact that action models have a precondition function that assigns a formula instead of a set of basic propositions.

```
type AM = Model State Form
```

Pointed action models:

```
type PoAM = Pmod State Form
```

The preconditions of a pointed action model:

```
preconditions :: PoAM -> [Form]
preconditions (Pmod states pre acc points) =
  map (table2fct pre) points
```

Sometimes we need a single precondition:

```
precondition :: PoAM -> Form
precondition am = canonF (Conj (preconditions am))
```

The zero action model **0**:

```
zero :: PoAM
zero = Pmod [] [] [] []
```

The purpose of action models is to define relations on the class of all static models. States with precondition \perp can be pruned from an action model. For this we define a specialized version of the `gsm` function:

```
gsmPoAM :: PoAM -> PoAM
gsmPoAM (Pmod states pre acc points) =
  let
    points' = [ p | p <- points, consistent (table2fct pre p) ]
    states' = [ s | s <- states, consistent (table2fct pre s) ]
    pre'    = filter (\ (x,_) -> elem x states') pre
    f       = \ (_,s,t) -> elem s states' && elem t states'
    acc'    = filter f acc
  in
  if points' == []
  then zero
  else gsm (Pmod states' pre' acc' points')
```

6.5 Program Transformation

For every action model A with states s_0, \dots, s_{n-1} we define a set of n^2 program transformers $T_{i,j}^A$ ($0 \leq i < n, 0 \leq j < n$), as follows [16]:

:

$$\begin{aligned}
T_{ij}^A(a) &= \begin{cases} ?\text{pre}(s_i); a & \text{if } s_i \xrightarrow{a} s_j, \\ ?\perp & \text{otherwise} \end{cases} \\
T_{ij}^A(? \varphi) &= \begin{cases} ?(\text{pre}(s_i) \wedge [A, s_i] \varphi) & \text{if } i = j, \\ ?\perp & \text{otherwise} \end{cases} \\
T_{ij}^A(\pi_1; \pi_2) &= \bigcup_{k=0}^{n-1} (T_{ik}^A(\pi_1); T_{kj}^A(\pi_2)) \\
T_{ij}^A(\pi_1 \cup \pi_2) &= T_{ij}^A(\pi_1) \cup T_{ij}^A(\pi_2) \\
T_{ij}^A(\pi^*) &= K_{ijn}^A(\pi)
\end{aligned}$$

where $K_{ijk}^A(\pi)$ is a (transformed) program for all the π^* paths from s_i to s_j that can be traced through A while avoiding a pass through intermediate states s_k and higher. Thus, $K_{ijn}^A(\pi)$ is a program for all the π^* paths from s_i to s_j that can be traced through A , period.

$K_{ijk}^A(\pi)$ is defined by recursing on k , as follows:

$$K_{ij0}^A(\pi) = \begin{cases} ?\top \cup T_{ij}^A(\pi) & \text{if } i = j, \\ T_{ij}^A(\pi) & \text{otherwise} \end{cases}$$

$$K_{ij(k+1)}^A(\pi) = \begin{cases} (K_{kkk}^A(\pi))^* & \text{if } i = k = j, \\ (K_{kkk}^A(\pi))^*; K_{kjk}^A(\pi) & \text{if } i = k \neq j, \\ K_{ikk}^A(\pi); (K_{kkk}^A(\pi))^* & \text{if } i \neq k = j, \\ K_{ijk}^A(\pi) \cup (K_{ikk}^A(\pi); (K_{kkk}^A(\pi))^*; K_{kjk}^A(\pi)) & \text{otherwise } (i \neq k \neq j). \end{cases}$$

Lemma 3 (Kleene Path) *Suppose $(w, w') \in \llbracket T_{ij}^A(\pi) \rrbracket^{\mathbf{M}}$ iff there is a π path from (w, s_i) to (w', s_j) in $\mathbf{M} \otimes A$. Then $(w, w') \in \llbracket K_{ijn}^A(\pi) \rrbracket^{\mathbf{M}}$ iff there is a π^* path from (w, s_i) to (w', s_j) in $\mathbf{M} \otimes A$.*

The Kleene path lemma is the key ingredient in the proof of the following program transformation lemma.

Lemma 4 (Program Transformation) *Assume A has n states s_0, \dots, s_{n-1} . Then:*

$$\mathbf{M} \models_w [A, s_i][\pi]\varphi \text{ iff } \mathbf{M} \models_w \bigwedge_{j=0}^{n-1} [T_{ij}^A(\pi)][A, s_j]\varphi.$$

The implementation of the program transformation functions is given here:

```

transf :: PoAM -> Integer -> Integer -> Program -> Program
transf am@(Pmod states pre acc points) i j (Ag ag) =
  let
    f = table2fct pre i
  in
    if elem (ag,i,j) acc && f == Top          then Ag ag
    else if elem (ag,i,j) acc && f /= Neg Top then Conc [Test f, Ag ag]
    else Test (Neg Top)
transf am@(Pmod states pre acc points) i j (Ags ags) =
  let ags' = nub [ a | (a,k,m) <- acc, elem a ags, k == i, m == j ]
      ags1 = intersect ags ags'
      f     = table2fct pre i
  in
    if ags1 == [] || f == Neg Top          then Test (Neg Top)
    else if f == Top && length ags1 == 1 then Ag (head ags1)
    else if f == Top                      then Ags ags1
    else Conc [Test f, Ags ags1]
transf am@(Pmod states pre acc points) i j (Test f) =
  let
    g = table2fct pre i
  in
    if i == j
    then Test (Conj [g,(Up am f)])
    else Test (Neg Top)
transf am@(Pmod states pre acc points) i j (Conc []) =
  transf am i j (Test Top)
transf am@(Pmod states pre acc points) i j (Conc [p]) = transf am i j p
transf am@(Pmod states pre acc points) i j (Conc (p:ps)) =
  Sum [ Conc [transf am i k p, transf am k j (Conc ps)] | k <- [0..n] ]
  where n = toInteger (length states - 1)
transf am@(Pmod states pre acc points) i j (Sum []) =
  transf am i j (Test (Neg Top))
transf am@(Pmod states pre acc points) i j (Sum [p]) = transf am i j p
transf am@(Pmod states pre acc points) i j (Sum ps) =
  Sum [ transf am i j p | p <- ps ]
transf am@(Pmod states pre acc points) i j (Star p) = kleene am i j n p
  where n = toInteger (length states)

```

Implementation of K_{ijk}^A :

```

kleene :: PoAM -> Integer -> Integer -> Integer -> Program -> Program
kleene am i j 0 pr =
  if i == j
    then Sum [Test Top, transf am i j pr]
    else transf am i j pr
kleene am i j k pr
  | i == j && j == pred k = Star (kleene am i i i pr)
  | i == pred k           =
    Conc [Star (kleene am i i i pr), kleene am i j i pr]
  | j == pred k           =
    Conc [kleene am i j j pr, Star (kleene am j j j pr)]
  | otherwise             =
    Sum [kleene am i j k' pr,
          Conc [kleene am i k' k' pr,
                Star (kleene am k' k' k' pr), kleene am k' j k' pr]]
  where k' = pred k

```

Transformation plus simplification:

```

tfm :: PoAM -> Integer -> Integer -> Program -> Program
tfm am i j pr = simpl (transf am i j pr)

```

The program transformations can be used to translate Update PDL to PDL, as follows:

$$\begin{aligned}
t(\top) &= \top \\
t(p) &= p \\
t(\neg\varphi) &= \neg t(\varphi) \\
t(\varphi_1 \wedge \varphi_2) &= t(\varphi_1) \wedge t(\varphi_2) \\
t([\pi]\varphi) &= [r(\pi)]t(\varphi) \\
t([A, s]\top) &= \top \\
t([A, s]p) &= t(\text{pre}(s)) \rightarrow p \\
t([A, s]\neg\varphi) &= t(\text{pre}(s)) \rightarrow \neg t([A, s]\varphi) \\
t([A, s](\varphi_1 \wedge \varphi_2)) &= t([A, s]\varphi_1) \wedge t([A, s]\varphi_2) \\
t([A, s_i][\pi]\varphi) &= \bigwedge_{j=0}^{n-1} [T_{ij}^A(r(\pi))]t([A, s_j]\varphi) \\
t([A, s][A', s']\varphi) &= t([A, s]t([A', s']\varphi)) \\
t([A, S]\varphi) &= \bigwedge_{s \in S} t([A, s]\varphi) \\
r(a) &= a \\
r(B) &= B \\
r(? \varphi) &= ?t(\varphi) \\
r(\pi_1; \pi_2) &= r(\pi_1); r(\pi_2) \\
r(\pi_1 \cup \pi_2) &= r(\pi_1) \cup r(\pi_2) \\
r(\pi^*) &= (r(\pi))^*.
\end{aligned}$$

The correctness of this translation follows from direct semantic inspection, using the program transformation lemma for the translation of $[A, s_i][\pi]\varphi$ formulas.

The crucial clauses in this translation procedure are those for formulas of the forms $[A, S]\varphi$ and $[A, s]\varphi$, and more in particular the one for formulas of the form $[A, s][\pi]\varphi$. It makes sense to give separate functions for the steps that pull the update model through program π given formula φ .

```

step0, step1 :: PoAM -> Program -> Form -> Form
step0 am@(Pmod states pre acc []) pr f = Top
step0 am@(Pmod states pre acc [i]) pr f = step1 am pr f
step0 am@(Pmod states pre acc is) pr f =
  Conj [ step1 (Pmod states pre acc [i]) pr f | i <- is ]
step1 am@(Pmod states pre acc [i]) pr f =
  Conj [ Pr (transf am i j (rpr pr))
        (Up (Pmod states pre acc [j]) f) | j <- states ]

```


Perform a single step, and put in canonical form:

```
step :: PoAM -> Program -> Form -> Form
step am pr f = canonF (step0 am pr f)
```

```
t :: Form -> Form
t Top = Top
t (Prop p) = Prop p
t (Neg f) = Neg (t f)
t (Conj fs) = Conj (map t fs)
t (Disj fs) = Disj (map t fs)
t (Pr pr f) = Pr (rpr pr) (t f)
t (K x f) = Pr (Ag x) (t f)
t (EK xs f) = Pr (Ags xs) (t f)
t (CK xs f) = Pr (Star (Ags xs)) (t f)
```

Translations of formulas starting with an action model update:

```
t (Up am@(Pmod states pre acc [i]) f) = t' am f
t (Up am@(Pmod states pre acc is) f) =
  Conj [ t' (Pmod states pre acc [i]) f | i <- is ]
```

Translations of formulas starting with a single pointed action model update are performed by t' :

```
t' :: PoAM -> Form -> Form
t' am Top = Top
t' am (Prop p) = impl (precondition am) (Prop p)
t' am (Neg f) = Neg (t' am f)
t' am (Conj fs) = Conj (map (t' am) fs)
t' am (Disj fs) = Disj (map (t' am) fs)
t' am (K x f) = t' am (Pr (Ag x) f)
t' am (EK xs f) = t' am (Pr (Ags xs) f)
t' am (CK xs f) = t' am (Pr (Star (Ags xs)) f)
t' am (Up am' f) = t' am (t (Up am' f))
```

The crucial case: update action having scope over a program. We may assume that the update action is single pointed.

```
t' am@(Pmod states pre acc [i]) (Pr pr f) =
  Conj [ Pr (transf am i j (rpr pr))
        (t' (Pmod states pre acc [j]) f) | j <- states ]
t' am@(Pmod states pre acc is) (Pr pr f) =
  error "action model not single pointed"
```

Translations for programs:

```
rpr :: Program -> Program
rpr (Ag x)      = Ag x
rpr (Ags xs)    = Ags xs
rpr (Test f)    = Test (t f)
rpr (Conc ps)   = Conc (map rpr ps)
rpr (Sum ps)    = Sum (map rpr ps)
rpr (Star p)    = Star (rpr p)
```

Translating and putting in canonical form:

```
tr :: Form -> Form
tr = canonF . t
```

Some example translations:

```
ActEpist> tr (Up (public p) (Pr (Star (Ags [b,c])) p))
T
ActEpist> tr (Up (public (Disj [p,q]) (Pr (Star (Ags [b,c])) p))
[(U[?T,C[?v[p,q],[b,c]])*]v[p,&[-p,-q]]
ActEpist> tr (Up (groupM [a,b] p) (Pr (Star (Ags [b,c])) p))
[C[C[(U[?T,C[?p,[b,c]])*],C[?p,[c]]],(U[U[?T,[b,c]],C[c,(U[?T,C[?p,[b,c]])*],C[?p,[c]]])*)]p
ActEpist> tr (Up (secret [a,b] p) (Pr (Star (Ags [b,c])) p))
[C[C[(U[?T,C[?p,[b]]])*],C[?p,[c]]],(U[U[?T,[b,c]],C[?-T,(U[?T,C[?p,[b]]])*],C[?p,[c]]])*)]p
```

6.6 Automata

Probably the translation from the previous section is all you will ever need. This section was left in for sentimental reasons.

Alphabet: accessibility steps for individual agents plus tests on formulas of the language:

```
data Symbol = Acc Agent | Tst Form deriving (Eq,Ord,Show)
```

Moves are triples consisting of start state, a symbol read and a next state.

```
data (Eq a,Ord a,Show a) => Move a = Move a Symbol a deriving (Eq,Ord,Show)
```

Automata are triples consisting of a start state, a list of possible moves and a final state (thus, the state set is left implicit).

```
data (Eq a,Ord a,Show a) => NFA a = NFA a [Move a] a deriving (Eq,Ord,Show)
```

The states of an automaton:

```
states :: (Eq a,Ord a,Show a) => NFA a -> [a]
states (NFA s delta f) = (sort . nub) (s:f:rest)
  where rest = [ s' | Move s' a t' <- delta ]
              ++
              [ t' | Move s' a t' <- delta ]
```

The symbols of an NFA:

```
symbols :: (Eq a,Ord a,Show a) => NFA a -> [Symbol]
symbols (NFA s moves f) = (sort . nub) [ symb | Move s symb t <- moves ]
```

Recognizing strings of symbols. If there is no input left, check if the start state coincides with the final state. Otherwise, construct the automata that result from reading the first symbol with the current automaton, and check if any of these accepts the rest of the input.

```

recog :: (Eq a,Ord a,Show a) => NFA a -> [Symbol] -> Bool
recog (NFA start moves final) [] = start == final
recog (NFA start moves final) (symbol:symbols) =
  any (\ aut -> recog aut symbols)
    [ NFA new moves final |
      Move s symb new <- moves, s == start, symb == symbol ]

```

Computing the reachable states of an automaton, using a well-known algorithm for graph reachability [35, Ch 1]:

```

reachable :: (Eq a,Ord a,Show a) => NFA a -> [a]
reachable (NFA start moves final) = acc moves [start] []
  where
    acc :: (Show a,Ord a) => [Move a] -> [a] -> [a] -> [a]
    acc moves [] marked = marked
    acc moves (b:bs) marked = acc moves (bs ++ (cs \\ bs)) (marked ++ cs)
      where
        cs = nub [ c | Move b' symb c <- moves, b' == b, notElem c marked ]

```

Simplify an automaton by removing non-accessible states from the transition relation.

```

accNFA :: (Eq a,Ord a,Show a) => NFA a -> NFA a
accNFA nfa@(NFA start moves final) =
  if
    notElem final fromStart
  then
    NFA start [] final
  else
    NFA start moves' final
  where
    fromStart = reachable nfa
    moves' = [ Move x symb y | Move x symb y <- moves, elem x fromStart ]

```

Minimizing the number of states of a finite automaton can be done with the following algorithm [27]:

- Start out with a partition $\{S - \{f\}, \{f\}\}$ of the state set of the automaton (S is the set of all states, f is the final state).

- Given a partition Π , for each block b in Π , partition b into sub-blocks such that two states s, t of b are in the same sub-block iff for all symbols σ it holds that s and t have $\xrightarrow{\sigma}$ transitions to states in the same block of Π . Update Π to Π' by replacing each b in Π by the newly found set of sub-blocks for b .
- Halt as soon as $\Pi = \Pi'$.

The initial partition:

```
initPart :: (Eq a, Ord a, Show a) => NFA a -> [[a]]
initPart nfa@(NFA start moves final) = [states nfa \\ [final], [final]]
```

Refining a partition:

```
refinePart :: (Eq a, Ord a, Show a) => NFA a -> [[a]] -> [[a]]
refinePart nfa p = refineP nfa p p
  where
    refineP :: (Eq a, Ord a, Show a) => NFA a -> [[a]] -> [[a]] -> [[a]]
    refineP nfa part [] = []
    refineP nfa@(NFA start moves final) part (block:blocks) =
      newblocks ++ (refineP nfa part blocks)
      where
        newblocks =
          rel2part block (\ x y -> sameAccBl nfa part x y)
```

Function that checks whether two states have the same accessible blocks under a partition:

```
sameAccBl :: (Eq a, Ord a, Show a) => NFA a -> [[a]] -> a -> a -> Bool
sameAccBl nfa part s t =
  and [ accBl nfa part s symb == accBl nfa part t symb |
        symb <- symbols nfa ]
```

Accessible blocks for a symbol from a given state, given an NFA and a partition:

```
accBl :: (Eq a, Ord a, Show a) => NFA a -> [[a]] -> a -> Symbol -> [[a]]
accBl nfa@(NFA start moves final) part s symb =
  nub [ bl part y | Move x symb' y <- moves, symb' == symb, x == s ]
```

The whole algorithm:

```
compress :: (Eq a, Ord a, Show a) => NFA a -> [[a]]
compress nfa = compress' nfa (initPart nfa)
  where
    compress' :: (Eq a, Ord a, Show a) => NFA a -> [[a]] -> [[a]]
    compress' nfa part = if rpart == part
                        then part
                        else compress' nfa rpart
    where rpart = refinePart nfa part
```

Use this to construct the minimal automaton. We employ the opportunity to put the test formulas in the transition table of the automaton in canonical form.

```
minimalAut' :: (Eq a, Ord a, Show a) => NFA a -> NFA [a]
minimalAut' nfa@(NFA start moves final) = NFA start' moves' final'
  where
    (NFA st mov fin) = accNFA nfa
    partition        = compress (NFA st mov fin)
    f                = bl partition
    g (Acc ag)       = Acc ag
    g (Tst frm)      = Tst (canonF frm)
    start'           = f st
    final'           = f fin
    moves'           = (nub.sort)
                     (map (\ (Move x y z) -> Move (f x) (g y) (f z)) mov)
```

Converting an automaton of type NFA a to one of type NFA State:

```
convAut :: (Eq a, Ord a, Show a) => NFA a -> NFA State
convAut aut@(NFA s delta t) =
  NFA
    (f s)
    (map (\ (Move x symb y) -> Move (f x) symb (f y)) delta)
    (f t)
  where f = convert (states aut)
```

Simplify the output of minimalAut':

```
minimalAut :: (Eq a, Ord a, Show a) => NFA a -> NFA State
minimalAut = convAut . minimalAut'
```

Automaton that accepts nothing:

```
nullAut = (NFA 0 [] 1)
```

General knowledge automaton:

```
genKnown :: [Agent] -> NFA State
genKnown agents = (NFA 0 [Move 0 (Acc a) 1 | a <- agents ] 1)
```

Relativized common knowledge automaton:

```
relCknown :: [Agent] -> Form -> NFA State
relCknown agents form = (NFA 0 (Move 0 (Tst form) 1 :
                               [Move 1 (Acc a) 0 | a <- agents]) 0)
```

Common knowledge automaton:

```
cKnown :: [Agent] -> NFA State
cKnown agents = (NFA 0 [Move 0 (Acc a) 0 | a <- agents] 0)
```

Implementation of the function AUT:

```

aut' :: (Show a, Ord a) =>
  PoAM -> State -> State -> NFA a -> NFA (State, Int, a)
aut' (Pmod sts pre acc _) s t (NFA start delta final) =
  (NFA (s, 0, start) delta' (t, 1, final)) where
    delta' = [ Move (u, 1, w) (Acc a) (v, 0, x) |
      (a, u, v) <- acc,
      Move w (Acc a') x <- delta,
      a == a' ]
    ++
    [ Move (u, 0, w) (Tst (table2fct pre u)) (u, 1, w) |
      u <- sts,
      w <- states (NFA start delta final) ]
    ++
    [ Move (u, 1, v)
      (Tst (Neg (Up (Pmod sts pre acc [u])
        (Neg form)))) (u, 1, w) |
      u <- sts,
      Move v (Tst form) w <- delta ]

```

Simplify the output of the aut' function:

```

aut :: (Show a, Ord a) => PoAM -> State -> State -> NFA a -> NFA State
aut am s t nfa = minimalAut (aut' am s t nfa)

```

The aut function is the key ingredient of the following translation function from the language of epistemic update logic to APDL:

```

tr' :: Form -> Form
tr' Top = Top
tr' (Prop p) = Prop p
tr' (Neg form) = Neg (tr' form)
tr' (Conj forms) = Conj (map tr' forms)
tr' (Disj forms) = Disj (map tr' forms)
tr' (K agent form) = K agent (tr' form)
tr' (EK agents form) = Aut (genKnown agents) (tr' form)
tr' (CK agents form) = Aut (cKnown agents) (tr' form)

```



```

tr' (Aut nfa form) = Aut (tAut nfa) (tr' form)
tr' (Up (Pmod sts pre rel []) form) = Top
tr' (Up (Pmod sts pre rel [s]) Top) = Top
tr' (Up (Pmod sts pre rel [s]) (Prop p)) =
  impl (tr' (table2fct pre s)) (Prop p)
tr' (Up (Pmod sts pre rel [s]) (Neg form)) =
  impl (tr' (table2fct pre s))
    (Neg (tr' (Up (Pmod sts pre rel [s]) form)))
tr' (Up (Pmod sts pre rel [s]) (Conj forms)) =
  Conj [ tr' (Up (Pmod sts pre rel [s]) form) | form <- forms ]
tr' (Up (Pmod sts pre rel [s]) (Disj forms)) =
  Disj [ tr' (Up (Pmod sts pre rel [s]) form) | form <- forms ]
tr' (Up (Pmod sts pre rel [s]) (K agent form)) =
  impl (tr' (table2fct pre s))
    (Conj [ K agent (tr' (Up (Pmod sts pre rel [t]) form)) |
            t <- sts ])
tr' (Up (Pmod sts pre rel [s]) (EK agents form)) =
  tr' (Up (Pmod sts pre rel [s]) (Aut (genKnown agents) form))
tr' (Up (Pmod sts pre rel [s]) (CK agents form)) =
  tr' (Up (Pmod sts pre rel [s]) (Aut (cKnown agents) form))

```

```

tr' (Up (Pmod sts pre rel [s]) (Aut nfa form)) =
  Conj [ tr' (Aut (aut (Pmod sts pre rel [s]) s t nfa)
                (Up (Pmod sts pre rel [t]) form)) | t <- sts ]
tr' (Up (Pmod sts pre rel [s]) (Up (Pmod sts' pre' rel' points) form)) =
  tr' (Up (Pmod sts pre rel [s])
        (tr' (Up (Pmod sts' pre' rel' points) form)))
tr' (Up (Pmod sts pre rel points) form) =
  Conj [ tr' (Up (Pmod sts pre rel [s]) form) | s <- points ]

```

Note that this translation generalizes the translation function from [30] to the dynamic epistemic language with multiple pointed action models. The translation demonstrates that non-deterministic dynamic epistemic logic also reduces to automata PDL.

Translating and putting in canonical form:

```

kvbtr :: Form -> Form
kvbtr = canonF . tr'

```

Translation of the test formulas inside the transition relation of an automaton:

```
tAut :: NFA State -> NFA State
tAut (NFA s delta f) = NFA s (map trans delta) f
  where trans (Move u (Acc x) v)      = Move u (Acc x) v
        trans (Move u (Tst form) v) = Move u (Tst (kvbtr form)) v
```

Some example translations:

```
ActEpist> kvbtr (Up (public p) (K a p))
T
ActEpist> kvbtr (Up (public p) (K a q))
v[&[p,[a]v[&[p,q],-p]],-p]
ActEpist> kvbtr (Up (public p) (CK [a,b] p))
T
ActEpist> kvbtr (Up (public p) (CK [a,b] q))
[NFA 0 [Move 0 (Tst p) 1,Move 1 (Acc a) 0,Move 1 (Acc b) 0] 1]v[&[p,q],-p]
ActEpist> tr (Up (public p) (CK [a,b] q))
[(U[?T,C[?p,[a,b]]])*]v[&[p,q],-p]
```

Chapter 7

Model Minimization under Action Emulation

7.1 Module Declaration

```
module MinAE
where

import List
import Models
import MinBis
import ActEpist
```

Module `Models` is given in Chapter 3, module `MinBis` in Chapter 5, module `ActEpist` in Chapter 6.

7.2 Action Emulation

A structural relation of action emulation, weaker than bisimulation and intended to exactly capture the update effects of action models is defined in [18], as follows.

Definition 5 (Action Emulation) *If \mathbf{A} and \mathbf{B} are actions with sets of action states W_A, W_B , respectively, then a relation $R \subseteq W_A \times W_B$ is an action emulation if whenever sRt the following hold:*

Preconditions $pre(s) \wedge pre(t)$ *is consistent.*

Zig If $s \xrightarrow{a} s'$ then there are t_1, \dots, t_n with

$$t \xrightarrow{a} t_1, \dots, t \xrightarrow{a} t_n, s'Rt_1, \dots, s'Rt_n \text{ and } \text{pre}(s') \models \text{pre}(t_1) \vee \dots \vee \text{pre}(t_n).$$

Zag If $t \xrightarrow{a} t'$ then there are s_1, \dots, s_n with

$$s \xrightarrow{a} s_1, \dots, s \xrightarrow{a} s_n, s_1Rt', \dots, s_nRt' \text{ and } \text{pre}(t') \models \text{pre}(s_1) \vee \dots \vee \text{pre}(s_n).$$

Definition 6 A total action emulation between \mathbf{A} and \mathbf{B} is an action emulation $R \subseteq W_A \times W_B$ satisfying the following extra requirement: For every $s \in W_A$ there are $t_1, \dots, t_n \in W_B$ such that sRt_1, \dots, sRt_n and $\text{pre}(s) \models \text{pre}(t_1) \vee \dots \vee \text{pre}(t_n)$, and for every $t \in W_B$ there are $s_1, \dots, s_n \in W_A$ with s_1Rt, \dots, s_nRt and $\text{pre}(t) \models \text{pre}(s_1) \vee \dots \vee \text{pre}(s_n)$.

7.3 Partition Refinement Again

Any action model can be simplified by replacing each state s by its action emulation class $[s]$. The problem of finding the smallest action model modulo action emulation is similar to the problem of minimizing under bisimulation. We again use partition refinement [34]. Here is the adapted algorithm:

1. Take the generated submodel of the action model, while throwing away all states s with $\text{pre}(s) \equiv \perp$.
2. Use partition refinement to compute the bisimulation minimal version of the action model.
3. Do a root unfolding if the action model is multi-pointed.
4. Generate a partition Π of the new state set where all states with the same predecessors (for all agents) and the same successors (for all agents) are in the same class.
5. Use partition refinement, starting from Π , to minimize the result.
6. Set the precondition of each partition block B equal to $\bigvee_{s \in B} \text{pre}(s)$.
7. Use partition refinement again to compute the bisimulation minimal version of the new model.

This is an extension of the algorithm for finding bisimulation minimal models.

The function for doing a root unfolding.

```

unfold :: PoAM -> PoAM
unfold (Pmod states pre acc [])      = zero
unfold am@(Pmod states pre acc [p]) = am
unfold (Pmod states pre acc points) =
  Pmod states' pre' acc' points'
  where
    len = toInteger (length states)
    points' = [ p + len | p <- points ]
    states' = states ++ points'
    pre'    = pre ++ [ (j+len,f) | (j,f) <- pre, k <- points, j == k ]
    acc'    = acc ++ [ (ag,i+len,j) | (ag,i,j) <- acc, k <- points, i == k ]

```

Finding the predecessors and successors of a state, for a given relation:

```

preds, sucs :: (Eq a, Ord a, Eq b, Ord b) => [(a,b,b)] -> b -> [(a,b)]
preds rel s = (sort.nub) [ (ag,s1) | (ag,s1,s2) <- rel, s == s2 ]
sucs rel s = (sort.nub) [ (ag,s2) | (ag,s1,s2) <- rel, s == s1 ]

```

Initializing a partition based on same predecessors and same successors:

```

psPartition :: (Eq a, Ord a, Eq b) => Model a b -> [[a]]
psPartition (Mo states pre rel) =
  rel2part states (\ x y -> preds rel x == preds rel y
                    &&
                    sucs rel x == sucs rel y)

```

Implementation of parts 4,5,6 of the algorithm for finding the action emulation minimal Pmod:

```

minPmod :: (Eq a, Ord a) => Pmod a Form -> Pmod [a] Form
minPmod pm@(Pmod states pre rel pts) =
  (Pmod states' pre' rel' pts')
  where
    m          = Mo states pre rel
    partition = refine m (psPartition m)
    states'    = partition
    f          = bl partition
    g          = \ xs -> canonF (Disj (map (table2fct pre) xs))
    rel'       = (nub.sort) (map (\ (x,y,z) -> (x, f y, f z)) rel)
    pre'       = zip states' (map g states')
    pts'       = map (bl states') pts

```

Reducing Pmods under action emulation:

```

aePmod :: (Eq a, Ord a, Show a) => Pmod a Form -> Pmod State Form
--aePmod :: PoAM -> PoAM
aePmod = (bisimPmod propEquiv) . minPmod . unfold .
          (bisimPmod propEquiv) . gsmPoAM . convPmod

```

Chapter 8

Semantics

We now turn to the implementation of the semantics module.

8.1 Module Declaration

```
module Semantics
where

import List
import Char
import Models
import Display
import MinBis
import ActEpist
import MinAE
import DPLL
```

Here `List` and `Char` are standard Haskell modules. Module `Models` gives a general definition of Kripke models and is given in Chapter 3. Module `Display`, given in Chapter 4, handles model display and visualisation. `MinBis` is a module for model minimization under bisimulation, and is given in Chapter 5. `ActEpist` defines action models and epistemic models as specific cases of models; see Chapter 6. Module `MinAE` for minimization under action emulation is given in Chapter 7. `DPLL` is a module for propositional reasoning with the Davis, Putnam, Logemann, Loveland procedure [9, 10] listed in Appendix B.

8.2 Semantics

The group alternatives of group of agents a are the states that are reachable through $\bigcup_{a \in A} R_a$.

```
groupAlts :: [(Agent,State,State)] -> [Agent] -> State -> [State]
groupAlts rel agents current =
  (nub . sort . concat) [ alternatives rel a current | a <- agents ]
```

The common knowledge alternatives of group of agents a are the states that are reachable through a finite number of R_a links, for $a \in A$.

```
commonAlts :: [(Agent,State,State)] -> [Agent] -> State -> [State]
commonAlts rel agents current =
  closure rel agents (groupAlts rel agents current)
```

The model update function takes a static model and an action model and returns an object of type `Model (State,State) [Prop]`. The up function takes an epistemic model and a PoAM and returns a Pmod. Its states are the `(State,State)` pairs that result from the cartesian product construction described in [2]. Note that the update function uses the truth definition (given below as `isTrAt`).

```
up :: EpistM -> PoAM -> Pmod (State,State) [Prop]
up m@(Pmod worlds val acc points) am@(Pmod states pre susp actuals) =
  Pmod worlds' val' acc' points'
  where
    worlds' = [ (w,s) | w <- worlds, s <- states,
                  formula <- maybe [] (\ x -> [x]) (lookup s pre),
                  isTrAt w m formula
                ]
    val'     = [ ((w,s),props) | (w,props) <- val,
                              s <- states,
                              elem (w,s) worlds'
                            ]
    acc'     = [ (ag1,(w1,s1),(w2,s2)) | (ag1,w1,w2) <- acc,
                              (ag2,s1,s2) <- susp,
                              ag1 == ag2,
                              elem (w1,s1) worlds',
                              elem (w2,s2) worlds'
                            ]
    points' = [ (p,a) | p <- points, a <- actuals,
                  elem (p,a) worlds'
                ]
```


The appropriate notion of equivalence for the base case of the bisimulation for epistemic models is “having the same valuation”.

```
sameVal :: [Prop] -> [Prop] -> Bool
sameVal ps qs = (nub . sort) ps == (nub . sort) qs
```

Bisimulation minimal version of generated submodel of update result for epistemic model and PoAM:

```
upd :: EpistM -> PoAM -> EpistM
upd sm am = (bisimPmod (sameVal) . convPmod) (up sm am)
```

Non-deterministic update with a list of PoAMs:

```
upds :: EpistM -> [PoAM] -> EpistM
upds = foldl upd
```

For the interpretation of automata operators we need to find the worlds that are reachable from a given world in a model through a path accepted by a given automaton. This uses the following modification of the familiar graph reachability algorithm described in [35, Ch 1] (L and K are lists of pairs consisting of worlds in the model and states in the automaton):

- Start out with a list $L = [(w, s)]$, where w is the given world and s is the start state of the automaton, and with an empty list K of marked pairs.
- Repeat until L is empty:
 - Delete the first member (w', s') from L .
 - If $(s', ?\varphi, s'')$ is a move in the automaton, for some φ true at w' in the model, with $(w', s'') \notin K$, then add (w', s'') to L and to K .
 - If (s', a, s'') is a move in the automaton, for some agent label a , and $w' \xrightarrow{a} w''$ in the model, with $(w'', s'') \notin K$, then add (w'', s'') to L and to K .
- The reachable worlds are the worlds occurring in K paired with the final state of the automaton.

This algorithm is implemented by the following function:

```

reachableAut :: SM -> NFA State -> State -> [State]
reachableAut model nfa@(NFA start moves final) w =
  acc model nfa [(w,start)] []
  where
    acc :: SM -> NFA State -> [(State,State)] -> [(State,State)] -> [State]
    acc model (NFA start moves final) [] marked =
      (sort.nub) (map fst (filter (\ x -> snd x == final) marked))
    acc m@(Mo states _ rel) nfa@(NFA start moves final)
      ((w,s):pairs) marked =
      acc m nfa (pairs ++ (cs \ \ pairs)) (marked ++ cs)
    where
      cs = nub ([ (w, s') | Move t (Tst f) s' <- moves,
                      t == s, notElem (w,s') marked,
                      isTrueAt w m f ]
                ++
                [ (w',s') | Move t (Acc ag) s' <- moves, t == s,
                      w' <- states,
                      notElem (w',s') marked,
                      elem (ag,w,w') rel ])

```

At last we have all ingredients for the truth definition.

```

isTrueAt :: State -> SM -> Form -> Bool
isTrueAt w m Top = True
isTrueAt w m@(Mo worlds val acc) (Prop p) =
  elem p (concat [ props | (w',props) <- val, w'==w ])
isTrueAt w m (Neg f) = not (isTrueAt w m f)
isTrueAt w m (Conj fs) = and (map (isTrueAt w m) fs)
isTrueAt w m (Disj fs) = or (map (isTrueAt w m) fs)

```

The clauses for individual knowledge, general knowledge and common knowledge use the functions `alternatives`, `groupAlts` and `commonAlts` to compute the relevant accessible worlds:

```

isTrueAt w m@(Mo worlds val acc) (K ag f) =
  and (map (flip ((flip isTrueAt) m) f) (alternatives acc ag w))
isTrueAt w m@(Mo worlds val acc) (EK agents f) =
  and (map (flip ((flip isTrueAt) m) f) (groupAlts acc agents w))
isTrueAt w m@(Mo worlds val acc) (CK agents f) =
  and (map (flip ((flip isTrueAt) m) f) (commonAlts acc agents w))

```

In the clause for $[\mathbf{M}]\varphi$, the result of updating the static model M with action model \mathbf{M} may be undefined, but in this case the precondition $P(s_0)$ of the designated state s_0 of \mathbf{M} will fail in the designated world w_0 of M . By making the clause for $[\mathbf{M}]\varphi$ check for $M \models_{w_0} P(s_0)$, truth can be defined as a total function.

```
isTrueAt w m (Up am f) =
  and [ isTrueAt w' m' f |
        (m',w') <- decompose (upd (mod2pmod m [w]) am) ]
isTrueAt w m (Aut nfa f) =
  and [ isTrueAt w' m f | w' <- reachableAut m nfa w ]
```

Checking for truth in the actual world of an epistemic model:

```
isTrAt :: State -> EpistM -> Form -> Bool
isTrAt w (Pmod worlds val rel pts) = isTrueAt w (Mo worlds val rel)
```

Checking for truth in *all* the designated points of an epistemic model:

```
isTrue :: EpistM -> Form -> Bool
isTrue (Pmod worlds val rel pts) form =
  and [ isTrueAt w (Mo worlds val rel) form | w <- pts ]
```

8.3 Tools for Constructing Epistemic Models

The following function constructs an initial epistemic model where the agents are completely ignorant about their situation, as described by a list of basic propositions. The input is a list of basic propositions used for constructing the valuations.

```

initE :: [Prop] -> EpistM
initE allProps = (Pmod worlds val accs points)
  where
    worlds = [0..(2k - 1)]
    k      = length allProps
    val     = zip worlds (sortL (powerList allProps))
    accs    = [ (ag,st1,st2) | ag <- all_agents,
                                st1 <- worlds,
                                st2 <- worlds      ]

    points = worlds

```

This uses the following utilities:

```

powerList :: [a] -> [[a]]
powerList [] = [[]]
powerList (x:xs) = (powerList xs) ++ (map (x:) (powerList xs))

sortL :: Ord a => [[a]] -> [[a]]
sortL = sortBy (\ xs ys -> if length xs < length ys then LT
                           else if length xs > length ys then GT
                           else compare xs ys)

```

Some initial models:

```

e00 :: EpistM
e00 = initE [P 0]

e0 :: EpistM
e0 = initE [P 0, Q 0]

```

8.4 From Communicative Actions to Action Models

Computing the update for a public announcement:

```

public :: Form -> PoAM
public form =
  (Pmod [0] [(0,form)] [ (a,0,0) | a <- all_agents ] [0])

```

Public announcements are S5 models:

```

DEMO> showM (public p)
==> [0]
[0]
(0,p)
(a,[[0]])
(b,[[0]])
(c,[[0]])

```

Computing the update for passing a group announcement to a list of agents: the other agents may or may not be aware of what is going on. In the limit case where the message is passed to all agents, the message is a public announcement.

```

groupM :: [Agent] -> Form -> PoAM
groupM agents form =
  if (sort agents) == all_agents
  then public form
  else
    (Pmod
      [0,1]
      [(0,form),(1,Top)]
      ([ (a,0,0) | a <- all_agents ]
        ++ [ (a,0,1) | a <- all_agents \\ agents ]
        ++ [ (a,1,0) | a <- all_agents \\ agents ]
        ++ [ (a,1,1) | a <- all_agents ]))
    [0])

```

Group announcements are S5 models:

```

DEMO> showM (groupM [a,b] p)
==> [0]
[0,1]
(0,p)(1,T)
(a,[[0],[1]])
(b,[[0],[1]])
(c,[[0],[1]])

```

Computing the update for an individual message to b that φ :

```
message :: Agent -> Form -> PoAM
message agent form = groupM [agent] form
```

Computing the update for passing a *secret* message to a list of agents: the other agents remain unaware of the fact that something goes on. In the limit case where the secret is divulged to all agents, the secret becomes a public update.

```
secret :: [Agent] -> Form -> PoAM
secret agents form =
  if (sort agents) == all_agents
  then public form
  else
    (Pmod
     [0,1]
     [(0,form),(1,Top)]
     ([ (a,0,0) | a <- agents ]
      ++ [ (a,0,1) | a <- all_agents \\ agents ]
      ++ [ (a,1,1) | a <- all_agents ])
     [0])
```

Secret messages are KD45 models:

```
DEMO> showM (secret [a,b] p)
==> [0]
[0,1]
(0,p)(1,T)
(a,[[],[0]],[[],[1]])
(b,[[],[0]],[[],[1]])
(c,[([0],[1]))
```

To Do 4 *Add functions for messages with bcc.*

A special case of a secret is a test. Tests are updates that are kept secret from all agents:

```
test :: Form -> PoAM
test = secret []
```

Testing for $p \vee q$ is done with the following KD45 action model:

```

DEMO> showM (test (Disj [p,q]))
==> [0]
[0,1]
(0,v[p,q])(1,T)
(a,[[[0],[1]]])
(b,[[[0],[1]]])
(c,[[[0],[1]]])

```

Updating a model (M, w_0) with a test φ ? yields a unit list containing (M, w_0) in case $M \models_{w_0} \varphi$, the empty list otherwise:

```

DEMO> showMs (upd (initM [P 0, Q 0] [P 0]) (test p))
==> 1
[0,1,2,3]
(0,[])(1,[p])(2,[q])(3,[p,q])
(a,[[0,1,2,3]])
(b,[[0,1,2,3]])
(c,[[0,1,2,3]])

```

```

DEMO> showMs (upd (initM [P 0, Q 0] [P 0]) (test (Neg p)))

```

Here is a multiple pointed action model for the communicative action of revealing one of a number of alternatives to a list of agents, in such a way that it is common knowledge that one of the alternatives gets revealed (in [3] this is called *common knowledge of alternatives*).

```

reveal :: [Agent] -> [Form] -> PoAM
reveal ags forms =
  (Pmod
    states
    (zip states forms)
    ([ (ag,s,s) | s <- states, ag <- ags ]
     ++
     [ (ag,s,s') | s <- states, s' <- states, ag <- others ])
    states)
  where states = map fst (zip [0..] forms)
        others = all_agents \\ ags

```

Here is an action model for the communication that reveals to a one of p_1, q_1, r_1 .

```

DEMO> showM (reveal [a] [p1,q1,r1])
==> [0,1,2]
[0,1,2]
(0,p1)(1,q1)(2,r1)
(a,[[[0],[1],[2]]])
(b,[[0,1,2]])
(c,[[0,1,2]])

```

A group of agents B gets (transparently) informed about a formula φ if B get to know φ when φ is true, and B get to know the negation of φ otherwise. Transparency means that all other agents are aware of the fact that B get informed about φ , i.e., the other agents learn that $(\varphi \rightarrow C_B \varphi) \wedge (\neg \varphi \rightarrow C_B \neg \varphi)$. This action model can be defined in terms of **reveal**, as follows:

```
info :: [Agent] -> Form -> PoAM
info agents form = reveal agents [form, negation form]
```

An example application:

```
DEMO> showMs (upd (initM [P 0, Q 0] [P 0]) (info [a,b] q))
==> 1
[0,1,2,3]
(0,[]) (1,[p]) (2,[q]) (3,[p,q])
(a,[[0,1],[2,3]])
(b,[[0,1],[2,3]])
(c,[[0,1,2,3]])

DEMO> isTrue (head (upd (initM [P 0, Q 0] [P 0]) (info [a,b] q))) (CK [a,b] (Neg q))
True
```

8.5 Operations on Action Models

The trivial update action model is a special case of public announcement. Call this the **one** action model, for it behaves as 1 for the operation \otimes of action model composition.

```
one :: PoAM
one = public Top
```

Composition \otimes of multiple pointed action models.


```

cmpP :: PoAM -> PoAM ->
      Pmod (State,State) Form
cmpP m@(Pmod states pre susp ss) (Pmod states' pre' susp' ss') =
  (Pmod nstates npre nsusp npoints)
  where
    npoints = [ (s,s') | s <- ss, s' <- ss' ]
    nstates = [ (s,s') | s <- states, s' <- states' ]
    npre     = [ ((s,s'), g) | (s,f)      <- pre,
                              (s',f')    <- pre',
                              g          <- [computePre m f f']      ]
    nsusp    = [ (ag,(s1,s1'),(s2,s2')) | (ag,s1,s2)    <- susp,
                              (ag',s1',s2') <- susp',
                              ag == ag'      ]

```

Utility function for this: compute the new precondition of a state pair. If the preconditions of the two states are purely propositional, we know that the updates at the states commute and that their combined precondition is the conjunction of the two preconditions, provided this conjunction is not a contradiction. If one of the states has a precondition that is not purely propositional, we have to take the epistemic effect of the update into account in the new precondition.

```

computePre :: PoAM -> Form -> Form -> Form
computePre m g g' | pureProp conj = conj
                  | otherwise      = Conj [ g, Neg (Up m (Neg g')) ]
  where conj      = canonF (Conj [g,g'])

```

To Do 5 *Refine the precondition computation, by making more clever use of what is known about the update effect of the first action model.*

Compose pairs of multiple pointed action models, and reduce the result to its simplest possible form under action emulation.

```

cmpPoAM :: PoAM -> PoAM -> PoAM
cmpPoAM pm pm' = aePmod (cmpP pm pm')

```

Use `one` as unit for composing lists of PoAMs:

```

cmp :: [PoAM] -> PoAM
cmp = foldl cmpPoAM one

```

Here is the result of composing two messages:

```

DEMO> showM (cmp [groupM [a,b] p, groupM [b,c] q])
==> [0]
[0,1,2,3]
(0,&[p,q])(1,p)(2,q)(3,T)
(a,[ [0,1], [2,3] ])
(b,[ [0], [1], [2], [3] ])
(c,[ [0,2], [1,3] ])

```

This gives the resulting action model. Here is the result of composing the messages in the reverse order:

```

DEMO> showM (cmp [groupM [b,c] q, groupM [a,b] p])
==> [0]
[0,1,2,3]
(0,&[p,q])(1,q)(2,p)(3,T)
(a,[ [0,2], [1,3] ])
(b,[ [0], [1], [2], [3] ])
(c,[ [0,1], [2,3] ])

```

These two action models are bisimilar under the renaming $1 \mapsto 2, 2 \mapsto 1$.

Here is an illustration of an observation from [15].

```

m2 = initE [P 0,Q 0]
psi = Disj[Neg(K b p),q]

```

```

DEMO> showM (upds m2 [message a psi, message b p])
==> [1,4]
[0,1,2,3,4,5]
(0,[])(1,[p])(2,[p])(3,[q])(4,[p,q])
(5,[p,q])
(a,[ [0,1,2,3,4,5] ])
(b,[ [0,2,3,5], [1,4] ])
(c,[ [0,1,2,3,4,5] ])
DEMO> showM (upds m2 [message b p, message a psi])
==> [7]
[0,1,2,3,4,5,6,7,8,9,10]

```

```

(0, []) (1, []) (2, [p]) (3, [p]) (4, [p])
(5, [q]) (6, [q]) (7, [p,q]) (8, [p,q]) (9, [p,q])
(10, [p,q])
(a, [[0,3,5,7,9], [1,2,4,6,8,10]])
(b, [[0,1,3,4,5,6,9,10], [2,7,8]])
(c, [[0,1,2,3,4,5,6,7,8,9,10]])

```

Power of action models:

```

pow :: Int -> PoAM -> PoAM
pow n am = cmp (take n (repeat am))

```

The Van Benthem test: keep updating an epistemic model with the same action model until a fixpoint is reached.

```

vBtest :: EpistM -> PoAM -> [EpistM]
vBtest m a = map (upd m) (star one cmpPoAM a)

star :: a -> (a -> a -> a) -> a -> [a]
star z f a = z : star (f z a) f a

```

Putting in the fixpoint:

```

vBfix :: EpistM -> PoAM -> [EpistM]
vBfix m a = fix (vBtest m a)

fix :: Eq a => [a] -> [a]
fix (x:y:zs) = if x == y then [x]
               else x: fix (y:zs)

```

Illustration of the above for the update with an action model **S** based on

$$p_1 \wedge (\Box_a p_1 \Rightarrow p_2) \wedge (\Box_a p_2 \Rightarrow p_3).$$

```

m1 = initE [P 1,P 2,P 3]
phi = Conj[p1,Neg (Conj[K a p1,Neg p2]),
           Neg (Conj[K a p2,Neg p3])]
a1 = message a phi

```

Three updates with $A = \mathbf{a1}$ are needed before a is aware of the three facts p_1, p_2, p_3 , and four before the update process reaches its fixpoint. The example shows that $A \not\Leftarrow A^2 \not\Leftarrow A^3 \not\Leftarrow A^4 \Leftarrow A^5$. Note that it takes a *long* time to generate these five updates.

```

DEMO> showMs (vBtest m1 a1)
==> [0,1,2,3,4,5,6,7]
[0,1,2,3,4,5,6,7]
(0, []) (1, [p1]) (2, [p2]) (3, [p3]) (4, [p1,p2])
(5, [p1,p3]) (6, [p2,p3]) (7, [p1,p2,p3])
(a, [[0,1,2,3,4,5,6,7]])
(b, [[0,1,2,3,4,5,6,7]])
(c, [[0,1,2,3,4,5,6,7]])
==> [1,5,7,10]
[0,1,2,3,4,5,6,7,8,9,10,11]
(0, []) (1, [p1]) (2, [p1]) (3, [p2]) (4, [p3])
(5, [p1,p2]) (6, [p1,p2]) (7, [p1,p3]) (8, [p1,p3]) (9, [p2,p3])
(10, [p1,p2,p3]) (11, [p1,p2,p3])
(a, [[0,2,3,4,6,8,9,11], [1,5,7,10]])
(b, [[0,1,2,3,4,5,6,7,8,9,10,11]])
(c, [[0,1,2,3,4,5,6,7,8,9,10,11]])
==> [5,11]
[0,1,2,3,4,5,6,7,8,9,10,11,12,13]
(0, []) (1, [p1]) (2, [p1]) (3, [p2]) (4, [p3])
(5, [p1,p2]) (6, [p1,p2]) (7, [p1,p2]) (8, [p1,p3]) (9, [p1,p3])
(10, [p2,p3]) (11, [p1,p2,p3]) (12, [p1,p2,p3]) (13, [p1,p2,p3])
(a, [[0,2,3,4,7,9,10,13], [1,6,8,12], [5,11]])
(b, [[0,1,2,3,4,5,6,7,8,9,10,11,12,13]])
(c, [[0,1,2,3,4,5,6,7,8,9,10,11,12,13]])
==> [8]
[0,1,2,3,4,5,6,7,8,9,10]
(0, []) (1, [p1]) (2, [p2]) (3, [p3]) (4, [p1,p2])
(5, [p1,p2]) (6, [p1,p3]) (7, [p2,p3]) (8, [p1,p2,p3]) (9, [p1,p2,p3])
(10, [p1,p2,p3])
(a, [[0,1,2,3,5,6,7,10], [4,9], [8]])
(b, [[0,1,2,3,4,5,6,7,8,9,10]])
(c, [[0,1,2,3,4,5,6,7,8,9,10]])
==> [7]
[0,1,2,3,4,5,6,7,8]
(0, []) (1, [p1]) (2, [p2]) (3, [p3]) (4, [p1,p2])
(5, [p1,p3]) (6, [p2,p3]) (7, [p1,p2,p3]) (8, [p1,p2,p3])
(a, [[0,1,2,3,4,5,6,8], [7]])
(b, [[0,1,2,3,4,5,6,7,8]])
(c, [[0,1,2,3,4,5,6,7,8]])
==> [7]
[0,1,2,3,4,5,6,7,8]
(0, []) (1, [p1]) (2, [p2]) (3, [p3]) (4, [p1,p2])
(5, [p1,p3]) (6, [p2,p3]) (7, [p1,p2,p3]) (8, [p1,p2,p3])
(a, [[0,1,2,3,4,5,6,8], [7]])
(b, [[0,1,2,3,4,5,6,7,8]])
(c, [[0,1,2,3,4,5,6,7,8]])

```

Non-deterministic sum \oplus of multiple-pointed action models:

```

ndSum' :: PoAM -> PoAM -> PoAM
ndSum' m1 m2 = (Pmod states val acc ss)
  where
    (Pmod states1 val1 acc1 ss1) = convPmod m1
    (Pmod states2 val2 acc2 ss2) = convPmod m2
    f = \ x -> toInteger (length states1) + x
    states2' = map f states2
    val2'    = map (\ (x,y) -> (f x, y)) val2
    acc2'    = map (\ (x,y,z) -> (x, f y, f z)) acc2
    ss       = ss1 ++ map f ss2
    states   = states1 ++ states2'
    val      = val1 ++ val2'
    acc      = acc1 ++ acc2'

```

Example action models:

```

am0 = ndSum' (test p) (test (Neg p))

am1 = ndSum' (test p) (ndSum' (test q) (test r))

```

Examples of minimization for action emulation:

```

EMO> showM am0
==> [0,2]
[0,1,2,3]
(0,p)(1,T)(2,-p)(3,T)
(a,[(0],[1]),([2],[3]))
(b,[(0],[1]),([2],[3]))
(c,[(0],[1]),([2],[3]))

```

```

DEMO> showM (aePmod am0)
==> [0]
[0]
(0,T)
(a,[[0]])
(b,[[0]])
(c,[[0]])

```

```

DEMO> showM am1
==> [0,2,4]
[0,1,2,3,4,5]
(0,p)(1,T)(2,q)(3,T)(4,r)
(5,T)
(a,[(0],[1]),([2],[3]),([4],[5]))
(b,[(0],[1]),([2],[3]),([4],[5]))
(c,[(0],[1]),([2],[3]),([4],[5]))

DEMO> showM (aePmod am1)
==> [0]
[0,1]
(0,v[p,&[-p,q],&[-p,-q,r]])(1,T)
(a,[(0],[1]))
(b,[(0],[1]))
(c,[(0],[1]))

```

Non-deterministic sum \oplus of multiple-pointed action models, reduced for action emulation:

```

ndSum :: PoAM -> PoAM -> PoAM
ndSum m1 m2 = aePmod (ndSum' m1 m2)

```

Notice the difference with the definition of alternative composition of Kripke models for processes given in [25, Ch 4].

The **zero** action model is the 0 for the \oplus operation, so it can be used as the base case in the following list version of the \oplus operation:

```

ndS :: [PoAM] -> PoAM
ndS = foldl ndSum zero

```

Performing a test whether φ and announcing the result:

```

testAnnounce :: Form -> PoAM
testAnnounce form = ndS [ cmp [ test form, public form ],
                          cmp [ test (negation form),
                                public (negation form)] ]

```

testAnnounce form is equivalent to info all_agents form:

```

DEMO> showM (testAnnounce p)
==> [0,1]
[0,1]
(0,p)(1,-p)
(a,[[0],[1]])
(b,[[0],[1]])
(c,[[0],[1]])

DEMO> showM (info all_agents p)
==> [0,1]
[0,1]
(0,p)(1,-p)
(a,[[0],[1]])
(b,[[0],[1]])
(c,[[0],[1]])

```

The function `testAnnounce` gives the special case of revelations where the alternatives are a formula and its negation, and where the result is publicly announced.

Note that *DEMO* correctly computes the result of the sequence and the sum of two contradictory propositional tests:

```

DEMO> showM (cmp [test p, test (Neg p)])
==> []
[]

(a,[])
(b,[])
(c,[])

DEMO> showM (ndS [test p, test (Neg p)])
==> [0]
[0]
(0,T)
(a,[[0]])
(b,[[0]])
(c,[[0]])

```

Chapter 9

Main Module

9.1 Module Declaration

```
module DEMO
(
  module List,
  module Char,
  module Models,
  module Display,
  module MinBis,
  module ActEpist,
  module MinAE,
  module DPLL,
  module Semantics
)
where

import List
import Char
import Models
import Display
import MinBis
import ActEpist
import MinAE
import DPLL
import Semantics
```


9.2 Version

The first version of *DEMO* was written in March 2004. This first version was extended in May 2004 with an implementation of automata and a translation function from epistemic update logic to Automata PDL. In September 2004, I discovered a direct reduction of epistemic update logic to PDL [16]. This motivated a switch to a PDL-like language, with extra modalities for action update and automata update. I decided to leave in the automata for the time being, for nostalgic reasons.

Also in September 2004, Ji Ruan and I found a simple definition for action emulation and an algorithm for computing emulation-minimal action models [18, 38], so reduction under action emulation is also implemented in this October version of DEMO.

In Summer 2005, several example modules with DEMO programs for epistemic puzzles (some of them contributed by Ji Ruan) and for checking of security protocols (with contributions by Simona Orzan) were added, and the program was rewritten in a modular fashion.

```
version :: String
version = "DEMO 1.03, Summer 2005"
```

Chapter 10

Examples

10.1 The Riddle of the Caps

Picture a situation¹ of four people a, b, c, d standing in line, with a, b, c looking to the left, and d looking to the right. a can see no-one else; b can see a ; c can see a and b , and d can see no-one else. They are all wearing caps, and they cannot see their own cap. If it is common knowledge that there are two white and two black caps, then in the following situation c knows what colour cap she is wearing.



If c now announces that she knows the colour of her cap (without revealing the colour), b can infer from this that he is wearing a white cap, for b can reason as follows: “ c knows her colour, so she must see two caps of the same colour. The cap I can see is white, so my own cap must be white as well.” In this situation b draws a conclusion from the fact that c knows her colour.

In the following situation b can draw a conclusion from the fact that c does not know her colour.



In this case c announces that she does not know her colour, and b can infer from this that he is wearing a black cap, for b can reason as follows: “ c does not know her colour, so she must see two caps of different colours in front of her. The cap I can see is white, so my own cap must be black.”

To account for this kind of reasoning, we use model checking for epistemic updating, as follows. Proposition p_i expresses the fact that the i -th cap, counting from the left, is white. Thus, the facts of our first example situation are given by $p_1 \wedge p_2 \wedge \neg p_3 \wedge \neg p_4$, and those of our second example by $p_1 \wedge \neg p_2 \wedge \neg p_3 \wedge p_4$.

Here is the DEMO code for this example (details to be explained below):

¹See [17].

```

module Caps
where

import DEMO

-- in Models, set last_agent = d.

capsInfo :: Form
capsInfo = Disj [Conj [f, g, Neg h, Neg j] |
                  f <- [p1, p2, p3, p4],
                  g <- [p1, p2, p3, p4] \\ [f],
                  h <- [p1, p2, p3, p4] \\ [f,g],
                  j <- [p1, p2, p3, p4] \\ [f,g,h],
                  f < g, h < j
                ]

awarenessFirstCap = info [b,c] p1
awarenessSecondCap = info [c] p2

cK = Disj [K c p3, K c (Neg p3)]
bK = Disj [K b p2, K b (Neg p2)]

mo0 = upd (initE [P 1, P 2, P 3, P 4]) (test capsInfo)
mo1 = upd mo0 (public capsInfo)
mo2 = upds mo1 [awarenessFirstCap, awarenessSecondCap]
mo3a = upd mo2 (public cK)
mo3b = upd mo2 (public (Neg cK))

```

An initial situation with four agents a, b, c, d and four propositions p_1, p_2, p_3, p_4 , with exactly two of these true, where no-one knows anything about the truth of the propositions, and everyone is aware of the ignorance of the others, is modelled like this:

```

Caps> showM mo0
==> [5,6,7,8,9,10]
[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15]
(0, []) (1, [p1]) (2, [p2]) (3, [p3]) (4, [p4])
(5, [p1,p2]) (6, [p1,p3]) (7, [p1,p4]) (8, [p2,p3]) (9, [p2,p4])
(10, [p3,p4]) (11, [p1,p2,p3]) (12, [p1,p2,p4]) (13, [p1,p3,p4]) (14, [p2,p3,p4])
(15, [p1,p2,p3,p4])
(a, [[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15]])
(b, [[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15]])
(c, [[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15]])
(d, [[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15]])

```

The first line indicates that worlds 5,6,7,8,9,10 are compatible with the facts of the matter (the facts being that there are two white and two black caps). E.g., 5 is the world where a and b are wearing the white caps. The second line lists all the possible worlds; there are 2^4 of them, since every world has a different valuation. The third through sixth lines give the valuations of worlds. The last four lines represent the accessibility relations for the agents. All accessibilities are total relations, and they are represented here as the corresponding partitions on the set of worlds. Thus, the ignorance of the agents is reflected in the fact that for all of them all worlds are equivalent: none of the agents can tell any of them apart.

The information that two of the caps are white and two are black is expressed by the formula

$$(p_1 \wedge p_2 \wedge \neg p_3 \wedge \neg p_4) \vee (p_1 \wedge p_3 \wedge \neg p_2 \wedge \neg p_4) \vee (p_1 \wedge p_4 \wedge \neg p_2 \wedge \neg p_3) \\ \vee (p_2 \wedge p_3 \wedge \neg p_1 \wedge \neg p_4) \vee (p_2 \wedge p_4 \wedge \neg p_1 \wedge \neg p_3) \vee (p_3 \wedge p_4 \wedge \neg p_1 \wedge \neg p_2).$$

A public announcement with this information has the following effect:

```
Caps> showM (upd mo0 (public capsInfo))
==> [0,1,2,3,4,5]
[0,1,2,3,4,5]
(0, [p1,p2]) (1, [p1,p3]) (2, [p1,p4]) (3, [p2,p3]) (4, [p2,p4])
(5, [p3,p4])
(a, [[0,1,2,3,4,5]])
(b, [[0,1,2,3,4,5]])
(c, [[0,1,2,3,4,5]])
(d, [[0,1,2,3,4,5]])
```

Let this model be called `mo1`. The representation above gives the partitions for all the agents, showing that nobody knows anything. A perhaps more familiar representation for this multi-agent Kripke model is given in Figure 10.1. In this picture, all worlds are connected for all agents, all worlds are compatible with the facts of the matter (indicated by the double ovals).

Next, we model the fact that (everyone is aware that) b can see the first cap and that c can see the first and the second cap, as follows:

```
Caps> showM (upds mo1 [info [b,c] p1, info [c] p2])
==> [0,1,2,3,4,5]
[0,1,2,3,4,5]
(0, [p1,p2]) (1, [p1,p3]) (2, [p1,p4]) (3, [p2,p3]) (4, [p2,p4])
(5, [p3,p4])
(a, [[0,1,2,3,4,5]])
(b, [[0,1,2], [3,4,5]])
(c, [[0], [1,2], [3,4], [5]])
(d, [[0,1,2,3,4,5]])
```

Notice that this model reveals that in case a, b wear caps of the same colour (situations 0 and 5), c knows the colour of all the caps, and in case a, b wear caps of different colours, she does not (she confuses the cases 1,2 and the cases 3,4). Figure 10.2 gives a picture representation.




Figure 10.1: Caps situation where nobody knows anything about p_1, p_2, p_3, p_4 .

Let this model be called `mo2`. Knowledge of c about her situation is expressed by the epistemic formula $K_cp_3 \vee K_c\neg p_3$, ignorance of c about her situation by the negation of this formula. Knowledge of b about his situation is expressed by $K_bp_2 \vee K_b\neg p_2$. Let `bK`, `cK` express that b, c know about their situation. Then updating with public announcement of `cK` and with public announcement of the negation of this have different effects:

```
Caps> showM (upd mo2 (public cK))
==> [0,1]
[0,1]
(0, [p1,p2]) (1, [p3,p4])
(a, [[0,1]])
(b, [[0], [1]])
(c, [[0], [1]])
(d, [[0,1]])

Caps> showM (upd mo2 (public (Neg cK)))
==> [0,1,2,3]
[0,1,2,3]
(0, [p1,p3]) (1, [p1,p4]) (2, [p2,p3]) (3, [p2,p4])
(a, [[0,1,2,3]])
(b, [[0,1], [2,3]])
(c, [[0,1], [2,3]])
(d, [[0,1,2,3]])
```




Figure 10.2: Caps situation after updating with awareness of what b and c can see.

In both results, b knows about his situation, though:

```
Caps> isTrue (upd mo2 (public cK)) bK
True
Caps> isTrue (upd mo2 (public (Neg cK))) bK
True
```

10.2 Muddy Children

For this example we need four agents a, b, c, d . Four children a, b, c, d are sitting in a circle. They have been playing outside, and they may or may not have mud on their foreheads. Their father announces: “At least one child is muddy!” Suppose in the actual situation, both c and d are muddy.

a	b	c	d
○	○	●	●

Then at first, nobody knows whether he is muddy or not. After public announcement of these facts, $c(d)$ can reason as follows. “Suppose I am clean. Then $d(c)$ would have known in the first round that she was dirty. But she didn’t. So I am muddy.” After c, d announce that they know their state, $a(b)$ can reason as follows: “Suppose I am dirty. Then c and d would not have known in the second round that they were dirty. But they knew. So I am clean.” Note that the reasoning involves awareness about *perception*.

In the actual situation where b, c, d are dirty, we get:

a	b	c	d
○	●	●	●
?	?	?	?
?	?	?	?
?	!	!	!
!	!	!	!

Reasoning of b : “Suppose I am clean. Then c and d would have known in the second round that they are dirty. But they didn’t know. So I am dirty. Similarly for c and d .” Reasoning of a : “Suppose I am dirty. Then b, c and d would not have known their situation in the third round. But they did know. So I am clean.” And so on ... [19].

Here is the DEMO implementation of the second case of this example, with b, c, d dirty.

```

module Muddy
where

import DEMO

-- in Models, set last_agent = d.

bcd_dirty = Conj [Neg p1, p2, p3, p4]

awareness = [info [b,c,d] p1,
              info [a,c,d] p2,
              info [a,b,d] p3,
              info [a,b,c] p4 ]

aK = Disj [K a p1, K a (Neg p1)]
bK = Disj [K b p2, K b (Neg p2)]
cK = Disj [K c p3, K c (Neg p3)]
dK = Disj [K d p4, K d (Neg p4)]

mu0 = upd (initE [P 1, P 2, P 3, P 4]) (test bcd_dirty)
mu1 = upds mu0 awareness
mu2 = upd mu1 (public (Disj [p1, p2, p3, p4]))
mu3 = upd mu2 (public (Conj[Neg aK, Neg bK, Neg cK, Neg dK]))
mu4 = upd mu3 (public (Conj[Neg aK, Neg bK, Neg cK, Neg dK]))
mu5 = upds mu4 [public (Conj[bK, cK, dK])]

```

The initial situation, where nobody knows anything, and they are all aware of the common ignorance (say, all children have their eyes closed, and they all know this) looks like this:

```

Muddy> showM mu0
==> [14]
[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15]
(0,[]) (1,[p1]) (2,[p2]) (3,[p3]) (4,[p4])
(5,[p1,p2]) (6,[p1,p3]) (7,[p1,p4]) (8,[p2,p3]) (9,[p2,p4])
(10,[p3,p4]) (11,[p1,p2,p3]) (12,[p1,p2,p4]) (13,[p1,p3,p4]) (14,[p2,p3,p4])
(15,[p1,p2,p3,p4])
(a,[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15])
(b,[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15])
(c,[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15])
(d,[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15])

```

The awareness of the children about the mud on the foreheads of the others is expressed in terms of update models. Here is the update model that expresses that *b, c, d* can see whether *a* is muddy or not:


```

Muddy> showM (info [b,c,d] p1)
==> [0,1]
[0,1]
(0,p1)(1,-p1)
(a,[0,1])
(b,[0],[1])
(c,[0],[1])
(d,[0],[1])

```

Let **awareness** be the list of update models expressing what happens when they all open their eyes and see the foreheads of the others. Then updating with this has the following result:

```

Muddy> showM (upds mu0 awareness)
==> [14]
[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15]
(0,[]) (1,[p1]) (2,[p2]) (3,[p3]) (4,[p4])
(5,[p1,p2]) (6,[p1,p3]) (7,[p1,p4]) (8,[p2,p3]) (9,[p2,p4])
(10,[p3,p4]) (11,[p1,p2,p3]) (12,[p1,p2,p4]) (13,[p1,p3,p4]) (14,[p2,p3,p4])
(15,[p1,p2,p3,p4])
(a,[0,1],[2,5],[3,6],[4,7],[8,11],[9,12],[10,13],[14,15])
(b,[0,2],[1,5],[3,8],[4,9],[6,11],[7,12],[10,14],[13,15])
(c,[0,3],[1,6],[2,8],[4,10],[5,11],[7,13],[9,14],[12,15])
(d,[0,4],[1,7],[2,9],[3,10],[5,12],[6,13],[8,14],[11,15])

```

Call the result **mu1**. An update of **mu1** with the public announcement that at least one child is muddy gives:


```

Muddy> showM (upd mu1 (public (Disj [p1, p2, p3, p4])))
==> [13]
[0,1,2,3,4,5,6,7,8,9,10,11,12,13,14]
(0,[p1]) (1,[p2]) (2,[p3]) (3,[p4]) (4,[p1,p2])
(5,[p1,p3]) (6,[p1,p4]) (7,[p2,p3]) (8,[p2,p4]) (9,[p3,p4])
(10,[p1,p2,p3]) (11,[p1,p2,p4]) (12,[p1,p3,p4]) (13,[p2,p3,p4]) (14,[p1,p2,p3,p4])

(a,[0],[1,4],[2,5],[3,6],[7,10],[8,11],[9,12],[13,14])
(b,[0,4],[1],[2,7],[3,8],[5,10],[6,11],[9,13],[12,14])
(c,[0,5],[1,7],[2],[3,9],[4,10],[6,12],[8,13],[11,14])
(d,[0,6],[1,8],[2,9],[3],[4,11],[5,12],[7,13],[10,14])

```


Picture representation (the double oval indicates the actual world):



Call this model μ_2 , and use aK , bK , cK , dK for the formulas expressing that a, b, c, d know whether they are muddy (see the code above). Then we get:

```
Muddy> showM (upd mu2 (public (Conj[Neg aK, Neg bK, Neg cK, Neg dK])))
==> [9]
[0,1,2,3,4,5,6,7,8,9,10]
(0, [p1,p2]) (1, [p1,p3]) (2, [p1,p4]) (3, [p2,p3]) (4, [p2,p4])
(5, [p3,p4]) (6, [p1,p2,p3]) (7, [p1,p2,p4]) (8, [p1,p3,p4]) (9, [p2,p3,p4])
(10, [p1,p2,p3,p4])
(a, [[0], [1], [2], [3,6], [4,7], [5,8], [9,10]])
(b, [[0], [1,6], [2,7], [3], [4], [5,9], [8,10]])
(c, [[0,6], [1], [2,8], [3], [4,9], [5], [7,10]])
(d, [[0,7], [1,8], [2], [3,9], [4], [5], [6,10]])
```

Picture representation:



Call this model μ_3 , and update again with the same public announcement of general ignorance:

```
Muddy> showM (upd mu3 (public (Conj[Neg aK, Neg bK, Neg cK, Neg dK])))
==> [3]
[0,1,2,3,4]
```

$(0, [p_1, p_2, p_3]) (1, [p_1, p_2, p_4]) (2, [p_1, p_3, p_4]) (3, [p_2, p_3, p_4]) (4, [p_1, p_2, p_3, p_4])$


$(a, [[0], [1], [2], [3, 4]])$

$(b, [[0], [1], [2, 4], [3]])$

$(c, [[0], [1, 4], [2], [3]])$

$(d, [[0, 4], [1], [2], [3]])$

Picture representation:



Call this model μ_4 . In this model, b, c, d know about their situation:

```
Muddy> isTrue mu4 (Conj [bK, cK, dK])
```

```
True
```

Updating with the public announcement of this information determines everything:

```
Muddy> showM (upd mu4 (public (Conj[bK, cK, dK])))
```

```
==> [0]
```

```
[0]
```

```
(0, [p2, p3, p4])
```

```
(a, [[0]])
```

```
(b, [[0]])
```

```
(c, [[0]])
```

```
(d, [[0]])
```

10.3 The Riddle of Sum and Product

The following Sum and Product riddle was stated by the Dutch mathematician Hans Freudenthal in a Dutch mathematics journal in 1969:

A says to S and P: I have chosen two integers x, y such that $1 < x < y$ and $x + y \leq 100$. In a moment, I will inform S only of $s = x + y$, and P only of $p = xy$. These announcements remain private. You are required to determine the pair (x, y) . He acts as said. The following conversation now takes place:

1. P says: "I do not know the pair."

2. S says: “I knew you didn’t.”
3. P says: “I now know it.”
4. S says: “I now also know it.”

Determine the pair (x, y) .

This was solved by combinatorial means in a later issue of the journal. A model checking solution with DEMO (based on a DEMO program written by Ji Ruan) was presented in [13].

```
module SumProduct
where

import DEMO

-- Program written by Ji Ruan

-- Make sure to set last_agent = b in Models
```

The list of candidate pairs:

```
pairs :: [(Int,Int)]
pairs = [ (x,y) | x <- [2..100], y <- [2..100], x < y, x+y <= 100 ]
```

The number of pairs gives the number of different possible worlds that we are going to need:

```
numpairs = toInteger (length pairs)
```

By indexing the pairs we get a mapping from possible worlds to pairs:

```
indexed_pairs = zip [0..numpairs-1] pairs
```

The initial epistemic model is such that a (representing S) cannot distinguish number pairs with the same sum, and b (representing P) cannot distinguish number pairs with the same product. In the valuation, we let p_x, q_y represent the pair (x, y) .

```

msnp :: EpistM
msnp = (Pmod [0..numpairs-1] val acc [0..numpairs-1])
  where
    val  = [ (w,[P x, Q y]) | (w,(x,y)) <- indexed_pairs]
    acc  = [ (a,w,v) | (w,(x1,y1)) <- indexed_pairs,
                    (v,(x2,y2)) <- indexed_pairs,
                    x1+y1 == x2+y2 ]
          ++
          [ (b,w,v) | (w,(x1,y1)) <- indexed_pairs,
                    (v,(x2,y2)) <- indexed_pairs,
                    x1*y1 == x2*y2 ]

```

Identifying a pair in a particular state is done by the following formula:

```

pform :: Int -> Int -> Form
pform x y = Conj [Prop (P x), Prop (Q y)]

```

The statement by b that he does not know the pair:

```

statement_1 =
  Conj [ Neg (K b (pform x y)) | (x,y) <- pairs ]

```

To check this statement is expensive. A computationally cheaper equivalent statement is the following (see [13]).

```

statement_1e =
  Conj [ pform x y 'impl' Neg (K b (pform x y)) | (x,y) <- pairs ]

```

In Freudenthal's story, the first public announcement is the statement where b confesses his ignorance, and the second public announcement is the statement by a about her knowledge about b 's state of knowledge *before* that confession. We can wrap the two together in a single public announcement to the effect that initially, a knows that b does not know the pair. This gives:

```

announcement_1 = public (K a statement_1e)

```

The second public announcement proclaims the statement by b that now he knows:

```
statement_2 =  
  Disj [ K b (pform x y) | (x,y) <- pairs ]
```

Equivalently, but computationally more efficient:

```
statement_2e =  
  Conj [ pform x y 'impl' K b (pform x y) | (x,y) <- pairs ]
```

This gives:

```
announcement_2 = public statement_2e
```

The final public announcement concerns the statement by a that now she knows as well. In the computationally optimized version:

```
statement_3e =  
  Conj [ pform x y 'impl' K a (pform x y) | (x,y) <- pairs ]
```

This gives:

```
announcement_3 = public statement_3e
```

The solution:

```
solution = showM (upds msnp [announcement_1,announcement_2,announcement_3])
```

This works (watch the result after typing in the command, and a good night's sleep):

```
SumProduct> solution
==> [0]
[0]
(0,[p4,q13])
(a,[[0]])
(b,[[0]])
```

10.4 Sums and Sums-of-Squares

A variation of the Sum and Product riddle goes as follows.²

Someone says to *A* and *B*: I have chosen two integers x, y such that $1 \leq x \leq y \leq 20$, and I will inform *A* only of $x^2 + y^2$ and *B* only of $x + y$. He acts as said. The following conversation takes place:

1. A says: “I do not know the pair.”
2. B says: “I do not know the pair.”
3. A says: “I do not know the pair.”
4. B says: “I do not know the pair.”
5. A says: “I do not know the pair.”
6. B says: “I do not know the pair.”
7. A says: “Now I know the pair.”

What are x and y ?

```
module SumSumSquares
where

import DEMO

-- Make sure to set last_agent = b in Models
```

The list of candidate pairs:

```
pairs :: [(Int,Int)]
pairs = [ (x,y) | x <- [1..20], y <- [1..20], x <= y ]
```

²With thanks to Karst Koymans.

The number of pairs gives the number of different possible worlds that we are going to need:

```
numpairs = toInteger (length pairs)
```

By indexing the pairs we get a mapping from possible worlds to pairs:

```
indexed_pairs = zip [0..numpairs-1] pairs
```

The initial epistemic model is such that a cannot distinguish number pairs with the same sum of squares, and b cannot distinguish number pairs with the same sum. In the valuation, we let p_x, q_y represent the pair (x, y) .

```
msss :: EpistM
msss = (Pmod [0..numpairs-1] val acc [0..numpairs-1])
  where
    val = [ (w,[P x, Q y]) | (w,(x,y)) <- indexed_pairs]
    acc = [ (a,w,v) | (w,(x1,y1)) <- indexed_pairs,
                      (v,(x2,y2)) <- indexed_pairs,
                      x1^2+y1^2 == x2^2+y2^2 ]
      ++
      [ (b,w,v) | (w,(x1,y1)) <- indexed_pairs,
                  (v,(x2,y2)) <- indexed_pairs,
                  x1+y1 == x2+y2 ]
```

Identifying a pair in a particular state is done by the following formula:

```
pform :: Int -> Int -> Form
pform x y = Conj [Prop (P x), Prop (Q y)]
```

The statements by a and b that they do not know the pair:


```

not_knows_a =
  Conj [ pform x y 'impl' Neg (K a (pform x y)) | (x,y) <- pairs ]
not_knows_b =
  Conj [ pform x y 'impl' Neg (K b (pform x y)) | (x,y) <- pairs ]

```

The statement by a that she knows the pair:

```

knows_a =
  Conj [ pform x y 'impl' K a (pform x y) | (x,y) <- pairs ]

```

The solution:

```

solution = showM (upds msss [public not_knows_a,
                             public not_knows_b,
                             public not_knows_a,
                             public not_knows_b,
                             public not_knows_a,
                             public not_knows_b,
                             public knows_a ])

```

This yields the solution $x = 8, y = 9$.

Inspecting the computations, we can come up with variations of the puzzle:

Variation 1:

1. A says: "I do not know the pair."
2. B says: "I do not know the pair."
3. A says: "I do not know the pair."
4. B says: "I do not know the pair."
5. A says: "Now I know the pair."

What are x and y ?

```
variation1 = showM (upds msss [public not_knows_a,
                               public not_knows_b,
                               public not_knows_a,
                               public not_knows_b,
                               public knows_a ])
```

The outcome is now $x = 6, y = 7$.

Variation 2:

1. A says: “I do not know the pair.”
2. B says: “I do not know the pair.”
3. A says: “I do not know the pair.”
4. B says: “I do not know the pair.”
5. A says: “I do not know the pair.”
6. B says: “Now I know the pair.”

What are x and y ?

The statement by b that he knows the pair:

```
knows_b =
  Conj [ pform x y 'impl' K b (pform x y) | (x,y) <- pairs ]
```

```
variation2 = showM (upds msss [public not_knows_a,
                               public not_knows_b,
                               public not_knows_a,
                               public not_knows_b,
                               public not_knows_a,
                               public knows_b ])
```

The outcome is now $x = 1, y = 12$.

10.5 Card Showing

```

module Cards
where

import DEMO

-- in Models, set last_agent = c.

```

A simple card showing situation goes as follows.³ Alice, Bob and Carol each hold one of cards Purple, Qaki (Khaki), Red. The actual deal is: Alice holds Purple, Bob holds Qaki, Carol holds Red. The actual action is: Alice shows Purple to Bob with Carol looking on.

The initial state of the game:

```

cards0 :: EpistM
cards0 = (Pmod [0..5] val acc [0])
  where
    val    = [(0,[P 1,Q 2,R 3]),(1,[P 1,R 2,Q 3]),
              (2,[Q 1,P 2,R 3]),(3,[Q 1,R 2,P 3]),
              (4,[R 1,P 2,Q 3]),(5,[R 1,Q 2,P 3])]
    acc    = [(a,0,0),(a,0,1),(a,1,0),(a,1,1),
              (a,2,2),(a,2,3),(a,3,2),(a,3,3),
              (a,4,4),(a,4,5),(a,5,4),(a,5,5),
              (b,0,0),(b,0,5),(b,5,0),(b,5,5),
              (b,2,2),(b,2,4),(b,4,2),(b,4,4),
              (b,3,3),(b,3,1),(b,1,3),(b,1,1),
              (c,0,0),(c,0,2),(c,2,0),(c,2,2),
              (c,3,3),(c,3,5),(c,5,3),(c,5,5),
              (c,4,4),(c,4,1),(c,1,4),(c,1,1)]

```

Here it is displayed:

```


Cards> showM cards0
==> 0
[0,1,2,3,4,5]
(0,[p1,q2,r3])(1,[p1,r2,q3])(2,[q1,p2,r3])(3,[q1,r2,p3])(4,[r1,p2,q3])
(5,[r1,q2,p3])
(a,[0,1],[2,3],[4,5])

```

³With thanks to Hans van Ditmarsch.

$(b, [[0,5], [1,3], [2,4]])$
 $(c, [[0,2], [1,4], [3,5]])$

Viewed as a graph:



Action: a shows p to b with c looking on (c sees that a card is shown, but does not see that it is p):

```

showABp :: PoAM
showABp = (Pmod [0,1] pre susp [0])
  where
    pre = [(0,p1),(1,q1)]
    susp = [(a,0,0),(a,1,1),
            (b,0,0),(b,1,1),
            (c,0,0),(c,0,1),
            (c,1,0),(c,1,1)]

```

This gives:


Cards> showM showABp

```

==> [0]
[0,1]
(0,p1)(1,q1)
(a,[[0],[1]])
(b,[[0],[1]])
(c,[[0,1]])

```

As a graph:




The result of updating with this:

```

Cards> showM (upd cards0 showABp)
==> [0]
[0,1,2,3]
(0,[p1,q2,r3])(1,[p1,r2,q3])(2,[q1,p2,r3])(3,[q1,r2,p3])
(a,[[0,1],[2,3]])
(b,[[0],[1],[2],[3]])
(c,[[0,2],[1],[3]])

```

Viewed as a graph:




This modelling of the show action is still rather crude, for it is hard-coded in the action model that p_1 holds in the actual world. We can do better if we use a multiple pointed reveal action model.

```

revealAB = reveal [b] [p1,q1,r1]

```

Viewed as a graph:




The result of updating with this:

```
result = upd cards0 revealAB
```

```

Cards> showM result
==> [0]
[0,1,2,3,4,5]
(0,[p1,q2,r3])(1,[p1,r2,q3])(2,[q1,p2,r3])(3,[q1,r2,p3])(4,[r1,p2,q3])
(5,[r1,q2,p3])
(a,[[0,1],[2,3],[4,5]])
(b,[[0],[1],[2],[3],[4],[5]])
(c,[[0,2],[1,4],[3,5]])
  
```

Viewed as a graph:



We have enough machinery now to handle quite subtle actions. Suppose the three cards are dealt to a, b, c but are still undisclosed on the table. We now want to find an action model for the action of a inspecting her own card, with the others looking on. Here it is:

```

Cards> showM (reveal [a] [p1,q1,r1])
==> [0,1,2]
[0,1,2]
(0,p1)(1,q1)(2,r1)
(a,[[0],[1],[2]])
(b,[[0,1,2]])
(c,[[0,1,2]])

```

And here is the action of a picking up her card and showing it to the others, without taking a look herself:

```

Cards> showM (reveal [b,c] [p1,q1,r1])
==> [0,1,2]
[0,1,2]
(0,p1)(1,q1)(2,r1)
(a,[[0,1,2]])
(b,[[0],[1],[2]])
(c,[[0],[1],[2]])

```


10.6 Elements of Secure Communication

An important element of secure communication is the ability to pass information from a to b along an insecure channel in such a way that an eavesdropper c cannot find out what b has learnt. A particular case of that is the Russian Card Problem [12, 14]. See Section 10.8. Security protocols are described in [33, 36]; [37] makes an attempt at model checking for analysing the security of communication. Applications of epistemic logic, but with rather ad-hoc updates, to the analysis of security protocols can be found in [26].

Here we will merely take an abstract look at the problem⁴. If agent a and b have a link between propositions p and q and a and b are the only ones with this link, then a can inform b in secret about q by means of a public communication about p . Here is how.

```

module SecCom
where

import DEMO

-- in Models, set last_agent = c.

link_ab_pq = ndSum (groupM [a,b] (equiv p q))
                  (groupM [a,b] (equiv p (Neg q)))
c_b_pq = CK [a,b] (Disj [K b (p 'equiv' q), K b (p 'equiv' (Neg q))])
ck_ab_link = Disj [CK [a,b] (p 'equiv' q), CK [a,b] (p 'equiv' (Neg q))]
k_tr      = K a (Neg (K b p)) 'impl' Neg (K b p)

aKq = Disj [K a q, K a (Neg q)]
bKq = Disj [K b q, K b (Neg q)]
cKq = Disj [K c q, K c (Neg q)]

mo0 = initE [P 0, Q 0]
mo1 = upds mo0 [info [a] p, link_ab_pq]
mo2 = upd mo1 (public p)

```

Assume there are three agents a, b, c . A situation where all agents are ignorant about p and q , and are aware of their common ignorance looks like this:

```

SecCom> showM mo0
==> [0,1,2,3]
[0,1,2,3]
(0, []) (1, [p]) (2, [q]) (3, [p,q])
(a, [[0,1,2,3]])

```

⁴See [17].

```
(b, [[0,1,2,3]])
(c, [[0,1,2,3]])
```

Suppose a has information about p , and a and b either have common knowledge that p and q are equivalent or they have common knowledge that p and $\neg q$ are:

```
SecCom> showM (upds mo0 [info [a] p, link_ab_pq])
==> [0,5,8,9]
[0,1,2,3,4,5,6,7,8,9,10,11]
(0, []) (1, []) (2, []) (3, [p]) (4, [p])
(5, [p]) (6, [q]) (7, [q]) (8, [q]) (9, [p,q])
(10, [p,q]) (11, [p,q])
(a, [[0], [1,6], [2,7], [3,10], [4,11], [5], [8], [9]])
(b, [[0,9], [1,3,6,10], [2,4,7,11], [5,8]])
(c, [[0,1,3,6,9,10], [2,4,5,7,8,11]])
```




Figure 10.3: Situation where a knows whether p and where a, b have common knowledge about a link between p and q .

Call this model `mo1` (see Figure 10.3 for an alternative representation). In this model, c still knows nothing about q :

```
SecCom> isTrue mo1 (Disj [K c q, K c (Neg q)])
```

False

Also, in this model it is common knowledge among a, b that b knows about the link between p and q . Thus, the result of public announcement of p in this situation is that b knows whether q , while c is still left in the dark about q :

```
SecCom> showM (upd mo1 (public p))
==> [2,3]
[0,1,2,3,4,5]
(0,[p])(1,[p])(2,[p])(3,[p,q])(4,[p,q])
(5,[p,q])
(a,[[0,4],[1,5],[2],[3]])
(b,[[0,4],[1,5],[2],[3]])
(c,[[0,3,4],[1,2,5]])

SecCom> isTrue (upd mo1 (public p)) (Disj [K b q, K b (Neg q)])
True
SecCom> isTrue (upd mo1 (public p)) (Disj [K c q, K c (Neg q)])
False
```

10.7 Public/secret key cryptography

Many security protocols assume a public/secret key infrastructure (PKI), meaning that each agent owns a pair of keys, a *public* one, available to everybody, and a *secret* one, only known to the agent herself. These keys are employed in encryption/decryption algorithms that match each other, i.e., a content encrypted with a public key can only be decrypted with the corresponding secret key, and the other way around. Therefore, when an agent a wants to send a message that only an agent b should be able to read, all she has to do is encrypt the message using b 's public key.

```
module PublSecr
where

import DEMO

-- set last_agent = c in Models
```

To see how this basic mechanism can be modeled in DEMO [17], we consider two agents a, b trying to communicate as described above, and an agent c , the eavesdropper, who intercepts all the messages. Let p_1, p_2 represent the secret keys of a and b , respectively, and let us fix their




Figure 10.4: The epistemic state of agents a, b, c before and after a has sent the message q encrypted with the public key of b .

truth values to $p_1 = \top$, $p_2 = \perp$. It is not convenient to model the corresponding public keys explicitly; instead, we model the encryption algorithms for the two parties as the implications $p_1 \Rightarrow x$ and $\neg p_2 \Rightarrow x$ (where x is the content to be encrypted). Finally, let q be the message that a wishes to send to b . The situation when a is the only one who knows p_1 and q , b the only one knowing p_2 and all agents are aware of who knows what is described in DEMO as follows:

```
ms1 = upds (initE [P 1, P 2, Q 0])
  [test (Conj [p1, (Neg p2), q]), info [b] p2, info [a] p1, info [a] q]
```

Sending the encrypted message is modeled by the following update step:

```
ms2 = upd ms1 (public ((Neg p2) 'impl' q))
```

$\neg p_2 \Rightarrow q$ is the result of applying b 's public encryption algorithm to the content q . The public communication expresses the fact that c eavesdrops. The generated visual representations of `ms1` and `ms2` are shown in Figure 10.4.

In `ms2`, it can be checked that b has learned q , while c hasn't:

```
PublSecr> isTrue ms2 (Disj [K b q, K b (Neg q)])
True
PublSecr> isTrue ms2 (Disj [K c q, K c (Neg q)])
False
```

It can also be verified in `ms2` that the encryption algorithm works as a blackbox, that is that a and b do not learn each other's secret key while encrypting/decrypting:

```
PublSecr> isTrue ms2 (Disj [K a p2, K a (Neg p2)])
False
PublSecr> isTrue ms2 (Disj [K b p1, K b (Neg p1)])
False
```

By replacing `info` with `secret` in the definition of `ms1`, it is possible to model the fact that a doesn't actually know whether b indeed has b 's secret key. This is a relevant subtlety in, for instance, authentication protocols.

10.8 The Russian Cards Problem

From a pack of seven known cards two players each draw three cards and a third player gets the remaining card. How can the players with three cards openly inform each other about their cards, without the third player learning from any of their cards who holds it?

This 'Russian Cards' problem originated at the Moscow Math Olympiad 2000. It is discussed in [12], and solved in DEMO in [14], with a DEMO program written by Ji Ruan. Here we first give a streamlined version of Ji Ruan's solution, and next a generalisation. In Ruan's solution, a particular deal is assumed to be the actual deal, and the protocol is hardwired to that deal. In the generalization, the protocol to be followed is computed from the model.

Call the players a, b, c . Let the cards be numbered 0 through 6. A possible deal is: a holds $\{0, 1, 2\}$, b holds $\{3, 4, 5\}$ and c holds 6. Abbreviate this as 012.345.6.

Let p_i express that card i is held by a , let q_i express that card i is held by b , let r_i express that card i is held by c . Then the conjunction $p_0 \wedge p_1 \wedge p_2 \wedge q_3 \wedge q_4 \wedge q_5 \wedge r_6$ is true of the deal 012.345.6.

```

module RussianCards
where

import DEMO

-- Contributed by Ji Ruan, revision and generalisation by Jan van Eijck

-- Make sure last_agent is set to c in Models

```

Represent the deals as sequences of integers of the form $((n_0, n_1, n_2), (n_3, n_4, n_5), n_6)$, taken from the set $\{0, \dots, 6\}$, where all the n_i are different, and where triples (n_0, n_1, n_2) and (n_3, n_4, n_5) are in ascending order.

```

deals = [((n0,n1,n2),(n3,n4,n5),n6) | n0 <- [0..6] :: [Int],
                                     n1 <- [0..6] \\ [n0],
                                     n2 <- [0..6] \\ [n0,n1],
                                     n3 <- [0..6] \\ [n0,n1,n2],
                                     n4 <- [0..6] \\ [n0,n1,n2,n3],
                                     n5 <- [0..6] \\ [n0,n1,n2,n3,n4],
                                     n6 <- [0..6] \\ [n0,n1,n2,n3,n4,n5],
                                     n0 < n1, n1 < n2,
                                     n3 < n4, n4 < n5 ]

```

Clearly, there are $\binom{7}{3}\binom{4}{3}\binom{1}{1} = 140$ deals, which provides a first check on the implementation:

```

RussianCards> length deals
140

```

For constructing the initial model, we index the deals, as follows:

```

indexed_deals = zip [0..139] deals

```

In the initial model, the indices 0..139 are the worlds, the valuations are given by the list of possible deals, and the accessibilities are such that the players can only distinguish deals that differ in their own hand. Treating the actual world as a parameter, this gives:

```

rus_init :: Integer -> EpistM
rus_init n = Pmod [0..139] val acc [n]
  where
    val = [ (i, [P n0, P n1, P n2, Q n3, Q n4, Q n5, R n6]) |
              (i, ((n0,n1,n2),(n3,n4,n5),n6)) <- indexed_deals ]
    acc = [ (a,w,v) | (w,(d,_,_)) <- indexed_deals,
                      (v,(e,_,_)) <- indexed_deals, d == e ]
      ++
      [ (b,w,v) | (w,(_,d,_)) <- indexed_deals,
                  (v,(_,e,_)) <- indexed_deals, d == e ]
      ++
      [ (c,w,v) | (w,(_,_,d)) <- indexed_deals,
                  (v,(_,_,e)) <- indexed_deals, d == e ]

```

Let `a_knows_bs` express that *a* knows *b*'s cards, let `b_knows_as` express that *b* knows *a*'s cards, and let `c_ignorant` express that *c* does not know any of *a*'s or *b*'s cards.

```

a_knows_bs =
  Conj[Disj[K a (Prop (Q i)), K a (Neg (Prop (Q i)))] | i<- [0..6] ]

b_knows_as =
  Conj[Disj[K b (Prop (P i)), K b (Neg(Prop (P i)))] | i <- [0..6] ]

c_ignorant =
  Conj[Conj[Neg (K c (Prop (P i))), Neg (K c (Prop (Q i)))]
      | i <- [0..6] ]

```

A correct solution to the Russian cards problem is a sequence of public announcements by *a* and *b* that has the effect that after each announcement it is publicly known that *c* is still ignorant about *a*'s and *b*'s cards. After execution of the entire protocol, it should be commonly known by *a* and *b* that (i) *a* knows that *b* knows *a*'s hand, and (ii) *b* knows that *a* knows *b*'s hand.

Now suppose that the deal is 012.345.6. Then *a* has hand 012, and one known solution to the Russian cards problem that we are going to check is where *a* announces that her hand is in {012, 034, 056, 135, 246}, followed by *b* announcing that *c* holds card 6.

Let `a_announce` express the public announcement that *a* knows that her hand is one of

{012, 034, 056, 135, 246}.

```
a_announce = public (K a (Disj [Conj [p0,p1,p2],
                                   Conj [p0,p3,p4],
                                   Conj [p0,p5,p6],
                                   Conj [p1,p3,p5],
                                   Conj [p2,p4,p6]]))
```

Let **b_announce** express the public announcement that *b* knows that *c* holds card 6.

```
b_announce = public (K b r6)
```

Since 0 is the world with deal 012.345.6, we can get the result of *a*'s announcement for this deal as follows:

```
rus_1 = upd (rus_init 0) a_announce
```

To check that this communication is safe for *a* and *b* we have to check that afterwards it is common knowledge that *c* is still ignorant about the cards of *a* and *b*:

```
check1 = isTrue rus_1 (CK [a,b,c] c_ignorant)
```

Also, after the announcement it should be common knowledge among *a, b* that *b* knows *a*'s cards:

```
check2 = isTrue rus_1 (CK [a,b] b_knows_as)
```

What this means is that *b* can truthfully announce that he knows *c*'s card:

```
rus_2 = upd rus_1 b_announce
```

Again, we have to check that after this public announcement it is common knowledge that *c* is still ignorant about the cards of *a* and *b*:


```
check3 = isTrue rus_2 (CK [a,b,c] c_ignorant)
```

The final check verifies that it is common knowledge among a, b that a knows b 's cards.

```
check4 = isTrue rus_2 (CK [a,b] a_knows_bs)
```

We get:

```
RussianCards> check1
True
RussianCards> check2
True
RussianCards> check3
True
RussianCards> check4
True
```

What the above amounts to is a check of a particular exchange between a and b for the case where the actual deal is 012.345.6. More interesting is a program that computes an appropriate protocol from a given deal, and a correctness check for such a program.

The following function computes the hand of a from an epistemic model, by letting a ‘look at her hand’.

```
hand_a :: EpistM -> [Int]
hand_a pmod = [ i | i <- [0..6], isTrue pmod (K a (Prop (P i))) ]
```

Next, we need to map a 's hand to an appropriate safe announcement of a . An announcement is appropriate if it announces that a knows a disjunction of conjunctions of formulas $(p_i \wedge p_j \wedge p_k)$ with every conjunction $(p_i \wedge p_j \wedge p_k)$ satisfying:

- either i, j, k is a 's hand, or exactly one of i, j, k occurs in a 's hand,

and, moreover, for each $m \in \{0, \dots, 6\}$ it holds that

- each index in $\{0, \dots, 6\} - \{m\}$ occurs at least once in an index triple i, j, k that does not contain m ,

- each index in $\{0, \dots, 6\} - \{m\}$ is absent at least once in an index triple i, j, k that does not contain m .

The following function accomplishes this:

```
hand2a_announce :: [Int] -> PoAM
hand2a_announce hand = public (K a (cmb2form $ dst hand)) where
  cmb2form :: [(Int,Int,Int)] -> Form
  cmb2form triples = Disj
    [ Conj [Prop (P n1), Prop (P n2), Prop (P n3)] | (n1,n2,n3) <- triples ]
  dst [x,y,z] =
    (x,y,z) : zipWith (\ u (v,w) -> (u,v,w)) [x,x,y,z] (cb [x,y,z])
  cb :: [Int] -> [(Int,Int)]
  cb hand = dist ([0..6] \\ hand) where
    dist [x,y,z,u] = [(x,y),(z,u),(x,z),(y,u)]
```

With this function we can compute a 's announcement from a 's knowledge in the model:

```
ms2a_announce :: EpistM -> PoAM
ms2a_announce pmod = hand2a_announce $ hand_a pmod
```


After this announcement, b knows c 's card, so we can compute b 's announcement from b 's knowledge:

```
ms2b_announce :: EpistM -> PoAM
ms2b_announce pmod = public (K b (Prop (R i))) where
  [i] = [ j | j <- [0..6], isTrue pmod (K b (Prop (R j))) ]
```

Now we can play out the generalized version of the protocol.

```
gen_rus_1 :: Integer -> EpistM
gen_rus_1 n = upd (rus_init n) (ms2a_announce (rus_init n))
```

World 100 in the initial model happens to be the world with deal 234.015.6. The result of an appropriate public announcement of a in this world, given by `gen_rus_1 100`, is the following:




In this model 20 situations are still considered possible by either a or c . Note that after a 's announcement b has complete knowledge about the distribution of the cards (there are no uncertainties for b).

For the second step in the protocol, we need:

```
gen_rus_2 :: Integer -> EpistM
gen_rus_2 n = upd (gen_rus_1 n) (ms2b_announce (gen_rus_1 n))
```

The result of the whole protocol, for the world with deal 234.015.6:



Note that there is no card among $\{0, \dots, 5\}$ that c can correctly attribute to its owner:

- 0 is in b 's hand in world 2 and in a 's hand in the other two worlds,
- 1 is in a 's hand in world 0 and in b 's hand in the other two worlds,
- 2 is in b 's hand in world 1 and in a 's hand in the other two worlds,
- ...

And here are the generalized checks to run:

```

gen_check1 :: State -> Bool
gen_check1 n = isTrue (gen_rus_1 n) (CK [a,b,c] c_ignorant)

gen_check2 :: State -> Bool
gen_check2 n = isTrue (gen_rus_1 n) (CK [a,b] b_knows_as)

gen_check3 :: State -> Bool
gen_check3 n = isTrue (gen_rus_2 n) (CK [a,b,c] c_ignorant)

gen_check4 :: State -> Bool
gen_check4 n = isTrue (gen_rus_2 n) (CK [a,b] a_knows_bs)

```

Wrap these all together:

```

all_1 = all gen_check1 [0..139]
all_2 = all gen_check2 [0..139]
all_3 = all gen_check2 [0..139]
all_4 = all gen_check2 [0..139]

check_all = all_1 && all_2 && all_3 && all_4

```

The protocol is correct (the execution of the following command may take a while):

```

RussianCards> check_all
True

```

10.9 Update Semantics

Update semantics was devised as a way of dealing with the semantics of modal expressions such as ‘might’ and ‘must’ [40, 41]. The basic system analyses a special case of public announcement logic, where the knowledge of a single agent is modelled.

```

module UpdateSem

where

import DEMO

-- make sure to set last_agent = a in Models

```

```

fact_update :: Form -> PoAM
fact_update form = public form

may_update :: Form -> PoAM
may_update form = public (Neg (K a (Neg form)))

```

Model with no information about p, q, r :

```

m0 = initE [P 0, Q 0, R 0]

```

“It might rain. It does not rain.”

```
example1 = upds m0 [may_update r, fact_update (Neg r)]
```

“It does rain. It might not rain.”

```
example2 = upds m0 [fact_update r, may_update (Neg r)]
```

The results:

```
UpdateSem> showM m0
==> [0,1,2,3,4,5,6,7]
[0,1,2,3,4,5,6,7]
(0, []) (1, [p]) (2, [q]) (3, [r]) (4, [p,q])
(5, [p,r]) (6, [q,r]) (7, [p,q,r])
(a, [[0,1,2,3,4,5,6,7]])
```

```
UpdateSem> showM example1
==> [0,1,2,3]
[0,1,2,3]
(0, []) (1, [p]) (2, [q]) (3, [p,q])
(a, [[0,1,2,3]])
```

```
UpdateSem> showM example2
==> []
[]

(a, [])
```

10.10 The Protocol of the Dining Cryptographers

The setting of Chaum’s dining cryptographers protocol [7] is a situation where three cryptographers are eating out. At the end of the dinner, they are informed that the bill has been paid, either by one of them, or by NSA (the National Security Agency). Respecting each others rights to privacy, they want to find out whether NSA paid or not, in such a way that in case one of them has paid the bill, the identity of the one who paid is not revealed to the two others.

They decide on the following protocol. Each cryptographer tosses a coin with his righthand neighbour, with the result of the toss remaining hidden from the third person. Each cryptographer then has a choice between two public announcements: that the coins that she has observed

agree or that they disagree. If she has not paid the bill she will say that they agree if the coins are the same and that they disagree otherwise; if she has paid the bill she will say the opposite: she will say that they agree if in fact they are different and she will say that they disagree if in fact they are the same. Clearly, if everyone is speaking the truth, the number of ‘disagree’ announcements will be even. This reveals that NSA has picked up the bill. If one person is lying, the number of ‘disagree’ announcements will be odd, indicating that one among them is paying.

Model checking of this protocol in terms of process theory is described in [39]. In this approach, every aspect of the situation is modelled as a process: the process for coins is defined in terms of processes for heads and for tails, the process for cryptographers following the protocol is defined in terms of their behaviour, and finally the process for the meal is composed from the processes for coins and for cryptographers. The correctness specification is captured in a process that outputs ‘crypt’ if a cryptographer pays and ‘nfa’ if NFA pays. After encoding these processes, a model checker confirms that the process for the whole system is indeed a refinement of the specification, and, thus, that it meets the specification.

An epistemic model checking approach is much more straightforward [17]. One starts with an epistemic situation where the diners have common knowledge of the fact that either NSA or one of them has paid. Next, one updates with the result of the coin tosses, and with communicative acts representing the sharing of information between a cryptographer and his neighbour about these results.

```

module DC
where
import DEMO

-- code by Simona Orzan and Jan van Eijck

-- in Models, set last_agent = c.

```

We use p_1, p_2, p_3 for the secret pay bits, with p_1 indicating that a has paid, p_2 that b has paid, and p_3 that c has paid. The aim of the protocol is that everybody learns whether the formula $p_1 \vee p_2 \vee p_3$ is true or not, but if the formula is true, nobody (except the payer herself) should learn which of the three propositions was true.

The fact that at most one of the cryptographers has paid is expressed as follows:

```

zero_or_one_payer =
  (Disj [Conj [Neg p1, Neg p2, Neg p3],
        Conj [Neg p1, Neg p2, p3],
        Conj [Neg p1, p2, Neg p3],
        Conj [p1, Neg p2, Neg p3]    ])

```

To model the protocol, we need three more propositions q_1, q_2, q_3 representing the result of flipping the coins shared by a and b , b and c , c and a , respectively.

The initial model expresses that at most one cryptographer pays, and that this is public info:

```
dc1 = upd (initE [P 1, P 2, P 3, Q 1, Q 2, Q 3])
      (public zero_or_one_payer)
```

For ease of exposition, we first assume that the result of the coin tosses is $q_1 = \top, q_2 = \top, q_3 = \perp$. We also assume, again for reasons of exposition, that in fact b has picked up the bill.

```
dc2 = upd dc1 (test
  (Conj [Neg p1, p2, Neg p3, q1, q2, Neg q3]))
```

We can assume that it is common knowledge that every participant knows whether she has paid or not:

```
dc3 = upds dc2 [info [a] p1, info [b] p2, info [c] p3 ]
```

The information about the coin tosses is shared in pairs, as follows:

```
dc4 = upds dc3 [info [a,b] q1, info [b,c] q2, info [a,c] q3]
```

Next, everyone makes an appropriate public announcement:

```
xor :: Form -> Form -> Form
xor x y = Neg (equiv x y)

a_statement = public (xor (xor q1 q3) p1)
b_statement = public (xor (xor q1 q2) p2)
c_statement = public (xor (xor q2 q3) p3)

dc5 = upds dc4 [a_statement, b_statement, c_statement]
```





Figure 10.5: The final epistemic state of the DC protocol, for actual situation p_2, q_1, q_2 .

Figure 10.5 shows the final epistemic state, `dc5`. Space does not permit us to list or display the (rather large) intermediate states. The textual representation of `dc5` is:

```
DC> showM dc5
==> [4]
[0,1,2,3,4,5]
(0, [p1,q2]) (1, [p2,q3]) (2, [p3,q1]) (3, [p1,q1,q3]) (4, [p2,q1,q2])
(5, [p3,q2,q3])
(a, [[0],[1,5],[2,4],[3]])
(b, [[0,5],[1],[2,3],[4]])
(c, [[0,4],[1,3],[2],[5]])
```

Here, world 4 is the actual world. This is the world where b has paid and where q_1 and q_2 have value \top . It is clear from the accessibility relations that a cannot distinguish the actual world from world 2 (a world where c has paid), and that c cannot distinguish the actual world from world 0 (a world where a has paid).

The most important properties to be checked in the final state are the fact that everybody learned $p_1 \vee p_2 \vee p_3$ and that a and c don't know that b was the payer:

```
DC> isTrue dc5 (CK [a,b,c] (Disj [p1,p2,p3]))
True
DC> isTrue dc5 (Conj [Neg(K a p2), Neg(K c p2)])
True
```

Now let us move on to the general case. The following function fixes the actual world of `dc1`:

```
gen_dc :: State -> EpistM
gen_dc s = mod2pmod m [s] where m = fst (pmod2mp dc1)
```

This allows us to define generalized updating, with the actual world as parameter.

```
gen_dc_u s =  
  upds (gen_dc s) [info [a] p1, info [b] p2, info [c] p3,  
                    info [a,b] q1, info [b,c] q2, info [a,c] q3,  
                    a_statement, b_statement, c_statement      ]
```

The first general check tests whether those who have not paid do not know who has:

```
gen_check1 s = isTrue (gen_dc_u s)  
  (Conj [Neg p1 'impl' Conj [Neg (K a p2), Neg (K a p3)],  
        Neg p2 'impl' Conj [Neg (K b p1), Neg (K b p3)],  
        Neg p3 'impl' Conj [Neg (K c p1), Neg (K c p2)]])
```

If one of the diners has paid, this is common knowledge:

```
gen_check2 s =  
  isTrue (gen_dc_u s)  
    (Disj [p1,p2,p3] 'impl' CK [a,b,c] (Disj [p1,p2,p3]))
```

Here are all checks lumped together:

```
all_checks = all gen_check1 [0..31] && all gen_check2 [0..31]
```

We get:

```
DC> all_checks  
True
```

This establishes the correctness of the dining cryptographers protocol.

10.11 A Measure for Ignorance

Looking at some update examples, we see the size of the models, measured in terms of number of transitions in the labelled transition component, may increase as we perform an update.

```
initM = initE [P 1, P 2, P 3, P 4]
upM    = upd initM (groupM [a,b] p1)
```

```
DEMO> length (access (fst (pmod2mp initM)))
1024
DEMO> length (access (fst (pmod2mp upM)))
1792
```

What this shows is that ‘number of transitions’ is *not* a good measure for the size of an epistemic model.

For KD45 (and hence S5) models the following very simple alternative is much more attractive. For each agent, use the

number of worlds still to be eliminated from the balloon pointed at by the actual world of the model or from the partition block containing the actual world of the model

as a measure of ignorance for that agent. This is implemented as follows:

```
measure :: (Eq a, Ord a) => (Model a b, a) -> Maybe [Int]
measure (m,w) =
  let
    f          = filter (\ (us,vs) -> elem w us || elem w vs)
    g [(xs,ys)] = length ys - 1
  in
    case kd45 (domain m) (access m) of
      Just a_balloons -> Just
        ( [ g (f balloons) | (a,balloons) <- a_balloons ] )
      Nothing          -> Nothing
```

We now get that the measure of ignorance of a model decreases with each epistemic update:

```
DEMO> measure (head $ decompose initM)
Just [15,15,15,15]
DEMO> measure (head $ decompose upM)
Just [7,7,23,23]
```

This may still be too crude.

To Do 6 *Refine the measure by distinguishing between factual measure, measure up to epistemic depth 1, and so on. The factual measure should count the worlds modulo “having the same valuation” (making the same factual statements true), the measure up to depth 1 should count the worlds modulo “being bisimilar up to depth 1 (making the same formulas with epistemic depth up to 1 true), and so on. How do the results compare with the implemented version of ‘measure’?*

To Do 7 *Investigate the general situation. Call an update on S5 or KD45 models honest if it does not increase the measure. What are the honest updates? Can we prove that these are precisely the updates that do not involve lying?*

10.12 Finding Axiom Schemes for Logics of Communication

Find an axiom scheme for the interaction of updates with epistemic preconditions and epistemic formulas:

```
DEMO> tr (Up (public (K a p)) (K a p))
[C[?[a]p,a]]v[p,&[-p,-[a]p]]
DEMO> tr (Up (public (K a p)) (Up (public (K a p)) (K a p)))
[U[C[?[a]p,C[?[a]p,a]]]v[p,&[-p,[a]p,-[a]p],&[-p,-[a]p]]
DEMO> step (public (K a p)) (Ag a) p
[C[?[a]p,a]]A[0]p
DEMO> step (public (K a p)) (Ag a) q
[C[?[a]p,a]]A[0]q
```

Find an axiom scheme for the interaction of public announcement and common knowledge:

```
DEMO> tr (Up (public p) (Pr (Star (Ags [a,b])) p))
T
DEMO> tr (Up (public p) (Pr (Star (Ags [a,b])) q))
[(U[?T,C[?p,[a,b]]])*v[&[p,q],-p]]
DEMO> step (public p) (Star (Ags [a,b])) q
[(U[?T,C[?p,[a,b]]])*]A[0]q
```

Find an axiom scheme for the interaction of secret group communication (email CC) and common knowledge:

```
DEMO> tr (Up (secret [a,b] p) (Pr (Star (Ags [a,b])) p))
[C[C[(U[?T,C[?p,[a,b]]])*],?-T],(U[U[?T,[a,b]],C[?-T,C[?p,[a,b]]])*],?-T]]*]]p
DEMO> step (secret [a,b] p) (Star (Ags [a,b])) p
&[[C[C[(U[?T,C[?p,[a,b]]])*],?-T],(U[U[?T,[a,b]],C[?-T,(U[?T,C[?p,[a,b]]])*],?-T]]*]]A[1]p,
[U[(U[?T,C[?p,[a,b]]])*],C[C[(U[?T,C[?p,[a,b]]])*],?-T],
```

```

(U[U[?T, [a, b]], C[?-T, (U[?T, C[?p, [a, b]])]*, ?-T]])*,
C[?-T, (U[?T, C[?p, [a, b]])]*]]]A[0]p]
DEMO> tr (Up (secret [a, b] p) (Pr (Star (Ags [a, b])) q))
&[[C[C[(U[?T, C[?p, [a, b]])]*, ?-T], (U[U[?T, [a, b]], C[?-T, (U[?T, C[?p, [a, b]])]*, ?-T]])*]]q,
[U[(U[?T, C[?p, [a, b]])]*, C[C[(U[?T, C[?p, [a, b]])]*, ?-T],
(U[U[?T, [a, b]], C[?-T, (U[?T, C[?p, [a, b]])]*, ?-T]])*,
C[?-T, (U[?T, C[?p, [a, b]])]*]]]v[&[p, q], -p]]

```

Find an axiom scheme for the interaction of group announcement and common knowledge:

```

DEMO> tr (Up (groupM [a, b] p) (Pr (Star (Ags [b, c])) q))
&[[C[C[(U[?T, C[?p, [b, c]])]*, C[?p, [c]]], (U[U[?T, [b, c]], C[c, (U[?T, C[?p, [b, c]])]*,
C[?p, [c]])]*]]q, [U[(U[?T, C[?p, [b, c]])]*, C[C[(U[?T, C[?p, [b, c]])]*, C[?p, [c]]],
(U[U[?T, [b, c]], C[c, (U[?T, C[?p, [b, c]])]*, C[?p, [c]])]*]]*]]]v[&[p, q], -p]]

```

And so on.

Acknowledgement This report and the tool that it describes were prompted by a series of questions voiced by Johan van Benthem in his talk at the annual meeting of the Dutch Association for Theoretical Computer Science, in Utrecht, on March 5, 2004. Thanks to Johan van Benthem, Hans van Ditmarsch, Barteld Kooi and Ji Ruan for valuable feedback and inspiring discussion.

Appendix A

Special Treatment for S5, KD45 and K45

In the introduction we saw in what sense S5, KD45 and K45 are special in an epistemic update setting. This appendix proves some useful facts about relations that we will employ to give S5, KD45 and K45 models special treatment.

Since equivalence relations are euclidean and serial, every S5 model is a KD45 model. Also, obviously, every KD45 model is a K45 model. This means that we can combine the tests for S5, KD45, and K45. For that, we first explore the relation between S5, KD45 and K45 in some detail.

If R is a relation, then a point x is called *isolated in R* if $\neg\exists y \ yRx$ and $\neg\exists y \ xRy$.

Theorem 7 *Removal of the isolated points from a K45 relation creates a KD45 relation.*

Proof. Suppose R is euclidean and transitive on X . We have to show that R is euclidean, transitive and serial on $Y = X - \{x \mid \neg\exists y \ yRx \wedge \neg\exists y \ xRy\}$. Euclideaness and transitivity are obvious. For seriality, take $y \in Y$. By definition of Y , y is not isolated for R . So either there is a $z \in Y$ with yRz or there is a $z \in Y$ with zRy . In the first case we are done. In the second case, yRy follows from zRy by euclideaness. \square

Theorem 8 *Adding a set of isolated points to a KD45 relation creates a K45 relation.*

Proof. Isolated points are trivially transitive and euclidean. \square

If R is a relation, then a pair $(x, y) \in R$ is called an *entry pair* if $(y, x) \notin R$.

Theorem 9 *Removing the set of entry pairs from a KD45 relation creates an equivalence relation.*

Proof. Let R be transitive, euclidean and serial on X . We have to show that

$$S = R - \{(x, y) \mid xRy \wedge \neg yRx\}$$

is an equivalence on X .

- S is transitive: Assume xSy, ySz . Since $S \subseteq R$, it follows that xRy, yRz , and by transitivity of R , xRz . By definition of S , it follows from xSy that yRx and from ySz that zRy . Again by transitivity of R , zRx . From xRz, zRx and the definition of S : xSz .

- S is symmetric: immediate from the definition.
- S is reflexive: Assume $x \in X$. By seriality of R , there is a $y \in X$ with xRy . By symmetry and transitivity of R , xRx . By definition of S , xSx .

□

If \sim is an equivalence on X , $x \in X$ and $y \notin X$, then let

$$y_{\sim x} = \{(y, z) \mid z \in [x]_{\sim}\}.$$

The pairs in $y_{\sim x}$ are the entry-pairs from y to the members of the $[x]_{\sim}$ block of the partition induced by \sim . Adding such sets of entry-pairs to an equivalence creates a KD45 relation. More precisely:

Theorem 10 *If \sim is an equivalence on X , $X \cap Y = \emptyset$, and $f : Y \rightarrow X$, then*

$$\sim \cup \bigcup \{y_{\sim f(y)} \mid y \in Y\}$$

is a KD45 relation on $X \cup Y$.

Proof. Suppose \sim is an equivalence on X , $Y \cap X = \emptyset$, and $f : Y \rightarrow X$. Let

$$R = \sim \cup \bigcup \{y_{\sim f(y)} \mid y \in Y\}.$$

We have to show that R is euclidean, serial and transitive.

- R is euclidean: Let xRy and xRz . We have to show yRz . Suppose $x \in X$. Then, by construction of R , $x \sim y$ and $x \sim z$. Thus $y \sim z$ by euclideaness of \sim , and hence yRz . Suppose $x \notin X$. Then $xRy \in x_{\sim f(x)}$ and $xRz \in x_{\sim f(x)}$, and therefore $y, z \in [f(x)]_{\sim}$. It follows that $y \sim z$, and hence yRz .
- R is serial: immediate from the construction of R .
- R is transitive. Similar to the reasoning for euclideaness.

□

The upshot of the above is the following:

- Any S5 relation can be represented by the partition it induces.
- Any KD45 relation can be represented by a barbed partition (a partition where blocks may be barbed by loose entry points) or a set of balloons (a partition where each partition block is held on a string of the entry points into the block).
- Any K45 relation can be represented by a barbed partition plus a set of isolated points, or by a set of balloons plus a set of isolated points.

Representing sets as lists, we can use the type $[[a]]$ (the type of lists of lists) for partitions, we can use the type $[([a], [a])]$ (the type of lists of list pairs) for sets of balloons, and the type $([a], [([a], [a])])$ (the type of pairs consisting of a list and a list of list pairs) for sets of isolated points together with sets of balloons.

Appendix B

The DPLL prover

Implementation of Davis, Putnam, Logemann, Loveland (DPLL) theorem proving [9, 10] for propositional logic. The implementation uses discrimination trees or *tries*, following [42].

B.1 Module Declaration

```
module DPLL

where

import List
```

B.2 Clauses, Clause Sets

If we let variables be represented by their indices, with the convention that positive indices represent positive literals and the negation of an index represents the negation of the variable, then clauses can be represented as integer lists, and clause sets as lists of integer lists.

```
type Clause    = [Integer]
type ClauseSet = [Clause]
```

A valuation is simply a list of integers:

```
type Valuation = [Integer]
```


Reorder the literals in a clause to make sure that lowest variable indices are listed first. Also, throw out duplicate literals from clauses and duplicate clauses from clause sets.

```
rSort :: ClauseSet -> ClauseSet
rSort = (srt1.nub) . (map (srt2. nub))
  where srt1 = sortBy cmp
        cmp [] ( _:_) = LT
        cmp [] []     = EQ
        cmp ( _:_) [] = GT
        cmp (x:xs) (y:ys) | (abs x) < (abs y) = LT
                          | (abs x) > (abs y) = GT
                          | (abs x) == (abs y) = cmp xs ys
        srt2 = sortBy (\ x y -> compare (abs x) (abs y))
```

A clause is trivial if the clause contains both positive and negative occurrences of some literal.

```
trivialC :: Clause -> Bool
trivialC [] = False
trivialC (i:is) = elem (-i) is || trivialC is
```

The function `clsNub` removes trivial clauses from a clause set.

```
clsNub :: ClauseSet -> ClauseSet
clsNub = filter (not.trivialC)
```

B.3 Tries

The datatype of discrimination trees.

```
data Trie = Nil | End | Tr Integer Trie Trie Trie deriving (Eq,Show)
```

`Nil` is the empty clause set, `End` a marker for a clause end.

In a trie of the form (Tr, v, P, N, R) , P encodes a clause set $\{P_1, \dots, P_n\}$ representing $\{v \vee P_1, \dots, v \vee P_n\}$, N a clause set $\{N_1, \dots, N_m\}$ representing $\{\neg v \vee N_1, \dots, \neg v \vee N_m\}$, and R a clause set $\{R_1, \dots, R_k\}$ with none of the R_i containing occurrences of v .

If the clause set corresponding to P equals \square , and the clause set corresponding to N equals \square , this means that both $\{v\}$ and $\{\neg v\}$ occur in the clause set, i.e., that the clause set is a contradiction (equals \square).

If the clause sets corresponding to P and Q are both empty, this means that v does not occur (positively or negatively) in the clause set.

If the clause set corresponding to R equals \square , this means that the whole clause set equals \square . If the clause set corresponding to R is empty, this means that the clause set does not contain clauses without positive or negative occurrences of v .

Bearing this in mind, the following operation can be used to perform some simplifications.

```
nubT :: Trie -> Trie
nubT (Tr v p n End) = End
nubT (Tr v End End r) = End
nubT (Tr v Nil Nil r) = r
nubT tr = tr
```

The trie merge operation (conjunction of the corresponding clause sets):

```
trieMerge :: Trie -> Trie -> Trie
trieMerge End _ = End
trieMerge _ End = End
trieMerge t1 Nil = t1
trieMerge Nil t2 = t2
trieMerge t1@(Tr v1 p1 n1 r1) t2@(Tr v2 p2 n2 r2)
  | v1 == v2 = (Tr
                v1
                (trieMerge p1 p2)
                (trieMerge n1 n2)
                (trieMerge r1 r2)
                )
  | v1 < v2 = (Tr
                v1
                p1
                n1
                (trieMerge r1 t2)
                )
  | v1 > v2 = (Tr
                v2
                p2
                n2
                (trieMerge r2 t1)
                )
```

Assuming clauses are r-sorted, mapping clause sets into ordered tries is done as follows:

```

cls2trie :: ClauseSet -> Trie
cls2trie [] = Nil
cls2trie ([]:_) = End
cls2trie cls@((i:is):_) =
  let j = abs i in
  (Tr
   j
   (cls2trie [ filter (/= j) cl | cl <- cls, elem j cl ])
   (cls2trie [ filter (/= -j) cl | cl <- cls, elem (-j) cl ])
   (cls2trie [ cl | cl <- cls, notElem j cl, notElem (-j) cl ]))

```

Tries are mapped back into clause sets by:

```

trie2cls :: Trie -> ClauseSet
trie2cls Nil = []
trie2cls End = [[]]
trie2cls (Tr i p n r) =
  [ i:rest | rest <- trie2cls p ]
++
  [ (-i):rest | rest <- trie2cls n ]
++
  trie2cls r

```

B.4 Unit Subsumption and Unit Resolution

Unit clause detection in the trie format; the following function finds all unit clauses:

```

units :: Trie -> [Integer]
units Nil = []
units End = []
units (Tr i End n r) = i : units r
units (Tr i p End r) = -i : units r
units (Tr i p n r) = units r

```

Unit propagation:

```

unitProp :: (Valuation,Trie) -> (Valuation,Trie)
unitProp (val,tr) = (nub (val ++ val'), unitPr val' tr)
  where
    val' = units tr
    unitPr :: Valuation -> Trie -> Trie
    unitPr [] tr = tr
    unitPr (i:is) tr = unitPr is (unitSR i tr)

```

Unit subsumption and resolution.

```

unitSR :: Integer -> Trie -> Trie
unitSR i = (unitR pol j) . (unitS pol j)
  where pol = i>0
        j   = abs i

```

Unit subsumption: delete every clause containing the literal. The literal is encoded as (pol,i) , where pol gives the sign and i is a positive integer giving the index of the variable.

```

unitS :: Bool -> Integer -> Trie -> Trie
unitS pol i Nil = Nil
unitS pol i End = End
unitS pol i tr@(Tr j p n r) | i == j = if pol
                                then nubT (Tr j Nil n r)
                                else nubT (Tr j p Nil r)
    | i < j = tr
    | i > j = nubT (Tr
                    j
                    (unitS pol i p)
                    (unitS pol i n)
                    (unitS pol i r)
                    )

```

Unit resolution: delete the mate of the literal from every clause containing it.

```

unitR :: Bool -> Integer -> Trie -> Trie
unitR pol i Nil = Nil
unitR pol i End = End
unitR pol i tr@(Tr j p n r) | i == j = if pol
                                then
                                    nubT (Tr
                                        j
                                        p
                                        Nil
                                        (trieMerge n r)
                                    )
                                else
                                    nubT (Tr
                                        j
                                        Nil
                                        n
                                        (trieMerge p r)
                                    )
    | i < j = tr
    | i > j = nubT (Tr
                    j
                    (unitR pol i p)
                    (unitR pol i n)
                    (unitR pol i r)
                )

```

B.5 Splitting

To treat splitting, we need functions for setting variables to true and false, respectively. Setting a variable to true is done by:

```

setTrue :: Integer -> Trie -> Trie
setTrue i Nil = Nil
setTrue i End = End
setTrue i tr@(Tr j p n r) | i == j = trieMerge n r
    | i < j = tr
    | i > j = (Tr
                j
                (setTrue i p)
                (setTrue i n)
                (setTrue i r)
            )

```

Setting a variable to false is done similarly:

```

setFalse :: Integer -> Trie -> Trie
setFalse i Nil = Nil
setFalse i End = End
setFalse i tr@(Tr j p n r) | i == j = trieMerge p r
                           | i < j  = tr
                           | i > j  = (Tr
                                       j
                                       (setFalse i p)
                                       (setFalse i n)
                                       (setFalse i r)
                                       )

```

Always split on the first variable:

```

split :: (Valuation,Trie) -> [(Valuation,Trie)]
split (v,Nil) = [(v,Nil)]
split (v,End) = []
split (v, tr@(Tr i p n r)) = [(v++[i], setTrue i tr),
                              (v++[-i],setFalse i tr)]

```

B.6 DPLL

Davis, Putnam, Loveland, Longeman (DPLL): keep splitting after unit propagation.

```

dpll :: (Valuation,Trie) -> [(Valuation,Trie)]
dpll (val,Nil) = [(val,Nil)]
dpll (val,End) = []
dpll (val,tr) =
  concat [ dpll vt | vt <- (split.unitProp) (val,tr) ]

```

Wrap it all up:

```

dp :: ClauseSet -> [(Valuation,Trie)]
dp cls = dpll ([], (cls2trie . rSort) (clsNub cls))

```

Bibliography

- [1] BALTAG, A. A logic for suspicious players: epistemic action and belief-updates in games. *Bulletin of Economic Research* 54, 1 (2002), 1–45.
- [2] BALTAG, A., MOSS, L., AND SOLECKI, S. The logic of public announcements, common knowledge, and private suspicions. Tech. Rep. SEN-R9922, CWI, Amsterdam, 1999.
- [3] BALTAG, A., MOSS, L., AND SOLECKI, S. The logic of public announcements, common knowledge, and private suspicions. Tech. rep., Dept of Cognitive Science, Indiana University and Dept of Computing, Oxford University, 2003.
- [4] BENTHEM, J. v. Language, logic, and communication. In *Logic in Action*, J. van Benthem, P. Dekker, J. van Eijck, M. de Rijke, and Y. Venema, Eds. ILLC, 2001, pp. 7–25.
- [5] BENTHEM, J. v. One is a lonely number: on the logic of communication. Tech. Rep. PP-2002-27, ILLC, Amsterdam, 2002.
- [6] BLACKBURN, P., DE RIJKE, M., AND VENEMA, Y. *Modal Logic*. Cambridge Tracts in Theoretical Computer Science. Cambridge University Press, 2001.
- [7] CHAUM, D. The dining cryptographers problem: unconditional sender and receiver untraceability. *Journal of Cryptology* 1 (1988), 65–75.
- [8] CHELLAS, B. *Modal Logic: An Introduction*. Cambridge University Press, 1980.
- [9] DAVIS, M., LOGEMANN, G., AND LOVELAND, D. A machine program for theorem proving. *Communications of the ACM* 5, 7 (1962), 394–397.
- [10] DAVIS, M., AND PUTNAM, H. A computing procedure for quantification theory. *Journal of the ACM* 7, 3 (1960), 201–215.
- [11] DITMARSCH, H. v. *Knowledge Games*. PhD thesis, ILLC, Amsterdam, 2000.
- [12] DITMARSCH, H. v. The Russian card problem. *Studia Logica* 75 (2003), 31–62.
- [13] DITMARSCH, H. v., RUAN, J., AND VERBRUGGE, R. Model checking sum and product. Tech. rep., Computer Science Department, University of Otago, New Zealand, 2005.
- [14] DITMARSCH, H. v., VAN DER HOEK, W., VAN DER MEYDEN, R., AND RUAN, J. Model checking Russian cards. To appear in Proceedings of MoChArt 05, 2005.
- [15] EIJCK, J. v. Communicative actions. CWI, Amsterdam, 2004.
- [16] EIJCK, J. v. Reducing dynamic epistemic logic to PDL by program transformation. Tech. Rep. SEN-E0423, CWI, Amsterdam, December 2004. Available from <http://db.cwi.nl/rapporten/>.
- [17] EIJCK, J. v., AND ORZAN, S. Modelling the epistemics of communication with functional programming. accepted for TFP’05, 2005.

- [18] ELJCK, J. V., AND RUAN, J. Action emulation. CWI, Amsterdam, www.cwi.nl/~papers/04/ae, 2004.
- [19] FAGIN, R., HALPERN, J., MOSES, Y., AND VARDI, M. *Reasoning about Knowledge*. MIT Press, 1995.
- [20] GERBRANDY, J. *Bisimulations on planet Kripke*. PhD thesis, ILLC, 1999.
- [21] GERBRANDY, J. Dynamic epistemic logic. In *Logic, Language and Information, Vol. 2*, L. Moss et al., Eds. CSLI Publications, Stanford, 1999.
- [22] GOLDBLATT, R. *Logics of Time and Computation, Second Edition, Revised and Expanded*, vol. 7 of *CSLI Lecture Notes*. CSLI, Stanford, 1992 (first edition 1987). Distributed by University of Chicago Press.
- [23] HAREL, D., KOZEN, D., AND TIURYN, J. *Dynamic Logic. Foundations of Computing*. MIT Press, Cambridge, Massachusetts, 2000.
- [24] HINTIKKA, J. *Knowledge and Belief: An Introduction to the Logic of the Two Notions*. Cornell University Press, Ithaca N.Y., 1962.
- [25] HOLLENBERG, M. *Logic and Bisimulation*. PhD thesis, Utrecht University, 1998.
- [26] HOMMERSOM, A., MEYER, J.-J., AND VINK, E. D. Update semantics of security protocols. *Synthese* 142 (2004), 229–267. Knowledge, Rationality and Action subseries.
- [27] J.E.HOPCROFT. An $n \log n$ algorithm for minimizing states in a finite automaton. In *Theory of Machines and Computations*, Z. Kohavi and A. Paz, Eds. Academic Press, 1971.
- [28] JONES, S. P., HUGHES, J., ET AL. Report on the programming language Haskell 98. Available from the Haskell homepage: <http://www.haskell.org>, 1999.
- [29] KNUTH, D. *Literate Programming*. CSLI Lecture Notes, no. 27. CSLI, Stanford, 1992.
- [30] KOOI, B., AND VAN BENTHEM, J. Reduction axioms for epistemic actions. In *AiML-2004: Advances in Modal Logic* (2004), R. Schmidt, I. Pratt-Hartmann, M. Reynolds, and H. Wansing, Eds., no. UMCS-04-9-1 in Technical Report Series, University of Manchester, pp. 197–211.
- [31] KOOI, B. P. *Knowledge, Chance, and Change*. PhD thesis, Groningen University, 2003.
- [32] KOUTSOFIOS, E., AND NORTH, S. Drawing graphs with *dot*. Available from <http://www.research.att.com/~north/graphviz/>.
- [33] LOWE, G. Breaking and fixing the Needham-Schroeder public-key protocol using FDR. In *Tools and Algorithms for the Construction and Analysis of Systems (TACAS)* (1996), vol. 1055, Springer-Verlag, Berlin Germany, pp. 147–166.
- [34] PAIGE, R., AND TARJAN, R. E. Three partition refinement algorithms. *SIAM J. Comput.* 16, 6 (1987), 973–989.
- [35] PAPADIMITRIOU, C. *Computational Complexity*. Addison-Wesley, 1994.
- [36] PEREIRA, O. Modelling and security analysis of authenticated group key agreement protocols, 2003.
- [37] ROSCOE, A. W. Modelling and verifying key-exchange protocols using CSP and FDR. In *Proc. 8th IEEE Computer Security Foundations Workshop* (1995), pp. 98–107.
- [38] RUAN, J. Exploring the update universe. Master’s thesis, ILLC, Amsterdam, 2004.
- [39] SCHNEIDER, S., AND SIDIROPOULOS, A. CSP and anonymity. In *Proc. ESORICS’96* (1996), LNCS 1146, pp. 198–218.
- [40] VELTMAN, F. Data semantics and the pragmatics of indicative conditionals. In *On Conditionals*, Traugott, ter Meulen, Reiley, and Ferguson, Eds. Cambridge University Press, 1986, pp. 147–168.

- [41] VELTMAN, F. Defaults in update semantics. *Journal of Philosophical Logic* (1996), 221–261.
- [42] ZHANG, H., AND STICKEL, M. E. Implementing the Davis-Putnam method. *Journal of Automated Reasoning* 24, 1/2 (2000), 277–296.

Index

- $C_B\varphi$, 43
- $E_B\varphi$, 43
- $K_{ijk}^A(\pi)$, 56
- T_{ij}^A , 55
- $\Box_a\varphi$, 43
- \oplus , 14
- AUT, 15
- \uplus , 14
- 0, 14, 55
- 1, 13
- a, 18
- accBl, 64
- accBlocks, 39
- access, 21
- accNFA, 63
- action emulation, 70
- action models, 12
- action update, 12
- aePmod, 73
- Agent, 18
- alice, 18
- all_agents, 19
- alternatives, 22
- AM, 54
- APDL, 67
- aut, 67
- aut', 66
- awareness, 98
- b, 18
- Baltag, Alexandru, 10
- Benthem, Johan van, 10
- bisim, 41
- bisimPmod, 41
- bl, 39
- Blackburn, Patrick, 10
- bob, 18
- bot, 48
- c, 18
- canonF, 52
- Caps, 93
- capsInfo, 93
- carol, 18
- cds2trie, 141
- cf, 49
- Chaum, David, 129
- Chellas, B., 10
- cKnown, 66
- Clause, 139
- ClauseSet, 139
- closure, 22
- clsNub, 140
- cmp, 7, 84
- cmp1, 32
- cmpP, 83
- cmpPoAM, 84
- commonAlts, 75
- composition
 - of action models, 13
- comprC, 46
- compress, 65
- computePre, 84
- consistency, 9
- consistent, 52
- containedIn, 23
- contrad, 52
- conv, 40
- convAut, 65
- convert, 40
- convPmod, 41
- d, 18
- dave, 18
- Davis, Martin, 139
- decompose, 20
- deMorgan, 51
- display, 29
- displayB, 30
- displayM, 30
- displayP, 30
- displayPB, 31
- Ditmarsch, Hans van, 10, 103, 110, 120
- domain, 21
- dot, 2

- dp, 51, 145
- dp11, 145
- DPLL procedure, 145
- e, 18
- e0, 3, 79
- e00, 79
- entryPair, 26
- EpistM, 54
- equiv, 44
- equiv2part, 25
- equivalenceR, 24
- ernie, 18
- euclidean relations, 8
- euclideanR, 25
- eval, 21
- expand, 22
- expansion of a relation, 22
- Fagin, R., 10
- finite automata, 14
- fix, 86
- Form, 43
- Freudenthal, H., 102
- fuseLists, 51
- genKnown, 66
- Gerbrandy, Jelle, 10
- glueWith, 31
- Graphviz, 31
- graphviz, 31, 34
- group message, 16
- group revelation, 17
- groupAlts, 75
- groupM, 7, 80
- gsm, 21
- gsmPoAM, 55
- Halpern, Joe, 10
- Hintikka, J., 8
- Hoek, Wiebe van der, 120
- idR, 24
- impl, 44
- individual message, 16
- individual revelation, 16
- info, 83
- initE, 2, 78
- initPart, 64
- initPartition, 38
- initRefine, 39
- instance, 32
- introspection
 - negative, 8
 - positive, 8
- isolated, 26
- isS5, 25
- isTrAt, 78
- isTrue, 4, 78
- isTrueAt, 77
- K45, 10, 137
- k45, 27
- k45PointsBalloons, 26
- k45R, 26
- KD45, 9, 137
- kd45, 28
- kd45Balloons, 27
- kd45psbs2balloons, 28
- kd45R, 26
- kleene, 57
- Kleene path lemma, 56
- knowledge distribution, 8, 9
- Koymans, Karst, 106
- kvbtr, 68
- labelled transition systems, 11
- last_agent, 19
- links, 32
- listPState, 33
- listState, 31
- Logemann, G., 139
- logic
 - of consistent belief, 9
 - of knowledge, 8
- lookupFs, 38
- Loveland, D., 139
- LTSs, 11
- mapping, 49
- measure, 134
- message, 3, 81
- Meyden, R. van der, 120
- minimalAut, 65
- minimalAut', 65
- minimalModel, 39
- minimalPmod, 40
- minimizing finite automata, 63
- minPmod, 72
- mod2pmod, 20
- Model, 19
- Moses, Y., 10
- Moss, Larry, 10
- Move, 62
- Muddy, 98

- ndS, 89
- ndSum, 89
- ndSum', 88
- negation, 44
- negative introspection, 8, 9
- NFA, 14
- nka_p, 48
- nkanp, 48
- nkap, 48
- nubT, 141
- nullAut, 66

- one, 83
- Orzan, Simona, 92, 130

- p, 48
- p0, 48
- p1, 48
- p2, 48
- p3, 48
- p4, 48
- p5, 48
- p6, 48
- pairs2rel, 25
- partition refinement, 37
- PDL
 - language, 11
- Pmod, 19
- pmod2mp, 20
- PoAM, 54
- points, 21
- positive introspection, 8, 9
- pow, 86
- powerList, 79
- precondition, 54
- preconditions, 54
- preds, 72
- Program, 43
- program simplification, 45
- program transformation lemma, 56
- Prop, 42
- propEquiv, 51
- psPartition, 72
- public, 6, 79
- public announcement, 16
- public announcement logic
 - language, 17
- pureProp, 48
- Putnam, Hilary, 139

- q, 48
- q0, 48
- q1, 48
- q2, 48
- q3, 48
- q4, 48
- q5, 48
- q6, 48

- r, 48
- r0, 48
- r1, 48
- r2, 48
- r3, 48
- r4, 48
- r5, 48
- r6, 48
- racheableAut, 76
- reachable, 63
- recog, 62
- reduction of dynamic epistemic logic to PDL, 13
- refine, 39
- refinePart, 64
- refinePartition, 38
- reflexive relations, 8
- reflR, 24
- rel2part, 20
- relKnown, 66
- reveal, 82
- revelation
 - group, 17
 - individual, 16
- Rijke, Maarten de, 10
- rpr, 61
- rSort, 140
- Ruan, Ji, 92, 103, 120

- S5, 8, 137
- s5ball2part, 28
- sameAccB1, 64
- sameAccBlocks, 38
- sameVal, 76
- satVals, 51
- secret, 8, 81
- secret group communication, 16
- secret individual communication, 16
- serial relations, 9
- serialR, 25
- setFalse, 144
- setTrue, 144
- showM, 2, 29
- showMo, 29
- showMs, 30
- simpl, 46

- SM, 53
- Solecki, S., 10
- split, 145
- splitU, 45
- star, 86
- State, 53
- states, 62
- step, 60
- step0, 59
- step1, 59
- Stickel, M.E., 139
- sucs, 72
- sum of action models, 14
- Symbol, 62
- symbols, 62
- symR, 24

- t, 60
- t', 60
- table2fct, 20
- tAut, 69
- test, 16
- test, 81
- testAnnounce, 89
- tfm, 58
- tr, 61
- tr', 67
- transf, 56
- transitive relations, 8
- translating update PDL to PDL, 59
- transparent informedness, 17
- transR, 24
- Trie, 140
- trie2cls, 142
- trieMerge, 141
- trivialC, 140
- truthfulness, 8

- u, 48
- unfold, 71
- unit
 - for composition, 13
 - for sum, 14
- unit resolution, 142
- unit subsumption, 142
- unitProp, 142
- unitR, 143
- unitS, 143
- units, 142
- unitSR, 143
- up, 75
- upd, 3, 76

- update PDL
 - language, 12
- upds, 4, 76

- Valuation, 139
- valuation, 54
- Van Benthem test, 86
- Vardi, Moshe, 10
- vBfix, 86
- vBtest, 86
- Veltman, Frank, 128
- Venema, Yde, 10
- Verbrugge, Rineke, 103
- version, 92

- writeGr, 35
- writeGraph, 35
- writeModel, 35
- writeP, 2, 36
- writePmod, 36

- zero, 55
- Zhang, H., 139